

Utrecht University
Master Thesis
Computing Science & Mathematical Sciences
Generic Numerical Representations as Ornaments

Samuel Klumpers
6057314

Supervisors
dr. Wouter Swierstra
dr. Paige North

December 5, 2023

Abstract

The concept of numerical representations as defined by Okasaki [Oka98] explains how certain datastructures resemble number systems, and motivates how number systems can be used as a basis to design datastructures, which can be made precise by using McBride’s ornaments [McB14]. In order to study a broad spectrum of indexed and unindexed numerical representations, we encode a universe allowing the expression of nested datatypes, and the internalization of descriptions of composite types. By equipping the universe with metadata, number systems and numerical representations can be described in the same setup. Adapting ornaments to this universe allows us to define two generic ornament computing functions, generalizing well-known sequences of ornaments, such as naturals-lists-vectors, by computing the indexed and unindexed numerical representations as a sequence of ornaments on a number system.

Contents

| | | |
|----------|---|-----------|
| 1 | Introduction | 4 |
| 2 | Background | 5 |
| 2.1 | Agda | 6 |
| 2.2 | Datatypes in Agda | 6 |
| 2.3 | Proving in Agda | 9 |
| 2.4 | Descriptions | 10 |
| 2.4.1 | Finite types | 11 |
| 2.4.2 | Recursive types | 12 |
| 2.4.3 | Sums of products | 12 |
| 2.4.4 | Parametrized types | 13 |
| 2.4.5 | Indexed types | 16 |
| 2.4.6 | Generic Programming | 18 |
| 2.5 | Ornaments | 19 |
| 2.6 | Ornamental Descriptions | 23 |
| 3 | Descriptions | 25 |
| 3.1 | Numerical Representations | 25 |
| 3.2 | Room for Improvement | 27 |
| 3.2.1 | Number systems | 27 |
| 3.2.2 | Nested types | 29 |
| 3.2.3 | Composite types | 31 |
| 3.2.4 | Hiding variables | 31 |
| 3.3 | A new Universe | 32 |
| 3.3.1 | Annotating Descriptions with Metadata | 35 |
| 4 | Ornaments | 37 |
| 4.1 | Ornamental descriptions | 37 |
| 5 | Generic Numerical Representations | 42 |
| 5.1 | Unindexed Numerical Representations | 42 |
| 5.2 | Indexed Numerical Representations | 45 |
| 6 | Conclusion and Discussion | 46 |
| 6.1 | Σ -descriptions are more natural for expressing finite types | 47 |
| 6.2 | Branching numerical representations | 48 |
| 6.3 | Indices do not depend on parameters | 49 |
| 6.4 | No RoseTrees | 49 |
| 6.5 | No levitation | 50 |
| 6.6 | Metadata more tasteful externally than internally | 50 |
| 6.7 | δ is conservative | 50 |
| 6.8 | No sparse numerical representations | 51 |
| | Appendices | 55 |
| A | Folding | 55 |
| B | Folding without Axiom K | 55 |
| C | Nested types as uniformly recursive indexed types | 57 |

1 Introduction

There is a close relation between *number systems* and *datastructures* containing certain numbers of elements. Examples of datastructures designed to resemble a number system, are explored in Okasaki's Purely Functional Data Structures ([Oka98], chapter 9) as *numerical representations*, relating some known datastructures to their underlying number system.

To illustrate such an example, consider the binary numbers in their bijective, or zeroless, form (least significant digit first):

```
data Bin : Type where
  0b      : Bin
  1b_ 2b_ : Bin → Bin
```

This definition declares that a binary number is either formed by `0b`, or by prepending either `1b` or `2b`. Here, `0b` represents the number 0, while `1b n` corresponds to $2n + 1$, representing the positive odd numbers, and `2b` corresponds to $2n + 2$, representing the positive even numbers. As a *positional number system*, `Bin` has digits 1 and 2; counting from the left, starting at 0, the weight of a digit at the i th position is 2^i . For example, the number 5 is represented by `1b 2b 0b`, since $1 \cdot 2^0 + 2 \cdot 2^1 + 0 \cdot 2^2 = 5$.

Compare this to the type of random-access lists (complete binary trees) in their *nested* (non-uniformly recursive) form ([Oka98], subsections 9.2.2 and 10.1.2):

```
data Random (A : Type) : Type where
  Zero : Random A
  One  : A → Random (A × A) → Random A
  Two  : A → A → Random (A × A) → Random A
```

Similarly, a random-access list can be formed by `Zero`, or by prepending `One x` for some x in A , or by `Two x y` with both x and y in A . Note that in the recursive fields of `One` and `Two`, we pass the type of pairs $A \times A$ as the parameter rather than simply A (hence the non-uniformity). In this recursive field, a `One` would thus ask for two values of A , and another level deeper for four, and so on.

By forgetting that a random-access list `xs` has fields, we find a binary number `size xs` again:

```
size : Random A → Bin
size Zero      = 0b
size (One _ xs) = 1b size xs
size (Two x y xs) = 2b size xs
```

For example, applying `size` to `One _ (Two _ _ Zero)` gives us back `1b 2b 0b`. Additionally, this number given by `size` coincides with the number of elements in `xs`: evidently, the `size` and number of elements of `Zero` are both zero. On the other hand, suppose that `xs` of type `Random (A × A)` has size n . Since $A \times A$ contains two values of A , we have doubled the weight of `xs`, so that it actually contains $2n$ values of A . Consequently, `One x xs` contains $2n + 1$ values, and `Two x y xs` contains $2n + 2$ values, so in general any `ys` contains `size ys` values.

In fact, if we remove the fields from random-access lists, binary numbers and random-access lists are essentially the same datatype. Conversely, we can de-

scribe random-access lists as binary numbers *decorated* with fields. Exactly such “informal human observations” can be made more precise and general using the language of *ornaments* as described by McBride [McB14]. This language effectively describes up to which modifications, such as adding or deleting fields, one datatype can be seen as a more elaborate version of another. Using ornaments, we can formulate random-access lists as a patch on top of binary numbers, and get `size` for free as the forgetful function.

Datastructures with relations to number systems occur more commonly, which raises the questions of how we can make this relation explicit in more general cases, but also which number systems have associated numerical representations, and which numerical representations arise from ornaments.

In this thesis we will explore how, for a certain generalization of positional number systems, we can construct all numerical representations as ornaments, and how some known examples of numerical representations fit into this framework, making the following contributions:

1. We define a *universe* in which we will encode number systems and numerical representations. This universe allows annotations, non-uniform datatypes, and composite datatypes. By encoding those datatypes in the universe, we gain the ability to write *generic programs* over them.
2. Then, we adapt ornaments to this universe, which lets us relate datatypes up to insertion of fields, nesting, and refinement of parameters, indices, and variables.
3. Finally, we prove the existence of two variants of numerical representations by implementing generic functions from number systems to ornaments, establishing that each number system has a numerical representation of the same structure.

As far as we are aware, a universe construction with this particular combination of features had not been studied before, hopefully allowing for further exploration of the interaction between features like non-uniform recursion and ornaments, and how incorporating generalized metadata can support more precise generic programming.

We formalize our work using the dependently typed proof assistant Agda [Tea23]. We use the unsafe `--type-in-type` option (manual page) so that the presented code is not diluted by the level variables, although our constructions can be modified to work without this flag [EC22]. We also use `--with-k`¹ and omit some type variables using variable generalization (manual page).

2 Background

Many of our constructions extend upon or are inspired by existing work in the domain of generic programming and ornaments, so let us take a closer look at the nuts and bolts to see what all the concepts are about.

¹In B Appendix B, we explain a variant on the universe which is compatible `--without-K`.

This section describes some common datatypes and their usages, exploring how dependent types let us prove properties of programs, or write programs that are correct-by-construction. We then discuss certain proofs or programs can be generalized to classes of types by encoding datatypes using descriptions. Finally, we take a look at ornaments as a means to relate datatypes by their structure, or construct more datatypes of a given structure, but also as a way to identify comparable programs on structurally similar datatypes.

2.1 Agda

We formalize our work in the programming language Agda [Tea23]. Agda is a total functional programming language with dependent types. Using dependent types we can use Agda as a proof assistant, reinterpreting types as formulas and functions as proofs, allowing us to state and prove theorems about our datastructures and programs. Since Agda is total, and hence all functions are total, all functions of a given type always terminate in a value of that type, ruling out invalid proofs as a bonus². While we will only occasionally reference Haskell, those more familiar with Haskell might understand (the reasonable part of) Agda as the subset of total Haskell programs [Coc+22].

In this section, we will explain and highlight some parts of Agda which we use in the later sections. Many of the types and functions we define in this section are also described and explained in most Agda tutorials ([Nor09], [WKS22], etc.), and can be imported from the standard library [The23].

2.2 Datatypes in Agda

At the level of (generalized) algebraic datatypes Agda is close to Haskell. In both languages, one can define objects using data declarations, and interact with them using function declarations. For example, we can define the type of booleans by declaring:

```
data Bool : Type where
  false : Bool
  true  : Bool
```

The constructors of this type state that values of `Bool` are produced in exactly two ways: `false` and `true`. We can then define functions on `Bool` by *pattern matching*, using that a variable of `Bool` is necessarily either `false` or `true`. As an example, we can define the conditional operator as:

```
if_then_else_ : Bool → A → A → A
if false then t else e = e
if true  then t else e = t
```

When we pattern match on a variable of type `A`, in this case `Bool`, the coverage checker ensures we define the function on all possible cases, and thus the function is completely defined.

We can also define a type representing the natural numbers:

²On the other hand, we sometimes have to put in some effort to convince Agda that a function indeed terminates.

```

data N : Type where
  zero : N
  suc  : N → N

```

Here, `N` always has a `zero` element, and for each element n the constructor `suc` expresses that there is also an element representing $n + 1$. Hence, `N` represents the natural numbers by encoding the existential axioms of the Peano axioms³. By pattern matching and *recursion* on `N`, we define the less-than operator:

```

_<?_ : (n m : N) → Bool
n    <? zero = false
zero <? suc m = true
suc n <? suc m = n <? m

```

One of the cases contains a recursive instance of `N`, so termination checker also verifies that this recursion indeed terminates, ensuring that we still define `n <? m` for all possible combinations of n and m . In this case the recursion is valid, since both arguments decrease before the recursive call, meaning that at some point n or m hits `zero` and the recursion terminates.

Like in Haskell, we can *parametrize* a datatype over other types to make a *polymorphic* type. By parameterizing a definition, the context of that definition is extended with a variable of the type parametrized over. Parametrizing lists over a type, we can define lists of values for all types:

```

data List (A : Type) : Type where
  [] : List A
  _:: : A → List A → List A

```

A list of A can either be empty `[]`, or contain an element of A and another list via `_::_`. In other words, `List` is a type of finite sequences in A , in the sense of sequences as an abstract type [Oka98].

Using polymorphic functions, we can manipulate and inspect lists by inserting or extracting elements. For example, we can define a function to look up the value at some position n in a list:

```

lookup? : List A → N → Maybe A
lookup? []      n      = nothing
lookup? (x :: xs) zero  = just x
lookup? (x :: xs) (suc n) = lookup? xs n

```

However, this function is *partial*, as we are relying on the type

```

data Maybe (A : Type) : Type where
  nothing : Maybe A
  just    : A → Maybe A

```

to handle the `[]` case, where the position does not lie in the list and we cannot return an element. If we know the length of the list `xs`, then we also know for which positions `lookup` will succeed, and for which it will not. We define

```

length : List A → N
length []      = zero
length (x :: xs) = suc (length xs)

```

³The equality, injectivity, and induction axioms follow from the corresponding principles for arbitrary datatypes.

so that we can test whether the position n lies inside the list by checking $n <? \text{length } xs$. If we declare `lookup` as a dependent function consuming a proof of $n <? \text{length } xs$, then `lookup` always succeeds. This, however, merely replaces the check whether `lookup` returns `nothing` with a check if $n <? \text{length } xs$ is before applying `lookup`.

More elegantly, we can combine natural numbers with an inequality by defining an *indexed type*, representing numbers below an upper bound:

```
data Fin : ℕ → Type where
  zero : Fin (suc n)
  suc   : Fin n → Fin (suc n)
```

Like parameters, *indices* add a variable to the context of a datatype, but unlike parameters, indices can influence the availability of constructors. The type `Fin` is defined such that a variable of type `Fin n` represents a number less than n . Since both constructors `zero` and `suc` dictate that the index is the `suc` of some natural number n , we see that `Fin zero` has no values. On the other hand, `suc` gives a value of `Fin (suc n)` for each value of `Fin n`, and `zero` gives exactly one additional value of `Fin (suc n)` for each n . We can thus conclude that `Fin n` has exactly n closed terms, each representing a number less than n .

To complement `Fin`, we define another indexed type representing lists of a known length, also known as vectors:

```
data Vec (A : Type) : ℕ → Type where
  [] : Vec A zero
  _:: : A → Vec A n → Vec A (suc n)
```

The `[]` constructor of this type produces the only term of type `Vec A zero`. The `_::` constructor ensures that a `Vec A (suc n)` always consists of an element of A and a `Vec A n`. Similar to `Fin`, we find that a `Vec A n` contains exactly n elements of A . Thus, we conclude that `Fin n` is exactly the type of positions in a `Vec A n`. In comparison to `List`, we can say that `Vec` is a type of arrays, in the sense of arrays as the abstract type of sequences of a fixed length. Furthermore, knowing the index of a term `xs` of type `Vec A n` uniquely determines the constructor it was formed by. Namely, if n is `zero`, then `xs` is `[]`, and if n is `suc m`, then `xs` is formed by `_::`.

Using this, we define a variant of `lookup` for `Fin` and `Vec`, taking a vector of length n and a position below n :

```
lookup : ∀ {n} → Vec A n → Fin n → A
lookup (x :: xs) zero = x
lookup (x :: xs) (suc i) = lookup xs i
```

We can now omit the `[]` case, where `lookup?` would return `nothing`. This happens because a variable of type `Fin n` is either `zero` or `suc i`, and both cases imply that n is `suc m` for some m . As we saw above, a `Vec A (suc m)` is always formed by `_::`, which ensures that finding `[]` for `xs` is impossible. Consequently, `lookup` always succeeds for vectors. However, this does not yet prove that `lookup` necessarily returns the right element, and we will need some more logic to verify this.

2.3 Proving in Agda

To describe the equality of terms we define a new type:

```
data _≡_ (a : A) : A → Type where
  refl : a ≡ a
```

If we have a value x of $a \equiv b$, then, as the only constructor of \equiv is `refl`, we must have that a is equal to b . We can use \equiv to describe the behaviour of functions like `lookup`.

To test `lookup`, we can insert elements into a vector with:

```
insert : ∀ {n} → Vec A n → Fin (suc n) → A → Vec A (suc n)
insert xs zero y = y :: xs
insert (x :: xs) (suc i) y = x :: insert xs i y
```

If `lookup` is correct, then we expect the following to hold

```
lookup-insert-type : ∀ {n} → Vec A n → Fin (suc n) → A → Type
lookup-insert-type xs i x = lookup (insert xs i x) i ≡ x
```

which essentially states that we find elements where we insert them.

To prove `lookup-insert-type`, we proceed as when defining any other function. By simultaneous induction on the position i and vector xs , we prove:

```
lookup-insert : ∀ {n} (xs : Vec A n) (i : Fin (suc n)) (y : A)
  → lookup-insert-type xs i y
lookup-insert [] zero y = refl
lookup-insert (x :: xs) zero y = refl
lookup-insert (x :: xs) (suc i) y = lookup-insert xs i y
```

In the first two cases, where we `lookup` the first position, `insert xs zero y` simplifies to $y :: xs$, so the `lookup` immediately returns y as wanted. In the last case, we have to prove that `lookup` is correct for $x :: xs$, so we use that the `lookup` ignores the term x , and appeal to the correctness of `lookup` on the smaller list xs to complete the proof.

Like \equiv , we can encode many other logical operations into datatypes, which establishes a correspondence between types and formulas, known as the *Curry-Howard correspondence*. For example, we can encode disjunctions (the logical ‘or’ operation) as

```
data _∪_ A B : Type where
  inj₁ : A → A ∪ B
  inj₂ : B → A ∪ B
```

Conjunction (logical ‘and’) can be represented by:⁴

```
record _×_ A B : Type where
  constructor _,_
  field
    fst : A
    snd : B
```

True and false are respectively represented by

⁴We use a record here, rather than a datatype with a constructor $A \rightarrow B \rightarrow A \times B$. The advantage of using a record is that this directly gives us projections like `fst` : $A \times B \rightarrow A$, and lets us use eta equality, making $(a, b) = (c, d) \iff a = c \wedge b = d$ holds automatically.

```
record  $\tau$  : Type where
  constructor tt
```

so that always $tt : \tau$, and:

```
data  $\perp$  : Type where
```

The body of \perp is intentionally empty: since \perp has no constructors, there is no proof of false⁵.

Because we identify function types with logical implications, we can also define the negation of a formula A as “ A implies false”:

```
 $\neg$  : Type  $\rightarrow$  Type
 $\neg A = A \rightarrow \perp$ 
```

The logical quantifiers \forall and \exists act on formulas with a free variable in a specific domain of discourse. We represent closed formulas by types, so we can represent a formula with one free variable of type A , also known as a *predicate*, by a function $A \rightarrow \text{Type}$, sending values of A to types. The universal quantifier $\forall a P(a)$ is true when for all a the formula $P(a)$ is true, so we represent the universal quantification of a predicate P as a dependent function type $(a : A) \rightarrow P\ a$, producing for each a of type A a proof of $P\ a$. The existential quantifier $\exists a P(a)$ is true when there is some a such that $P(a)$ is true, so we represent the existential quantification as

```
record  $\Sigma$  A (P : A  $\rightarrow$  Type) : Type where
  constructor _,_
  field
    fst : A
    snd : P fst
```

so that we have $\Sigma A P$ if and only if we have an element fst of A and a proof snd of $P\ a$. To avoid the need for lambda abstractions in existentials, we define the syntax

```
syntax  $\Sigma$ -syntax A ( $\lambda x \rightarrow P$ ) =  $\Sigma [x \in A] P$ 
```

letting us write $\Sigma [a \in A] P\ a$ for $\exists a P(a)$.

2.4 Descriptions

In the previous sections we completed a quadruple of types (\mathbb{N} , List , Vec , Fin) equipped with the nice interactions length and lookup . Similar to the type of $\text{length} : \text{List } A \rightarrow \mathbb{N}$, we can define

```
toList : Vec A n  $\rightarrow$  List A
toList [] = []
toList (x :: xs) = x :: toList xs
```

converting vectors back to lists. In the other direction, we can also promote a list to a vector by recomputing its index:

```
toVec : (xs : List A)  $\rightarrow$  Vec A (length xs)
toVec [] = []
toVec (x :: xs) = x :: toVec xs
```

⁵If we did not use `--type-in-type`, and even in that case I can only hope.

This is no coincidence, but happens because `N`, `List`, and `Vec` are structurally similar.

But how can we prove that datatypes have similar structures? In this section, we will explain a framework of datatype *descriptions* and ornaments, allowing us to describe datatypes as codes which can be compared directly, while also forming a foundation for generic programming in Agda [Nor09; AMM07; eff20; EC22].

Recall that while polymorphism allows us to write one program for many types at once, those programs act parametrically [Rey83; Wad89]: polymorphic functions must work for all types, thus they cannot inspect values of their type argument. Generic programs, by design, do use the structure of a datatype, allowing for more complex functions that do inspect values⁶.

Using datatype descriptions we can then relate `N`, `List` and `Vec`, explaining how `length` and `toList` are instances of *forgetful* functions. Let us walk through some ways of defining descriptions. We will start from simpler descriptions, building our way up to more general types, until we reach a framework in which we can describe `N`, `List`, `Vec` and `Fin`.

2.4.1 Finite types

An encoding of datatypes consists of two parts, a *type of descriptions* `U` of which the values are *codes* representing other datatypes, and an *interpretation* `U → Type` decoding those codes to the represented types. In the terminology of Martin-Löf type theory (MLTT) [Cha+10; Mar84], where types of types (e.g., `Type`) are called *universes*, we can think of a type of descriptions as an internal universe.

To start off, we define a basic universe with two codes `0` and `1`, respectively representing the types `⊥` and `τ`, and the requirement that the universe is closed under sums and products:

```
data U-fin : Type where
  0 1      : U-fin
  _+_ _*_ : U-fin → U-fin → U-fin
```

The meaning of the codes in this universe is then assigned by the interpretation:⁷

```
[_]fin : U-fin → Type
[ 0 ]fin = ⊥
[ 1 ]fin = τ
[ D + E ]fin = [ D ]fin ⊔ [ E ]fin
[ D * E ]fin = [ D ]fin × [ E ]fin
```

In this universe, we can encode the type of booleans simply as:

```
BoolD : U-fin
BoolD = 1 + 1
```

Since the types represented by `0` and `1` are finite, and sums and products of finite

⁶As examples, consider the generic JSON encoding of suitable datatypes [VL14], or the derivation of functor instances for a broad class of types [Mag+10].

⁷One might recognize that `[_]fin` is a morphism between the rings $(U\text{-fin}, +, *)$ and $(Type, \sqcup, \times)$. Similarly, `Fin` also gives a ring morphism from `N` with `+` and `*` to `Type`, and in fact `[_]fin` factors through `Fin` via the map sending the expressions in `U-fin` to their value in `N`.

types are also finite, we refer to `U-fin` as the universe of finite types. From this, one can immediately conclude that there is no code in `U-fin` representing the (infinite) type of natural numbers `N`.

2.4.2 Recursive types

We saw before that `N` differs from `Bool` by having a recursive field. So, in order to make a universe which can encode `N`, we begin by adding a code `ρ` to `U-fin` representing recursive type occurrences:

```
data U-rec : Type where
  1 ρ      : U-rec
  _⊕_ _⊗_ : U-rec → U-rec → U-rec
```

Then, we also have to redefine the interpretation: consider the interpretation of `1 ⊕ ρ`, for which we need to know that the whole type was `1 ⊕ ρ` while interpreting `ρ`. As a consequence, the interpretation splits in two phases.

In the first, we use functions from `Type` to `Type`⁸ to represent types with one free type variable. Interpreting a code `D`, we use the free variable `X` to represent “the type `D`”:

```
[_]rec : U-rec → Type → Type
[ 1 ]rec X = τ
[ ρ ]rec X = X
[ D ⊕ E ]rec X = ([ D ]rec X) ⊔ ([ E ]rec X)
[ D ⊗ E ]rec X = ([ D ]rec X) × ([ E ]rec X)
```

We can then model a recursive type by indeed setting the variable to the type itself, taking the *least fixpoint* of the functor:

```
data μ-rec (D : U-rec) : Type where
  con : [ D ]rec (μ-rec D) → μ-rec D
```

Recall the definition of `N`, which we can reinterpret as the declaration that `N` is the fixpoint of `N ≡ F N` for `F X = τ ⊔ X`. Hence, `N` can simply be encoded as:

```
NatD : U-rec
NatD = 1 ⊕ ρ
```

2.4.3 Sums of products

A downside of `U-rec` is that the definitions of types do not mirror their equivalents in user-written Agda very well. Using that polynomials can always be written as *sums of products*⁹, we can define a similar universe which more closely resembles handwritten code.

Unlike the arbitrarily shaped polynomials formed by `⊕` and `⊗`, a sum of products is analogous a datatype presented as a list of constructors. Thus, we split the descriptions into a stage in which we can form sums, equivalently datatypes

⁸We also refer to these functions as *polynomial functors*, which are polynomial here because they consist of sums and products, and are functors because they have a (functorial) mapping operation, as we will see later.

⁹We do not require these to be reduced, as different representations of the same polynomial represent different datatypes for us.

```

data U-sop : Type where
  [] : U-sop
  _::_ : Con-sop → U-sop → U-sop

```

on top of a stage where we can form products, equivalently constructors:

```

data Con-sop : Type where
  1 : Con-sop
  ρ : Con-sop → Con-sop
  σ : (S : Type) → (S → Con-sop) → Con-sop

```

When doing this, we also let the left-hand side of a product be any type, which allows us to represent ordinary fields. The interpretation of this universe, while similar to the one in the previous section, is also split into a part interpreting datatypes

```

[_]U-sop : U-sop → Type → Type
[ [] ]U-sop X = ⊥
[ C :: D ]U-sop X = [ C ]C-sop X × [ D ]U-sop X

```

and a part interpreting the constructors:

```

[_]C-sop : Con-sop → Type → Type
[ 1 ]C-sop X = τ
[ ρ C ]C-sop X = X × [ C ]C-sop X
[ σ S f ]C-sop X = Σ[ s ∈ S ] [ f s ]C-sop X

```

In this universe, we can define the type of lists as a description quantified over a type:

```

ListD : Type → U-sop
ListD A = nilD :: consD :: []
  where
    nilD = 1          -- : List A
    consD = σ A λ _ → --   A
      ρ          -- → List A
      1          -- → List A

```

Using this universe requires us to split functions on descriptions into multiple parts, but makes interconversion between representations and concrete types straightforward.

2.4.4 Parametrized types

The encoding of fields in `U-sop` makes the descriptions large in the following sense: by letting `S` in `σ` be an infinite type, we can get a description referencing infinitely many other descriptions. As a consequence, we cannot inspect an arbitrary description in its entirety. At the same time, we could not express `List` fully internally, and needed to handle the parameter externally.

We can resolve both quirks simultaneously by introducing parameters and variables using a new gadget. In a naive attempt, we can represent the parameters of a type as `List Type`. However, this cannot represent many useful types; in the existential quantifier `Σ_`, the type `A → Type` of second parameter `B` references back to the first parameter `A`, which is something that cannot be encoded in an ordinary list of types.

In a general parametrized type, parameters can refer to the values of all preceding parameters. The parameters of a type are thus a sequence of types depending on each other, which refer to as *telescopes* [Bru91] (also known as contexts in MLTT [Mar84]). We define telescopes using induction-recursion:

```
data Tel' : Type
[_]tel' : Tel' → Type

data Tel' where
  ∅ : Tel'
  _▷_ : (Γ : Tel') (S : [ Γ ]tel' → Type) → Tel'
```

A telescope can either be empty, or be formed from a telescope and a type in the context of that telescope, where we used the meaning of a telescope `[_]tel` to define types in the context of a telescope. This meaning represents valid assignments of values to parameters

```
[ ∅ ]tel' = τ
[ Γ ▷ S ]tel' = Σ [ Γ ]tel' S
```

interpreting a telescope into the dependent product of all the parameter types. This definition of telescopes enables us to write down the type of Σ :

```
Σ-Tel : Tel'
Σ-Tel = ∅ ▷ (λ _ → Type) ▷ (λ A → A → Type) ∘ snd
```

To encode Σ , we will need to be able to bind the argument `a` of `A` and reference it in the field `P a`. While viable, a universe built around `Tel'` would awkwardly confuse parameters and bound arguments.

By quantifying telescopes over a type [EC22; Sij16], we can distinguish parameters and bound arguments using almost the same setup:

```
data Tel (P : Type) : Type
[_]tel : Tel P → P → Type
```

A `Tel P` then represents a telescope for each value of `P`, which we can view as a telescope in the context of `P`. For readability, we redefine values in the context of a telescope as

```
_└_ : Tel P → Type → Type
Γ ⊢ A = Σ _ [ Γ ]tel → A
```

so we can define telescopes and their interpretations as:

```
data Tel P where
  ∅ : Tel P
  _▷_ : (Γ : Tel P) (S : Γ ⊢ Type) → Tel P

[ ∅ ]tel p = τ
[ Γ ▷ S ]tel p = Σ [ x ∈ [ Γ ]tel p ] S (p , x)
```

By setting `P = τ`, we recover the previous definition of parameter telescopes. We can then define an *extension* of a telescope as a telescope in the context of a parameter telescope

```
ExTel : Tel τ → Type
ExTel Γ = Tel ([ Γ ]tel tt)
```

representing a telescope of variables V over the fixed parameter telescope Γ , which can be extended independently of Γ . An extension of Γ can also be interpreted in the context of Γ :

```

[&_]tel : (Γ : Tel τ) (V : ExTel Γ) → Type
[Γ & V]tel = Σ ([Γ]tel tt) [V]tel

```

To describe conversions of telescopes, we give names to maps of telescopes and extensions:

```

Cxf : (Δ Γ : Tel P) → Type
Cxf Δ Γ = ∀ {p} → [Δ]tel p → [Γ]tel p

Vxf : Cxf Δ Γ → ExTel Δ → ExTel Γ → Type
Vxf g W V = ∀ {d} → [W]tel d → [V]tel (g d)

var→par : {g : Cxf Δ Γ} → Vxf g W V → [Δ & W]tel → [Γ & V]tel
var→par v (d , w) = _ , v w

Vxf-> : {g : Cxf Δ Γ} (v : Vxf g W V) (S : V ⊢ Type)
      → Vxf g (W ▷ (S ◦ var→par v)) (V ▷ S)
Vxf-> v S (p , w) = v p , w

```

We also defined two functions we will use extensively, `var→par` states that a map of extensions extends to a map between the whole telescopes, and `Vxf->` lets us extend a map of extensions by acting as the identity on a new variable.

In the descriptions, the parameter telescopes are relayed directly to the constructors, but the variable telescope is reset to `○` at the start of each constructor:

```

data U-par (Γ : Tel τ) : Type where
  [] : U-par Γ
  :: : Con-par Γ ○ → U-par Γ → U-par Γ

```

In the descriptions of constructors, we modify the `σ` code to request a type `S` in the context of `V`, which then also extends the context for the subsequent fields by `S`:

```

data Con-par (Γ : Tel τ) (V : ExTel Γ) : Type where
  1 : Con-par Γ V
  ρ : Con-par Γ V → Con-par Γ V
  σ : (S : V ⊢ Type) → Con-par Γ (V ▷ S) → Con-par Γ V

```

Replacing the function `S → U-sop` by `Con-par (V ▷ S)` allows us to bind the value of `S` while avoiding the higher order argument. The interpretation of the universe is then

```

[ ]U-par : U-par Γ → ([Γ]tel tt → Type) → [Γ]tel tt → Type
[ ]C-par : Con-par Γ V → ([Γ & V]tel → Type) → [Γ & V]tel → Type

[ [ ] ]U-par X p = 1
[ C :: D ]U-par X p = [ C ]C-par (X ◦ fst) (p , tt) × [ D ]U-par X p

[ 1 ]C-par X pv = τ
[ ρ C ]C-par X pv = X pv × [ C ]C-par X pv
[ σ S C ]C-par X pv@(p , v)
  = Σ [ s ∈ S pv ] [ C ]C-par (X ◦ var→par fst) (p , v , s)

```

where the `σ` case now provides the current parameters and variables to the field `S`, and extends the variables by `s` before passing them to the rest of the

interpretation. In this universe, we can describe the parameters of lists with a one-type telescope:

```
ListD : U-par (∅ ▷ λ _ → Type)
ListD = nilD
      :: consD
      :: []
where
  nilD = 1
  consD = σ (λ { ((_, A), _) → A })
          ( ρ
            1)
```

This description declares that `List` has two constructors, one with no fields, corresponding to `[]`, and the second with one field and a recursive field, representing `..::..`. In the second constructor, we use pattern lambdas to deconstruct the telescope¹⁰ and extract the type `A`.

Using the variable bound in `σ`, we can also give a description of the existential quantifier

```
SigmaD : U-par (∅ ▷ (λ _ → Type) ▷ λ { (_, _, A) → A → Type })
SigmaD = σ (λ { (((_, A), _), _) → A }) -- _,_ : (a : A)
          ( σ (λ { ((_, B), (_, a)) → B a }) --          → B a
            1) --          → Σ A B
          :: []
```

having one constructor with two fields. The first field of type `A` adds a value `a` to the variable telescope, which we pass to `B` in the second field by pattern matching on the variable telescope.

2.4.5 Indexed types

Lastly, we can integrate indexed types [DS06] into the universe by abstracting over indices:

```
data U-ix (Γ : Tel τ) (I : Type) : Type where
  [] : U-ix Γ I
  ..:: : Con-ix Γ ∅ I → U-ix Γ I → U-ix Γ I
```

Recall that in native Agda datatypes, a choice of constructor can fix the indices of the recursive fields and the resultant type, so we encode:

```
data Con-ix (Γ : Tel τ) (V : ExTel Γ) (I : Type) : Type where
  1 : V ⊢ I → Con-ix Γ V I
  ρ : V ⊢ I → Con-ix Γ V I → Con-ix Γ V I
  σ : (S : V ⊢ Type) → Con-ix Γ (V ▷ S) I → Con-ix Γ V I
```

Both `1` and `ρ` now take a value of `I` in context `V`, where for `1` this value represents the *actual index*, and for `ρ` it represents the *expected index* of the recursive field. Consider as an example `Fin`, where “`suc n`” bit of the constructor signatures tells us what the index of a constructor actually is, while the recursive type `Fin n` tells us which index the recursive field needs to have.

¹⁰Due to the interpretation of telescopes, the `∅` part always contributes a value `tt` we explicitly ignore, which also explicitly needs to be provided when passing parameters and variables.

If we are constructing a term of some indexed type, then the previous choices of constructors and arguments build up the actual index of this term. This actual index must then match the expected index of the declaration of this term. Hence, in the case of a leaf, we replace the unit type with the necessary equality between the expected i and actual index j [McB14], while for a recursive field, the expected index j is computed from the parameters and variables:

```
[_]C-ix : Con-ix  $\Gamma$  V I  $\rightarrow$  ([  $\Gamma$  ]tel tt  $\rightarrow$  I  $\rightarrow$  Type)  $\rightarrow$  ([  $\Gamma$  & V ]tel  $\rightarrow$  I  $\rightarrow$  Type)
[ 1 j ]C-ix X pv i = i  $\equiv$  (j pv)
[ p j C ]C-ix X pv@(p , v) i = X p (j pv)  $\times$  [ C ]C-ix X pv i
[  $\sigma$  S C ]C-ix X pv@(p , v) i =  $\Sigma$ [ s  $\in$  S pv ] [ C ]C-ix X (p , v , s) i

[_]D-ix : U-ix  $\Gamma$  I  $\rightarrow$  ([  $\Gamma$  ]tel tt  $\rightarrow$  I  $\rightarrow$  Type)  $\rightarrow$  ([  $\Gamma$  ]tel tt  $\rightarrow$  I  $\rightarrow$  Type)
[ [] ]D-ix X p i = 1
[ C :: Cs ]D-ix X p i = [ C ]C-ix X (p , tt) i  $\uplus$  [ Cs ]D-ix X p i
```

With indexed types, we can describe finite types and vectors¹¹ as

```
FinD : U-ix  $\emptyset$   $\mathbb{N}$ 
FinD = zeroD :: sucD :: []
where
zeroD =  $\sigma$  ( $\lambda$  _  $\rightarrow$   $\mathbb{N}$ ) -- : (n :  $\mathbb{N}$ )
      ( 1 ( $\lambda$  { (- , (- , n))  $\rightarrow$  suc n } ) ) --  $\rightarrow$  Fin (suc n)
sucD =  $\sigma$  ( $\lambda$  _  $\rightarrow$   $\mathbb{N}$ ) -- : (n :  $\mathbb{N}$ )
      ( p ( $\lambda$  { (- , (- , n))  $\rightarrow$  n } ) ) --  $\rightarrow$  Fin n
      ( 1 ( $\lambda$  { (- , (- , n))  $\rightarrow$  suc n } ) ) ) --  $\rightarrow$  Fin (suc n)
```

and:

```
VecD : U-ix ( $\emptyset \triangleright \lambda$  _  $\rightarrow$  Type)  $\mathbb{N}$ 
VecD = nilD
      :: consD
      :: []
where
nilD = 1 ( $\lambda$  _  $\rightarrow$  zero) -- : Vec A zero
consD =  $\sigma$  ( $\lambda$  _  $\rightarrow$   $\mathbb{N}$ ) -- : (n :  $\mathbb{N}$ )
      (  $\sigma$  ( $\lambda$  { ((- , A) , -)  $\rightarrow$  A } ) ) --  $\rightarrow$  A
      ( p ( $\lambda$  { (- , ((- , n) , -))  $\rightarrow$  n } ) ) --  $\rightarrow$  Vec A n
      ( 1 ( $\lambda$  { (- , ((- , n) , -))  $\rightarrow$  suc n } ) ) ) ) --  $\rightarrow$  Vec A (suc n)
```

In the first constructor of `VecD` we report an actual index of `zero`. In the second, we have a field `\mathbb{N}` to bring the index n into scope, which is used to request a recursive field with index n , and report the actual index of `suc n`. For completeness, let us replicate the natural numbers and lists in `U-ix`:

```
! : A  $\rightarrow$   $\tau$ 
! x = tt

NatD : U-ix  $\emptyset$   $\tau$ 
NatD = zeroD :: sucD :: []
where
```

¹¹Unlike some more elaborate encodings, we do not model implicit fields, so the descriptions of `Fin` and `Vec` look slightly different.

```

zeroD = 1 !
sucD  = ρ !
      ( 1 ! )

ListD : U-ix (∅ ▷ λ _ → Type) τ
ListD = nilD :: consD :: []
  where
    nilD = 1 !
    consD = σ (λ { ((_, A), _) → A })
           ( ρ !
             ( 1 ! ) )

```

Writing the descriptions `NatD`, `ListD` and `VecD` next to each other makes it easy to see the similarities: `ListD` is the same as `NatD` with a type parameter and an extra field `σ` of `A`. Likewise, `VecD` is the same as `ListD`, but now indexing over `N` and with another extra field `σ` of `N`. In Section 2.5, we will explain how this kind of analysis of descriptions can be performed formally inside Agda.

2.4.6 Generic Programming

As a bonus, we can also use `U-ix` for generic programming. For example, we can define the generic `fold` operation¹²:

```

→₃_ : (X Y : A → B → Type) → Type
X →₃ Y = ∀ a b → X a b → Y a b

fold : ∀ {D : U-ix ⊢ I} {X}
      → [ D ] D-ix X →₃ X → μ-ix D →₃ X

```

From the point of view of category theory, we actually get `fold` for free: since `μ-ix D` is the least fixpoint, or initial algebra, of `[D] D`, `fold f` is simply the induced map to the algebra `f : [D] D X →₃ X`.

More concretely, we can view `[D] D X` as a variant of `μ-ix D`, in which the recursive positions hold values of `X` rather than other values of `μ-ix D`. For example, the type `[ListD] D X` reduces (up to equivalence) to `τ ⊔ (A × X A)`, substituting `X A` for what would usually be the recursive field.

In a sense, a term of `[D] D X` is a kind of `D`-structured set of values of `X`. From this perspective, `fold` roughly states that an operation `f`, collapsing `D`-structured sets of `X` into `X`, extends to a function `fold f`. This function sends a whole value of `μ-ix D` into `X`, recursively collapsing applications of `con` from the bottom up.

As an example, we can instantiate `fold` to `ListD`, which, by the aforementioned equivalence, corresponds to:

```

foldr : {X : Type → Type}
       → (∀ A → τ ⊔ (A × X A) → X A)
       → ∀ B → List B → X B

```

Much like the familiar `foldr` operation lets us consume a `List A` to produce a value `X A`; provided we give a value `X A` for the `[]` case, and a means to convert a pair `A × X A` to `X A` for the `::` case.

¹²The full construction can be found in A Appendix A.

Do note that this version of `fold` takes a polymorphic function as an argument, as opposed to the usual `fold` which has the quantifiers on the outside:

```
foldr' : ∀ A B → (τ ∪ (A × B) → B) → List A → B
```

Like a couple of constructions we will encounter in later sections, we can recover the usual `fold` into a type `C` by generalizing `C` to the appropriate kind of maps into `C`. For example, by letting `X` be continuation-passing computations into `N`, we can recover:

```
sum' : ∀ A → List A → (A → N) → N
sum' = foldr {X = λ A → (A → N) → N} go
  where
    go : ∀ A → τ ∪ (A × ((A → N) → N)) → (A → N) → N
    go A (inj1 tt)      f = zero
    go A (inj2 (x , xs)) f = f x + xs f

sum : List N → N
sum xs = sum' N xs id
```

2.5 Ornaments

In this section we will introduce a simplified definition of ornaments, and use it to compare descriptions. Simply looking at their descriptions, `N` and `List` are rather similar, except that `List` has some things `N` does not have. We could say that we can form the type of lists by starting from `N` and adding a parameter and field, while keeping everything else the same. In the other direction, we see that each list corresponds to a natural by stripping this information. Likewise, the type of vectors is almost identical to `List`, can be formed from it by adding indices, and each vector corresponds to a list by dropping the indices.

Observations like these can be generalized using ornaments [McB14; KG16; Sij16], which define a binary relation describing which datatypes can be formed by “decorating” others. Conceptually, a type can be decorated by adding or modifying fields, extending its parameters, or refining its indices.

Essential to the concept of ornaments is the ability to convert back, forgetting the extra structure. After all, if there is an ornament from a description `D` to `E`, then `E` should be `D` with added fields, and more specific parameters and indices. This means that we should also be able to discard those extra fields, and revert to the less specific parameters and indices, obtaining a conversion from `E` to `D`. If `D` is a `U-ix` Γ `I` and `E` is a `U-ix` Δ `J`, then for a conversion from `E` to `D` to exist, there must also be functions `re-par` : `Cxf` Δ Γ and `re-index` : `J` \rightarrow `I` for re-parametrization and re-indexing.

In the same way that descriptions in `U-ix` are lists of constructor descriptions, ornaments are lists of constructor ornaments. We define the type of ornaments re-parametrizing with `re-par` and re-indexing with `re-index` as a type indexed over `U-ix`:

```
data Orn (re-par : Cxf Δ Γ) (re-index : J → I) :
  U-ix Γ I → U-ix Δ J → Type where
  [] : Orn re-par re-index [] []
```

```

 $\_::$  : ConOrn re-par id re-index CD CE
      → Orn re-par re-index D E
      → Orn re-par re-index (CD :: D) (CE :: E)

```

An ornament then induces a conversion between types via the forgetful map

```

bimap : {A B C D E : Type}
        → (A → B → C) → (D → A) → (E → B)
        → D → E → C
bimap f g h d e = f (g d) (h e)
ornForget : ∀ {re-par re-index} → Orn re-par re-index D E
            →  $\mu$ -ix E →3 bimap ( $\mu$ -ix D) re-par re-index

```

which reverts the modifications made by the constructor ornaments, and restores the original indices and parameters.

The definition of the constructor ornaments `ConOrn` controls the kinds of modification ornaments allow. Each allowed modification, equivalently each constructor of `ConOrn` also has to be reverted by `ornForget`. Accordingly, some modifications will have preconditions, which are in this case always pointwise equalities:

```

 $\_ \sim \_$  : {B : A → Type} → (f g : ∀ a → B a) → Type
f  $\sim$  g = ∀ a → f a ≡ g a

```

Since constructors exist in the context of variables, we let constructor ornaments transform variables with `re-var`, in addition to parameters and indices.

The first three constructors of `ConOrn` represent the operations which copy the corresponding constructors of `Con-ix`¹³. More precisely, as an example, the ornament `1` states that if actual indices `i` and `j`, of potentially different types, are related, then the datatype constructors of the same names `1 i` and `1 j` are related.

By contrast, the $\Delta\sigma$ ornament allows adding fields which are not present on the original datatype.:

```

data ConOrn (re-par : Cxf  $\Delta$   $\Gamma$ ) (re-var : Vxf re-par W V)
  (re-index : J → I) :
  Con-ix  $\Gamma$  V I → Con-ix  $\Delta$  W J → Type where
1 : ∀ {i j}
  → re-index o j  $\sim$  i o var→par re-var
  → ConOrn re-par re-var re-index (1 i) (1 j)

 $\rho$  : ∀ {i j CD CE}
  → re-index o j  $\sim$  i o var→par re-var
  → ConOrn re-par re-var re-index CD CE
  → ConOrn re-par re-var re-index ( $\rho$  i CD) ( $\rho$  j CE)

 $\sigma$  : ∀ {S CD CE}
  → ConOrn re-par (Vxf→ re-var S) re-index CD CE
  → ConOrn re-par re-var re-index ( $\sigma$  S CD) ( $\sigma$  (S o var→par re-var) CE)

```

¹³Viewing `ConOrn` as a binary relation on `Con-ix`, these represent the preservation of `ConOrn` by `1`, `ρ` , and `σ` , up to parameters, variables, and indices.

```

 $\Delta\sigma : \forall \{S \text{ CD CE}\}$ 
   $\rightarrow \text{ConOrn re-par (re-var } \circ \text{fst) re-index CD CE}$ 
   $\rightarrow \text{ConOrn re-par re-var re-index CD } (\sigma \text{ S CE})$ 

```

The commuting square $\text{re-index } \circ j \sim i \circ \text{var} \rightarrow \text{par re-var}$ in the first two constructors ensures that the indices on both sides are indeed related, up to re-index and re-var . As expected, we see that there can only be an ornament from a description D to E if there are constructor ornaments for all constructors; likewise, there can only be an ornament between constructors CD and CE if CE consists wholly of added fields and fields copied from CD , potentially refining parameters, variables, and indices.

Using these ornaments, we can make the claim that “lists are indeed natural numbers decorated with fields” more precise:

```

NatD-ListD : Orn ! id NatD ListD
NatD-ListD = nilO :: consO :: []
where
  nilO  = 1 (λ _ → refl) -- : List A
  consO =  $\Delta\sigma \{S = \lambda \{ (-, A), (-) \rightarrow A \} \}$  -- : A
        ( ρ (λ _ → refl) -- → List A
          ( 1 (λ _ → refl))) -- → List A

```

This ornament preserves most of the structure of \mathbb{N} , and only adds a field A using $\Delta\sigma$ ¹⁴. As \mathbb{N} has no parameters or indices, List has more specific parameters, namely a single type parameter. Consequently, all commuting squares factor through the unit type and can be satisfied with $\lambda _ \rightarrow \text{refl}$.

We can also ornament lists to get vectors by re-indexing them over \mathbb{N} :

```

ListD-VecD : Orn id ! ListD VecD
ListD-VecD = nilO :: consO :: []
where
  nilO  = 1 (λ _ → refl) -- : Vec A zero
  consO =  $\Delta\sigma \{S = \lambda _ \rightarrow \mathbb{N}\}$  -- : (n :  $\mathbb{N}$ )
        ( σ -- → A
          ( ρ {j = λ { (-, (-, n), -) → n }} -- → Vec A n
            (λ _ → refl)
            ( 1 {j = λ { (-, (-, n), -) → suc n }} -- → Vec A (suc n)
              (λ _ → refl))))

```

We bind a new field of \mathbb{N} with $\Delta\sigma$, extracting it in 1 and ρ to declare that the constructor corresponding to $_{::}$ takes a vector of length n and returns a vector of length $\text{suc } n$.

The conversions from lists to natural numbers (length), and from vectors to lists (toList) arise as ornForget , which we define using the fold over an algebra that erases a single layer of decorations:

```

ornForget 0 = fold (ornAlg 0)

```

Recursively applying this algebra, which reinterprets layers of E -data as D -data, lets us take apart a value in the fixpoint $\mu\text{-ix } E$ and rebuild it to a value of $\mu\text{-ix}$

¹⁴Note that S , and some later arguments we provide to ornaments, are implicit arguments: Agda would happily infer them from ListD and then later from VecD , had we omitted them.

D. This algebra

```

ornAlg : ∀ {D : U-ix Γ I} {E : U-ix Δ J} {re-par re-index}
  → Orn re-par re-index D E
  → [ E ]D-ix (bimap (μ-ix D) re-par re-index)
  →3 bimap (μ-ix D) re-par re-index
ornAlg O p j x = con (ornErase O p j x)

```

is a special case of the erasing function, which “undecorates” a single interpretation over an arbitrary type X:

```

ornErase : ∀ {re-par re-index} {X}
  → Orn re-par re-index D E
  → [ E ]D-ix (bimap X re-par re-index)
  →3 bimap ([ D ]D-ix X) re-par re-index
ornErase (CD :: D) p j (inj1 x) = inj1 (conOrnErase CD (p , tt) j x)
ornErase (CD :: D) p j (inj2 x) = inj2 (ornErase D p j x)

conOrnErase : ∀ {re-par re-index} {W V} {X} {re-var : Vxf re-par W V}
  {CD : Con-ix Γ V I} {CE : Con-ix Δ W J}
  → ConOrn re-par re-var re-index CD CE
  → [ CE ]C-ix (bimap X re-par re-index)
  →3 bimap ([ CD ]C-ix X) (var→par re-var) re-index
conOrnErase {re-index = i} (1 sq) p j x = trans (cong i x) (sq p)
conOrnErase {X = X} (ρ sq CD) p j (x , y) = subst (X _) (sq p) x
  , conOrnErase CD p j y
conOrnErase (σ CD) (p , w) j (s , x) = s
  , conOrnErase CD (p , w , s) j x
conOrnErase (Δσ CD) (p , w) j (s , x) = conOrnErase CD (p , w , s) j x

```

By pattern matching on the ornament, `conOrnErase` sees which bits of CE are new, and which are copied from CD. This tells us which parts of a term x under an interpretation of CE need to be forgotten, and which parts need to be copied or translated. Specifically, the first three cases of `conOrnErase` correspond to the structure-preserving ornaments, and merely translate equivalent structures from CE to CD.

For example, the `1 sq` case tells us that CD is `1 i'` and CE is `1 j'`. Recalling that a leaf `1 j'` at parameter p and expected index j is interpreted as the equality $j \equiv (j' \ p)$, we only need to produce the corresponding equality for `1 i'`, which is re-index $j \equiv i' \ (\text{var} \rightarrow \text{par} \text{ re-var } p)$. This is precisely accomplished by applying `re-index` to both sides and composing with the square `sq` at p . Likewise, in the case of `ρ` we have to show that x can be converted from one `ρ` to the other `ρ` by translating its parameters, but in `σ` case, we can directly copy the field. Only the ornament `Δσ` adds a field, which is easily undone by removing that field.

In this way `ornForget` reinforces the idea that E is a decorated version of D when there is an ornament from D to E by associating to each value of E an underlying value in D . This additionally makes it easier to relate functions between related types. For example, instantiating `ornForget` for `NatD-ListD` yields `length`. Hence, the statement that `length` sends concatenation `++` to addition `+`, that is `length (xs ++ ys) ≡ length xs + length ys`, is equivalent

to the statement that $_++_$ and $_+_$ are related, or that $_++_$ is a lifting of $_+_$ [DM14].

2.6 Ornamental Descriptions

By defining the ornaments `NatD-ListD` and `ListD-VecD` we demonstrated that lists are numbers with fields, and vectors are lists with fixed lengths. Even though we had to give `ListD` before we could define `NatD-ListD`, the value of `NatD-ListD` actually forces the right-hand side to be `ListD`.

If we somehow could leave out the right-hand side of ornaments, then we can also use ornaments to represent descriptions as patches on top of other descriptions. So, *ornamental descriptions* are precisely defined as ornaments without the right-hand side, effectively bundling a description and an ornament to it¹⁵. Their definition is analogous to that of ornaments, making the arguments which would only appear in the new description explicit:

```
data OrnDesc (Δ : Tel τ) (J : Type)
  (re-par : Cxf Δ Γ) (re-index : J → I)
  : U-ix Γ I → Type where
[] : OrnDesc Δ J re-par re-index []
_::_ : ConOrnDesc Δ ∅ J re-par ! re-index CD
      → OrnDesc Δ J re-par re-index D
      → OrnDesc Δ J re-par re-index (CD :: D)
data ConOrnDesc (Δ : Tel τ) (W : ExTel Δ) (J : Type)
  (re-par : Cxf Δ Γ) (re-var : Vxf re-par W V)
  (re-index : J → I)
  : Con-ix Γ V I → Type where
1 : V {i} (j : W ⊢ J)
  → re-index ∘ j ~ i ∘ var→par re-var
  → ConOrnDesc Δ W J re-par re-var re-index (1 i)

ρ : V {i} {CD} (j : W ⊢ J)
  → re-index ∘ j ~ i ∘ var→par re-var
  → ConOrnDesc Δ W J re-par re-var re-index CD
  → ConOrnDesc Δ W J re-par re-var re-index (ρ i CD)

σ : V (S : V ⊢ Type) {CD}
  → ConOrnDesc Δ (W ▷ S ∘ var→par re-var)
      J re-par (Vxf-▷ re-var S) re-index CD
  → ConOrnDesc Δ W J re-par re-var re-index (σ S CD)

Δσ : V (S : W ⊢ Type) {CD}
  → ConOrnDesc Δ (W ▷ S)
      J re-par (re-var ∘ fst) re-index CD
  → ConOrnDesc Δ W J re-par re-var re-index CD
```

¹⁵Consequently, `OrnDesc Δ J g i D` must simply be a convenient representation of $\Sigma (U\text{-}ix \Delta J) (Orn \ g \ i \ D)$.

Using `OrnDesc`, we can describe `ListD` as a patch on `NatD`, inserting a σ in the constructor corresponding to `suc`:

```
ListOD : OrnDesc (0 ▷ λ _ → Type) τ !! NatD
ListOD = nilOD :: consOD :: []
  where
    nilOD  = 1 (λ _ → tt) (λ _ → refl)  -- : List A
    consOD = Δσ (λ { ((_, A), _) → A }) -- : A
            ( ρ (λ _ → tt) (λ _ → refl)  -- → List A
              ( 1 (λ _ → tt) (λ _ → refl) ) -- → List A
```

Since an ornamental description simply represents a patch on top of a description, we can also extract the patched description and the ornament to it. To extract the description, we can use the projection applying the patch in an ornamental description

```
toDesc : {D : U-ix Γ I} → OrnDesc Δ J re-par re-index D
        → U-ix Δ J
toDesc [] = []
toDesc (COD :: OD) = toCon COD :: toDesc OD

toCon : ∀ {CD : Con-ix Γ V I} {re-par} {W} {re-var : Vxf re-par W V}
        → ConOrnDesc Δ W J re-par re-var re-index CD
        → Con-ix Δ W J
toCon (1 j j~i)          = 1 j
toCon (ρ j j~i COD)      = ρ j (toCon COD)
toCon {re-var = v} (σ S COD) = σ (S ◦ var→par v) (toCon COD)
toCon (Δσ S COD)        = σ S (toCon COD)
```

which would extract `ListD` out of `ListOD`.

The other projection reconstructs the ornament to the extracted description

```
toOrn : {D : U-ix Γ I}
        (OD : OrnDesc Δ J re-par re-index D)
        → Orn re-par re-index D (toDesc OD)
toOrn [] = []
toOrn (COD :: OD) = toConOrn COD :: toOrn OD

toConOrn : ∀ {CD : Con-ix Γ V I} {re-par} {W} {re-var : Vxf re-par W V}
            → (COD : ConOrnDesc Δ W J re-par re-var re-index CD)
            → ConOrn re-par re-var re-index CD (toCon COD)
toConOrn (1 j j~i)          = 1 j~i
toConOrn (ρ j j~i COD)      = ρ j~i (toConOrn COD)
toConOrn (σ S COD)          = σ (toConOrn COD)
toConOrn (Δσ S COD)        = Δσ (toConOrn COD)
```

and would extract `NatD-ListD` from `ListOD`. As a consequence, `OrnDesc` enjoys the features of both `Desc` and `Orn`, such as interpretation into a datatype by μ and the conversion to the underlying type by `ornForget`, by factoring through these projections.

In later sections, we will routinely use `OrnDesc` to view triples like `(NatD, ListD, VecD)` as a base type equipped with two patches in sequence.

3 Descriptions

Before we can analyze number systems and their numerical representations using generic programs, we first have to ensure that these types fit into the descriptions. Some numerical representations are hard to describe using only the descriptions of parametric indexed inductive types `U-ix`; in order to keep things running smoothly for the generic programmer, we present an extension of `U-ix` incorporating metadata, parameter transformation, description composition, and variable transformation.

3.1 Numerical Representations

Before we start rebuilding our universe, let us look at the construction of the simplest numerical representation `Vec` from `N`. At first, we defined `Vec` as the length-indexed variant of `List`, so that `lookup` becomes total and satisfies nice properties like `lookup-insert`. Later, we gave another description of `Vec` as an ornament on top of `List`. More abstractly, `Vec` is an implementation of finite maps with domain `Fin`, where finite maps are simply those types with operations like `insert`, `remove`, `lookup`, and `tabulate`¹⁶, satisfying relations or laws like `lookup-insert` and `lookup ∘ tabulate ≡ id`.

For comparison, we can define a trivial implementation of finite maps, by reading `lookup` as a prescript:

```
Lookup : Type → N → Type
Lookup A n = Fin n → A
```

Since `lookup` is simply the identity function on `Lookup`, this immediately satisfies the laws of finite maps, provided we define `insert` and `remove` correctly.

Unsurprisingly, `Vec` is *representable*, that is, we have that `Lookup` and `Vec` are equivalent, in the sense that there is an *isomorphism* between `Lookup` and `Vec`:¹⁷

```
record _≅_ A B : Type where
  constructor iso
  field
    fun : A → B
    inv : B → A
    rightInv : ∀ b → fun (inv b) ≡ b
    leftInv  : ∀ a → inv (fun a) ≡ a
```

An `Iso` from `A` to `B` is a map from `A` to `B` with a (two-sided) inverse¹⁸. In terms of elements, this means that elements of `A` and `B` are in one-to-one correspondence.

Rather than deriving them ourselves, we can also establish properties like `lookup-insert` from this equivalence. Instead of finding the properties of `Vec`

¹⁶The function `tabulate : (Fin n → A) → Vec A n` collects an assignment of elements `f` into a vector `tabulate f`.

¹⁷Since `lookup` is an isomorphism with `tabulate` as inverse, as we see from the relations `lookup ∘ tabulate ≡ id` and `tabulate ∘ lookup ≡ id`. Without further assumptions, we cannot use the equality type `≡` for this notion of equivalence of types: a type with a different name but exactly the same constructors as `Vec` would not be equal to `Vec`.

¹⁸Compare this to the other notion of equivalence: there is a map $f : A \rightarrow B$, and for each b in B there is exactly one a in A for which $f(a) = b$.

that were already there, let us view `Vec` as a consequence of the definition of `N` and `lookup`. By turning the `Iso` on its head, and starting from the equation that `Vec` is equivalent to `Lookup`, we derive a definition of `Vec` as if we were solving an equation [HS22].

As a warm-up, we can also derive `Fin` from the fact that `Fin n` should contain n elements, and thus be isomorphic to $\Sigma[m \in \mathbb{N}] m < n$. To express such a definition by isomorphism, we define

```
Def : Type → Type
Def A =  $\Sigma'$  Type  $\lambda B \rightarrow A \simeq B$ 

defined-by   : {A : Type} → Def A      → Type
by-definition : {A : Type} → (d : Def A) → A  $\simeq$  (defined-by d)
```

using:

```
record  $\Sigma'$  (A : Type) (B : A → Type) : Type where
  constructor _use-as-def
  field
    {fst} : A
    snd : B fst
```

The type `Def A` is deceptively simple, after all, there is (up to isomorphism) only one unique term in it! However, when using `Definitions`, the implicit `Σ'` extracts the right-hand side of a proof of an isomorphism, allowing us to reinterpret a proof as a definition.

To keep the isomorphisms we are trying to construct readable, we construct them as chains of simpler isomorphisms, using a variant of *equational reasoning* [The23; WKS22], letting us compose isomorphisms while displaying the intermediate steps. In the calculation of `Fin`, we will use the following lemmas:

```
 $\bot$ -strict : (A →  $\bot$ ) → A  $\simeq \bot$ 
 $\leftarrow$ -split :  $\forall n \rightarrow (\Sigma[ m \in \mathbb{N} ] m < \text{succ } n) \simeq (\tau \cup (\Sigma[ m \in \mathbb{N} ] m < n))$ 
```

If we allow reading isomorphisms as “*is*”, then the terminology of Section 2.3, `\bot -strict` states that “if A is false, then A *is* empty”, while `\leftarrow -split` states that the set of numbers below $n + 1$ has one more element than the set of numbers below n . Using these, we can calculate:¹⁹

```
Fin-def :  $\forall n \rightarrow \text{Def } (\Sigma[ m \in \mathbb{N} ] m < n)$ 
Fin-def zero =  $\Sigma[ m \in \mathbb{N} ] (m < \text{zero})$ 
               $\simeq$   $\langle \bot$ -strict ( $\lambda ()$ )  $\rangle$ 
               $\bot \simeq$   $\blacksquare$  use-as-def
Fin-def (succ n) =  $\Sigma[ m \in \mathbb{N} ] (m < \text{succ } n)$ 
                   $\simeq$   $\langle \leftarrow$ -split n  $\rangle$ 
                  ( $\tau \cup (\Sigma[ m \in \mathbb{N} ] m < n)$ )
                   $\simeq$   $\langle \text{cong } (\tau \cup \_) (\text{by-definition } (\text{Fin-def } n)) \rangle$ 
                  ( $\tau \cup \text{defined-by } (\text{Fin-def } n)$ )  $\simeq$   $\blacksquare$  use-as-def
```

This gives a different (but equivalent) definition of `Fin` compared to `FinD`; the description `FinD` describes `Fin` as an inductive family, whereas `Fin-def` describes `Fin` equivalently as a type-computing function [KG16]. From this `Def` we can

¹⁹Making non-essential use of `cong` for type families. In the derivation of `Vec` we use function extensionality, which has to be postulated, or can be obtained by using the cubical path types.

extract a definition of `Fin`:

```
Fin : ℕ → Type
Fin n = defined-by (Fin-def n)
```

To derive `Vec`, we use the isomorphisms

```
1→A≃τ : (1 → A) ≃ τ
τ→A≃A : (τ → A) ≃ A
ℳ→ℳ→× : ((A ℳ B) → C) ≃ ((A → C) × (B → C))
```

which one can compare to the familiar exponential laws. With these laws, we calculate the type of vectors

```
Vec-def : ∀ A n → Def (Lookup A n)
Vec-def A zero =
  (Fin zero → A) ≃⟨ ⟩
  (1 → A) ≃⟨ 1→A≃τ ⟩
  τ ≃-■ use-as-def

Vec-def A (suc n) =
  (Fin (suc n) → A) ≃⟨ ⟩
  (τ ℳ Fin n → A) ≃⟨ ℳ→ℳ→× ⟩
  (τ → A) × (Fin n → A) ≃⟨ cong (λ x (Fin n → A)) τ→A≃A ⟩
  A × (Fin n → A) ≃⟨ cong (A ×_) (by-definition (Vec-def A n)) ⟩
  A × (defined-by (Vec-def A n))
  ≃-■ use-as-def
```

yielding a definition of vectors and the `Iso` to `Lookup` in one go:

```
Vec : Type → ℕ → Type
Vec A n = defined-by (Vec-def A n)

Vec-Lookup : ∀ A n → Lookup A n ≃ Vec A n
Vec-Lookup A n = by-definition (Vec-def A n)
```

In conclusion, we computed a type of finite maps (the numerical representation `Vec`) from a number system (`ℕ`), by cases on the number system and making use of the values represented by the number system.

3.2 Room for Improvement

We could now carry on and attempt to generalize this calculation to other number systems, but we would quickly run into dead ends for certain numerical representations. Let us give an overview of what bits of `U-ix` are still missing if we are going to generically construct all numerical representations we promised.

3.2.1 Number systems

In the calculation `Vec` from `ℕ`, we analyze and replicate the structure of `ℕ`, for which we use the meaning of these constructors as numbers assigned to them by our explanation of `ℕ` in words²⁰. Based on that interpretation of constructors

²⁰More accurately, the meaning of `ℕ` comes from `Fin`, which gets its meaning from our definition of `1<-`.

as numbers, we deliberately choose to add zero fields in the case corresponding to `zero` and choose to add one field in the case of `one`.

However, if we want to compute numerical representations generically, we also have to convince Agda that our datatypes indeed represent number systems. As a first step, let us fix `N` as the primordial number system, so that we can compare other number systems by how they are mapped into `N`. Trivially, `N` can be interpreted as a number system via `id : N → N`.

The binary numbers, as described in the introduction, can be mapped to `N` by:

```
toN-Bin : Bin → N
toN-Bin 0b = 0
toN-Bin (1b n) = 1 + 2 * toN-Bin n
toN-Bin (2b n) = 2 + 2 * toN-Bin n
```

As a more exotic example, we can describe a number system

```
data Carpal : Type where
  0c : Carpal
  1c : Carpal
  2c : Phalanx → Carpal → Phalanx → Carpal

toN-Carpal : Carpal → N
toN-Carpal 0c = 0
toN-Carpal 1c = 1
toN-Carpal (2c l m r) = toN-Phalanx l
                        + 2 * toN-Carpal m
                        + toN-Phalanx r
```

which consists of smaller “number systems”:

```
data Phalanx : Type where
  1p 2p 3p : Phalanx

toN-Phalanx : Phalanx → N
toN-Phalanx 1p = 1
toN-Phalanx 2p = 2
toN-Phalanx 3p = 3
```

We could now define a general number system as a type `N` equipped with a map `N → N`, but this would both be too general for our purpose and opaque to generic programs. On the other hand, allowing only traditional positional number systems excludes number systems like `Carpal`, which would otherwise still have valid numerical representations, as we will see later.

Across the above examples, the interpretation of a number is computed by a simple fold. In particular, leaves have associated constants, recursive fields correspond to multiplication and addition, while fields can defer to another function. We can thus modify `Con-sop` to encode each of these systems. The modified constructor descriptions `Con-num` associate a single number to each `1` and `ρ`, and a function to each `σ`:

```
data Con-num : Type where
  1 : N → Con-num
```

```

ρ : ℕ → Con-num → Con-num
σ : (S : Type) → (S → ℕ) → Con-num → Con-num

```

This essentially encodes number systems as trees that evaluate nodes by linearly combining values of subnodes, generalizing *dense* representations of positional number systems²¹.

We can encode the examples we gave as follows:

```

Nat-num : U-num
Nat-num = zeroD :: sucD :: []
where
  zeroD = 1 0
  sucD  = ρ 1
        ( 1 1 )

```

The binary numbers admit a similar encoding, but multiply their recursive fields by two instead:

```

Bin-num : U-num
Bin-num = 0bD :: 1bD :: 2bD :: []
where
  0bD = 1 0
  1bD = ρ 2
        ( 1 1 )
  2bD = ρ 2
        ( 1 2 )

```

Finally, the *Carpal* system can be encoded by using the interpretation of *Phalanx*

```

Carpal-num : U-num
Carpal-num = 0cD :: 1cD :: 2cD :: []
where
  0cD = 1 0
  1cD = 1 1
  2cD = σ Phalanx toN-Phalanx -- : Phalanx
        ( ρ 2 -- → Carpal
          ( σ Phalanx toN-Phalanx -- → Phalanx
            ( 1 0 ))) -- → Carpal

```

3.2.2 Nested types

If our construction is going to cast *Random*, as defined in Section 1, as the numerical representation associated to *Bin*, then *Random* needs to have a description to begin with. However, the recursive fields of *Random* are not given the parameter *A*, but $A \times A$. This makes *Random* a nested type, as opposed to a uniformly recursive type, in which the parameters of the recursive fields would be identical to the top-level parameters. Consequently, *Random* has no adequate description in *U-ix*²².

²¹As a consequence, this excludes the *sparse* number systems, as we discuss in Section 6.8.

²²Here, the “inadequate” descriptions either hardly resemble the user defined *Random*, use indices to store the depth of a node (which we work out in C Appendix C), or only have a complicated isomorphism to *Random*.

Due to the work of Johann and Ghani [JG07], we can model general nested types as fixpoints of *higher-order functors* (i.e., endofunctors on the category of endofunctors):

```
Fun  = Type → Type
HFun = Fun → Fun

{-# NO_POSITIVITY_CHECK #-}
data HMu (H : HFun) (A : Type) : Type where
  con : H (HMu H) A → HMu H A
```

By placing the recursive field `Mu F` under `F`, the functor `F` can modify `Mu F` and `A` to determine the type of the recursive field. We can encode `Random` by a `HFun` as:

```
data HRandom (F : Fun) (A : Type) : Type where
  Zero : HRandom F A
  One  : A → F (A × A) → HRandom F A
  Two  : A → A → F (A × A) → HRandom F A
```

However, this definition is unsafe²³, so we settle for the weaker but safe inner nesting instead. Rather than placing the function that describes the nesting around the fixpoint like in `HMU`, we precisely emulate nested types which only modify their parameters.

When a type has parameters Γ , we can describe a change in parameters by a map $g : \text{Cxf } \Gamma \rightarrow \Gamma$ from Γ to itself. So, we modify the recursive field ρ of `U-ix` to be

```
 $\rho : V \vdash I \rightarrow \text{Cxf } \Gamma \rightarrow \text{Con-nest } \Gamma \vee I \rightarrow \text{Con-nest } \Gamma \vee I$ 
```

and update the interpretation of ρ to g before passing ρ to the recursive field X :

```
 $[ \rho \ j \ g \ C ] \text{C-nest } X \text{ pv} @ (p, v) \ i = X (g \ p) (j \ \text{pv}) \times [ C ] \text{C-nest } X \text{ pv } i$ 
```

With this modification, `Random` can be directly transcribed

```
RandomD : U-nest (0 ▷ λ _ → Type) τ
RandomD = ZeroD :: OneD :: TwoD :: []
where
  ZeroD = 1 _ -- : Random A
  OneD  = σ (λ { ((_, A), _) → A }) -- : A
         ( ρ _ (λ { ((_, A), _) → A × A }) -- → Random (A × A)
         ( 1 _ )) -- → Random A
  TwoD  = σ (λ { ((_, A), _) → A }) -- : A
         ( σ (λ { ((_, A), _) → A }) -- → A
         ( ρ _ (λ { ((_, A), _) → A × A }) -- → Random (A × A)
         ( 1 _ ))) -- → Random A
```

using the map $A \mapsto A \times A$ to describe its nesting as usual.

Uniformly recursive types then simply become the nested types which only use the identity as a parameter transformation:

```
 $\rho 0 : \forall \{V\} \rightarrow V \vdash I \rightarrow \text{Con-nest } \Gamma \vee I \rightarrow \text{Con-nest } \Gamma \vee I$ 
 $\rho 0 \ v \ C = \rho \ v \ \text{id } C$ 
```

²³As you might have deduced from the pragma disabling the positivity checker. Consider `HBad F A = F A → 1`.

3.2.3 Composite types

In Section 3.2.1, we defined the number system `Carpal-num` as a *composite type*, in the sense that its description references another concrete type `Phalanx`. By the same argument as there, the description `Carpal-num` which relies on `toN-Phalanx` to describe the value of `Phalanx`, turns out to be too imprecise to recover the complete numerical representation generically.

In comparison, generic programming facilities like the deriving-mechanism in Haskell allow for code like

```
{-# LANGUAGE DeriveFunctor #-}

data Two a = Two a a deriving Functor
data Even a = Zero | More (Two a) (Even a) deriving Functor
```

In this example, we can define lists of even numbers of elements as lists of pairs of elements, and the Functor instance for `Even` can be derived generically, using that `Two` has a (derived) Functor instance. This would not work for `U-ix` or `U-num`, as a generic function would not be able to decide whether a field is of the form μD to begin with.

Inlining the constructors of `Phalanx` into `Carpal` does allow generic constructions to see the structure of `Phalanx`, but is undesirable in this case and in general. Here, it would yield a type with two of the original constructors of `Carpal`, and 9 more constructors for each combination of constructors of `Phalanx`²⁴.

In order to make the descriptions of fields that have them visible to generics, we simply add a more specific former of fields to `U-ix` and call the resulting universe `U-comp` for now. The new former δ in `U-comp` is specialized to adding *composite fields* from provided descriptions

$$\begin{aligned} \delta &: (R : U\text{-comp } \Delta J) (d : Cxf \Gamma \Delta) (j : I \rightarrow J) \\ &\rightarrow Con\text{-comp } \Gamma V I \rightarrow Con\text{-comp } \Gamma V I \end{aligned}$$

This former then also has to take functions d and j to determine the parameters and indices passed to R . A composite field encoded by δ is then interpreted identically to how it would be if we used σ and μ instead²⁵:

$$\begin{aligned} [\delta R d j C] C\text{-comp } X \text{ pv}@ (p, v) i \\ = \mu\text{-comp } R (d p) (j i) \times [C] C\text{-comp } X \text{ pv } i \end{aligned}$$

Using δ rather than σ allows us to reveal the description of a field to a generic program. Instead of inserting a plain field containing `Phalanx` and `toN-Phalanx`, we can use δ to directly add `Phalanx-num` to `Carpal-num`.

3.2.4 Hiding variables

With the modifications described above, we can describe all the structures we want. However, there is one peculiarity in the way `U-ix` handles variables.

²⁴If working with 11 constructors sounds too feasible, consider that defining addition on types like `Carpal` (or concatenation on its numerical representation) is not (yet) generic and, if fully written out, will instead demand 121 manually written cases.

²⁵The omission of μR from the variable telescope is intentional. While adding it is workable, it also significantly complicates the treatment of ornaments.

Namely, each field S added by a σ is treated as a bound or dependent argument: Even if the value $(s : S)$ is then unused, all fields afterwards have to be treated as types depending on S . This only poses a minor inconvenience, but this does mean that two subsequent fields referring to the same variable will have to be encoded differently. Furthermore, adding fields of complicated types can quickly clutter the context when writing or inspecting a generic program.

With a simple modification to the handling of telescopes in **U-ix**, we can emulate both bound and unbound fields, without adding more formers to **U-ix**. By accepting a transformation of variables $Vxf \ \Gamma \ (V \triangleright S) \ W$ after a $\sigma \ S$ in the context of V , the remainder of the fields can be described in the context W :

$\sigma : (S : V \vdash \text{Type}) \rightarrow Vxf \ id \ (V \triangleright S) \ W \rightarrow \text{Con-var} \ \Gamma \ W \ I \rightarrow \text{Con-var} \ \Gamma \ V \ I$

Of course, it would be no use to redefine σ in an attempt to save the user some effort, while leaving them with the burden of manually adding these transformations. So, we define shorthands emulating precisely the bound field

$\sigma+ : \forall \{V\} \rightarrow (S : V \vdash \text{Type}) \rightarrow \text{Con-var} \ \Gamma \ (V \triangleright S) \ I \rightarrow \text{Con-var} \ \Gamma \ V \ I$

$\sigma+ \ S \ C = \sigma \ S \ id \ C$

and the unbound field

$\sigma- : \forall \{V\} \rightarrow (S : V \vdash \text{Type}) \rightarrow \text{Con-var} \ \Gamma \ V \ I \rightarrow \text{Con-var} \ \Gamma \ V \ I$

$\sigma- \ S \ C = \sigma \ S \ fst \ C$

3.3 A new Universe

Using the modifications described above we define a new universe based on **U-ix**, in which the descriptions are again lists of constructors:

data **DescI** (Me : **Meta**) (Γ : **Tel** τ) (I : **Type**) : **Type** **where**

[] : **DescI** Me Γ **I**

...:: : **ConI** Me Γ σ **I** \rightarrow **DescI** Me Γ **I** \rightarrow **DescI** Me Γ **I**

The universe **DescI** is also parametrized over the metadata **Meta**, generalizing the annotations from Section 3.2.1 which we will use later to encode number systems in **DescI**.

The constructors of described datatypes can be formed as follows:

data **ConI** (Me : **Meta**) (Γ : **Tel** τ) (V : **ExTel** Γ) (I : **Type**) : **Type** **where**

1 : {me : Me .**1i**} (i : $\Gamma \ \& \ V \vdash I$) \rightarrow **ConI** Me $\Gamma \ V \ I$

ρ : {me : Me .**pi**}
 (g : **Cxf** $\Gamma \ \Gamma$) (i : $\Gamma \ \& \ V \vdash I$) (C : **ConI** Me $\Gamma \ V \ I$)
 \rightarrow **ConI** Me $\Gamma \ V \ I$

σ : ($S : V \vdash \text{Type}$) {me : Me .**si** S }
 (w : **Vxf** $id \ (V \triangleright S) \ W$) (C : **ConI** Me $\Gamma \ W \ I$)
 \rightarrow **ConI** Me $\Gamma \ V \ I$

δ : {me : Me .**di** $\Delta \ J$ } {iff : **MetaF** Me' Me}
 (d : $\Gamma \ \& \ V \vdash [\Delta] \text{tel } tt$) (j : $\Gamma \ \& \ V \vdash J$)
 (R : **DescI** Me' $\Delta \ J$) (C : **ConI** Me $\Gamma \ V \ I$)
 \rightarrow **ConI** Me $\Gamma \ V \ I$

Remark that `1` is the same as before, but `ρ` now accepts the transformation `Cxf` Γ to encode non-uniform parameters. Likewise, `σ` takes a transformation `w` from $V \triangleright S$ to W , allowing us to replace the context $V \triangleright S$ after a field with a context W of our choice. Finally, `δ` is added to directly describe composite datatypes by giving a description R to represent a field μR .

Let us take a fresh look at some datatypes from before, now through the lens of `DescI`. We will leave the metadata aside for now by using:

```
Con = ConI Plain
Desc = DescI Plain
```

Like before, we use the shorthands `σ+`, `σ-`, and `ρ0` to avoid cluttering descriptions which do not make use of the corresponding features.

In `DescI`, we can encode `N` and `List` as before, replacing `σ` with `σ-` and `ρ` with `ρ0`:

```
NatD : Desc (∅ → τ)
NatD = zeroD :: sucD :: []
  where
    zeroD = 1 _ -- : ℕ
    sucD  = ρ0 _ -- : ℕ
           ( 1 _ ) -- → ℕ

ListD : Desc (∅ ▷ λ _ → Type) τ
ListD = nilD :: consD :: []
  where
    nilD = 1 _ -- : List A
    consD = σ- (λ ((_, A), _) → A) -- : A
           ( ρ0 _ -- → List A
             ( 1 _ ) -- → List A )
```

If we define `Vec`, we bind the length as a (implicit) field, for which we use `σ+` instead, so we can extract the length `n` in `ρ0` and `1`:

```
VecD : Desc (∅ ▷ λ _ → Type) ℕ
VecD = nilD :: consD :: []
  where
    nilD = 1 (λ _ → 0) -- : Vec A zero
    consD = σ- (λ ((_, A), _) → A) -- : A
           ( σ+ (λ _ → ℕ) -- → (n : ℕ)
             ( ρ0 (λ (_ , ( _ , n)) → n) -- → Vec A n
               ( 1 (λ (_ , ( _ , n)) → suc n))) -- → Vec A (suc n) )
```

By passing a recursive field `ρ` the function taking A to $A \times A$, we can almost repeat the definition of `Random` from `U-nest`:

```
RandomD : Desc (∅ ▷ λ _ → Type) τ
RandomD = ZeroD :: OneD :: TwoD :: []
  where
    ZeroD = 1 _ -- : RandomD A
    OneD  = σ- (λ ((_, A), _) → A) -- : A
           ( ρ (λ (_ , A) → (_ , (A × A))) _ -- → Random (A × A)
             ( 1 _ ) -- → Random A )
```

```

TwoD = σ- (λ ((- , A) , -) → A)      -- : A
      ( σ- (λ ((- , A) , -) → A)      -- → A
        ( ρ (λ (- , A) → (- , (A × A))) -- → Random (A × A)
          ( 1 - )))                  -- → Random A

```

Binary finger trees (as opposed to 2-3 finger trees [HP06]) are the numerical representation associated to `Carpal`. Like `Random`, they are a nested datatype, but instead store their elements in variably sized digits on both sides instead. In `DescI`, we can then first define digits as a datatype which holds one to three elements:

```

DigitD : Desc (∅ ▷ λ _ → Type) τ
DigitD = OneD :: TwoD :: ThreeD :: []
where
  OneD = σ- (λ ((- , A) , -) → A) -- : A
        ( 1 - )                  -- → Digit A
  TwoD = σ- (λ ((- , A) , -) → A) -- : A
        ( σ- (λ ((- , A) , -) → A) -- → A
          ( 1 - )                  -- → Digit A
        )
  ThreeD = σ- (λ ((- , A) , -) → A) -- : A
          ( σ- (λ ((- , A) , -) → A) -- → A
            ( σ- (λ ((- , A) , -) → A) -- → A
              ( 1 - )))              -- → Digit A

```

Then, we can use `δ` to include them into a separate description of finger trees:

```

FingerD : Desc (∅ ▷ λ _ → Type) τ
FingerD = EmptyD :: SingleD :: DeepD :: []
where
  EmptyD = 1 -                      -- : Finger A
  SingleD = σ- (λ ((- , A) , -) → A) -- : A
            ( 1 - )                  -- → Finger A
  DeepD = δ (λ (p , -) → p) - DigitD -- : Digit A
          ( ρ (λ (- , A) → (- , (A × A))) -- → Finger (A × A)
            ( δ (λ (p , -) → p) - DigitD -- → Digit A
              ( 1 - )))              -- → Finger A

```

These descriptions can be instantiated as before by taking the fixpoint

```

data μ (D : DescI Me Γ I) (p : [ Γ ]tel tt) : I → Type where
  con : ∀ {i} → [ D ]D (μ D) p i → μ D p i

```

of their interpretations as functors

```

[ ]C : ConI Me Γ V I → ( [ Γ ]tel tt → I → Type)
      → [ Γ & V ]tel → I → Type
[ 1 i' ]C X pv      i = i ≡ i' pv
[ ρ g i' D ]C X pv@(p , v) i = X (g p) (i' pv) × [ D ]C X pv i
[ σ S w D ]C X pv@(p , v) i = Σ[ s ∈ S pv ] [ D ]C X (p , w (v , s)) i
[ δ d j R D ]C X pv      i = Σ[ s ∈ μ R (d pv) (j pv) ] [ D ]C X pv i

[ ]D : DescI Me Γ I → ( [ Γ ]tel tt → I → Type)
      → [ Γ ]tel tt → I → Type
[ [ ] ]D X p i = 1

```

$\llbracket C :: D \rrbracket D X p i = (\llbracket C \rrbracket C X (p, tt) i) \uplus (\llbracket D \rrbracket D X p i)$
 inserting the transformations of parameters g in ρ and the transformations of variables w in σ .

Like `U-ix`, we can define a generic `fold` for `DescI`

`fold` : $\forall \{D : \text{DescI Me } \Gamma \text{ I}\} \{X\} \rightarrow \llbracket D \rrbracket D X \rightarrow_3 X \rightarrow \mu D \rightarrow_3 X$

which comes in equally handy when using ornaments.

3.3.1 Annotating Descriptions with Metadata

We promised encodings of number systems in `DescI`, so let us explain how number systems can be described as *metadata* and how this lets us use `DescI` in the same way we used `U-num` to describe type and numerical value in the same description.

By generalizing `DescI` over `Meta`, rather than coding the specification of number systems into the universe directly, we give ourselves the flexibility to both represent plain datatypes and number systems in the same universe. The specific `Meta` passed to `DescI` determines the types of information to be queried (in the implicit `me` fields) at each of the type-formers. A term of `Meta` simply lists the type of information to be queried at each type former²⁶:

```
record Meta : Type where
  field
    1i : Type
    pi : Type
    oi : (S :  $\Gamma \& V \vdash \text{Type}$ )  $\rightarrow$  Type
    di : Tel  $\tau \rightarrow \text{Type} \rightarrow \text{Type}$ 
```

In composite fields `di`, the metadata on the other description is not necessarily the same as the top-level metadata. When this happens, we ask that both sides are related by a transformation

```
record MetaF (L R : Meta) : Type where
  field
    1f : L .1i  $\rightarrow$  R .1i
    pf : L .pi  $\rightarrow$  R .pi
    of : {V : ExTel  $\Gamma$ } (S : V  $\vdash$  Type)  $\rightarrow$  L .oi S  $\rightarrow$  R .oi S
    df :  $\forall \Gamma A \rightarrow$  L .di  $\Gamma A \rightarrow$  R .di  $\Gamma A$ 
```

making it possible to downcast (or upcast) between the different types of metadata. This, for example, allows one to include an annotated type `DescI Me` into an ordinary datatype `Desc` without duplicating the former definition in `Desc` first.

The encoding of number systems by associating numbers to `1i` and `pi`, and functions to `oi`, can be summarized as:

```
Number : Meta
Number .1i = N
Number .pi = N
```

²⁶One can compare this to how generic representations of datatypes in Haskell can be (optionally) annotated with metadata making the names of datatypes, constructors and fields available on the type level.

```

Number . $\delta$ i S =  $\forall p \rightarrow S p \rightarrow \mathbb{N}$ 
Number . $\delta$ i  $\Gamma J = (\Gamma \equiv \emptyset) \times (J \equiv \tau) \times \mathbb{N}$ 

```

We let the composite field δ , which was not described when we discussed encoding number systems in `U-num`, act similar to ρ , also multiplying the value in its field by a constant; The equalities in the metadata of a δ ensure that number systems have no parameters or indices.

Using `Number`, we describe the binary numbers `Bin-num` in `DescI` as:

```

BinND : DescI Number  $\emptyset$   $\tau$ 
BinND = 0bD :: 1bD :: 2bD :: []
where
  0bD =  $\mathbf{1}$  {me = 0} _
  1bD =  $\rho 0$  {me = 2} _
        (  $\mathbf{1}$  {me = 1} _ )
  2bD =  $\rho 0$  {me = 2} _
        (  $\mathbf{1}$  {me = 2} _ )

```

The metadata transformations help us when we represent `Carpal-num` in its more accurate form, by first defining

```

PhalanxND : DescI Number  $\emptyset$   $\tau$ 
PhalanxND = 1pD :: 2pD :: 3pD :: []
where
  1pD =  $\mathbf{1}$  {me = 1} _
  2pD =  $\mathbf{1}$  {me = 2} _
  3pD =  $\mathbf{1}$  {me = 3} _

```

and directly including it into `Carpal`

```

CarpalND : DescI Number  $\emptyset$   $\tau$ 
CarpalND = 0cD :: 1cD :: 2cD :: []
where
  0cD =  $\mathbf{1}$  {me = 0} _
  1cD =  $\mathbf{1}$  {me = 1} _
  2cD =  $\delta$  {me = refl, refl, 1} {id-MetaF} _ _ PhalanxND
        (  $\rho 0$  {me = 2} _
          (  $\delta$  {me = refl, refl, 1} {id-MetaF} _ _ PhalanxND
            (  $\mathbf{1}$  {me = 0} _ ) ) )

```

where we can use the identity function to indicate both sides have metadata of type `Number`.

The metadata on a `DescI Number` can then be used to define a generic function sending terms of number systems to their `value` in \mathbb{N}

```

value : {D : DescI Number  $\Gamma$   $\tau$ }  $\rightarrow \forall \{p\} \rightarrow \mu D p \text{ tt} \rightarrow \mathbb{N}$ 

```

which is defined by generalizing over the inner metadata and `folding` using:

```

value-desc : (D : DescI Me  $\Gamma$   $\tau$ )  $\rightarrow \forall \{a b\} \rightarrow [ D ] D (\lambda \_ \_ \rightarrow \mathbb{N}) a b \rightarrow \mathbb{N}$ 
value-con : (C : ConI Me  $\Gamma$   $\tau$ )  $\rightarrow \forall \{a b\} \rightarrow [ C ] C (\lambda \_ \_ \rightarrow \mathbb{N}) a b \rightarrow \mathbb{N}$ 

```

```

value-desc (C :: D) (inj1 x) = value-con C x
value-desc (C :: D) (inj2 y) = value-desc D y

value-con ( $\mathbf{1}$  {me = k} j) refl

```

```

=  $\phi$  . 1f k

value-con ( $\rho$  {me = k} g j C) (n , x)
=  $\phi$  . pf k * n + value-con C x

value-con ( $\sigma$  S {me = S $\rightarrow$ N} h C) (s , x)
=  $\phi$  . sf _ S $\rightarrow$ N _ s + value-con C x

value-con ( $\delta$  {me = me} {iff = iff} g j R C) (r , x)
with  $\phi$  . df _ _ me
... | refl , refl , k
= k * value-lift R ( $\phi$   $\circ$  MetaF iff) r + value-con C x

```

Furthermore, also possible to use **Meta** to encode conventionally useful metadata such as field names:

```

Names : Meta
Names .1i =  $\tau$ 
Names .pi = String
Names .si _ = String
Names .di _ _ = String

```

On the other extreme, we can also declare that a description has no metadata at all by querying τ for all type-formers:

```

Plain : Meta
Plain .1i =  $\tau$ 
Plain .pi =  $\tau$ 
Plain .si _ =  $\tau$ 
Plain .di _ _ =  $\tau$ 

```

Because the queries for metadata are implicit in **DescI**, descriptions from **U-ix** can be imported into **Desc**, without having to insert metadata anywhere.

4 Ornaments

In the framework of **DescI** of the last section, we can write down a number system and its meaning in one description, and we can use this as the starting point for constructing numerical representations. To write down a generic construction of numerical representations from number systems, we will need a language in which we can describe modifications on the number systems.

In this section, we will describe the ornamental descriptions for the **DescI** universe, and explain their working by means of examples. As we will be constructing new datatypes, rather than relating pre-existing ones, we omit the definition of the ornaments.

4.1 Ornamental descriptions

The ornamental descriptions for **DescI** take the same shape as those in Section 2.6, generalized to handle nested types, variable transformations, and composite types. These ornamental descriptions are defined such that a **OrnDesc**

$\text{Me}' \Delta \text{re-par } J \text{ re-index } D$ represents a patch from a base description D to a description with metadata Me' , parameters Δ and indices J .

Note that metadata, as a non-structural property, has no direct influence on ornaments. So, we simply generalize over the metadata on D , querying the metadata for the new description without imposing constraints.

As always, we start off by defining ornamental descriptions as lists of constructor ornaments :

```
data OrnDesc {Me} (Me' : Meta) (Δ : Tel τ)
  (re-par : Cxf Δ Γ) (J : Type) (re-index : J → I)
  : DescI Me Γ I → Type where
[] : OrnDesc Me' Δ re-par J re-index []
... :: ConOrnDesc Me' {re-par = re-par} ! re-index {Me = Me} CD
  → OrnDesc Me' Δ re-par J re-index D
  → OrnDesc Me' Δ re-par J re-index (CD :: D)
```

Most of the modifications in `DescI` are reflected in the constructor ornaments, and as a consequence this is also where we pay the price for the flexibility we built into `ConI`. For example, because `ConI` allows us to transform variables, `ConOrnDesc` has to relate the transformations on both sides in order for `ornForget` to exist. We (have to) dedicate a lot of lines to such commutativity squares of variables, but these squares involving `Vxf` can generally be ignored; this is witnessed by the `Oσ+` and `Oσ-` variants of the `σ` ornament, automatically filling those squares in the usual cases of binding or ignoring fields.

The structure-preserving ornaments are defined as usual

```
data ConOrnDesc (Me' : Meta) {re-par : Cxf Δ Γ}
  (re-var : Vxf re-par W V) (re-index : J → I)
  : ConI Me Γ V I → Type where
1 : {i : Γ & V ⊢ I} (j : Δ & W ⊢ J)
  → re-index ∘ j ~ i ∘ var→par re-var
  → {me : Me .1i} {me' : Me' .1i}
  → ConOrnDesc Me' re-var re-index (1 {Me} {me = me} i)

ρ : {g : Cxf Γ Γ} (d : Cxf Δ Δ)
  {i : Γ & V ⊢ I} (j : Δ & W ⊢ J)
  → g ∘ re-par ~ re-par ∘ d
  → re-index ∘ j ~ i ∘ var→par re-var
  → {me : Me .pi} {me' : Me' .pi}
  → ConOrnDesc Me' re-var re-index CD
  → ConOrnDesc Me' re-var re-index (ρ {Me} {me = me} g i CD)

σ : (S : Γ & V ⊢ Type)
  {g : Vxf id (V ▷ S) V'} (h : Vxf id (W ▷ (S ∘ var→par re-var)) W')
  (v' : Vxf re-par W' V')
  → (V {p} → g ∘ Vxf▷ re-var S ~ v' {p = p} ∘ h)
  → {me : Me .σi S} {me' : Me' .σi (S ∘ var→par re-var)}
  → ConOrnDesc Me' v' re-index CD
  → ConOrnDesc Me' re-var re-index (σ {Me} S {me = me} g CD)
```

```

 $\delta : (R : \text{DescI If'' } \Theta K) (t : \Gamma \& V \vdash [\Theta] \text{tel tt}) (j : \Gamma \& V \vdash K)$ 
 $\rightarrow \{me : Me .\delta i \Theta K\} \{iff : \text{MetaF If'' Me}\}$ 
 $\{me' : Me' .\delta i \Theta K\} \{iff' : \text{MetaF If'' Me'}\}$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index CD}$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index}$ 
 $(\delta \{Me\} \{me = me\} \{iff = iff\} t j R CD)$ 

```

where ρ has a new field relating the old and new nesting transforms g and d . Likewise, σ now has a field relating the old and new variable transforms, which for example prevents us from unbinding a field in the new description which was used in the old description. The ornament δ now represents the direct copying of a δ in descriptions (up to re-par and re-var).

Where only $\Delta\sigma$ could add fields before, we can now also add fields described by δ using $\Delta\delta$:

```

 $\Delta\sigma : (S : \Delta \& W \vdash \text{Type}) (h : Vxf \text{ id } (W \triangleright S) W')$ 
 $(v' : Vxf \text{ re-par } W' V)$ 
 $\rightarrow (\forall \{p\} \rightarrow \text{re-var } \circ \text{fst} \sim v' \{p = p\} \circ h)$ 
 $\rightarrow \{me' : Me' .\sigma i S\}$ 
 $\rightarrow \text{ConOrnDesc Me' v' re-index CD}$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index CD}$ 

 $\Delta\delta : (R : \text{DescI If'' } \Theta J) (t : W \vdash [\Theta] \text{tel tt}) (j : W \vdash J)$ 
 $\rightarrow \{me' : Me' .\delta i \Theta J\} \{iff' : \text{MetaF If'' Me'}\}$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index CD}$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index CD}$ 

```

Again, $\Delta\sigma$ requires the relation of old and new variables.

Now, if we have a description D' with a composite field $\delta R d j R D$ referencing R , then we expect that a patch on R also induces a patch on D' . We generalize this by defining a kind of sequential composition of ornaments²⁷, taking two ornamental descriptions, one on R and one on D , and producing an ornamental description on D' :

```

 $\bullet\delta : \{R : \text{DescI If'' } \Theta K\} \{c' : Cxf \wedge \Theta\} \{f\Theta : V \vdash [\Theta] \text{tel tt}\}$ 
 $(f\wedge : W \vdash [\wedge] \text{tel tt}) \{k' : M \rightarrow K\} \{k : V \vdash K\}$ 
 $(m : W \vdash M)$ 
 $\rightarrow (RR' : \text{OrnDesc If''' } \wedge c' M k' R)$ 
 $\rightarrow (p_1 : \forall q w \rightarrow c' (f\wedge (q, w)) \equiv f\Theta (\text{re-par } q, \text{re-var } w))$ 
 $\rightarrow (p_2 : \forall q w \rightarrow k' (m (q, w)) \equiv k (\text{re-par } q, \text{re-var } w))$ 
 $\rightarrow \forall \{me\} \{iff\} \{me' : Me' .\delta i \wedge M\} \{iff' : \text{MetaF If''' Me'}\}$ 
 $\rightarrow (DE : \text{ConOrnDesc Me' re-var re-index CD})$ 
 $\rightarrow \text{ConOrnDesc Me' re-var re-index}$ 
 $(\delta \{Me\} \{me = me\} \{iff = iff\} f\Theta k R CD)$ 

```

If we try to forget $\bullet\delta$, the parameters to R can be computed in two ways. Namely, we can first convert back to the context of CD according to DE and compute the parameter for R there with the original $f\Theta$, or we can first compute the parameter

²⁷As opposed to Ko's parallel composition [Ko14], which composes two ornaments on the same description D , producing something that incorporates changes from both.

in the new context using the new $f\Lambda$ and then convert this back to the parameter for R according to RR' . We can summarize both routes in a diagram:

$$\begin{array}{ccc} W\&\Delta & \xrightarrow{f\Lambda} & \Lambda \\ \text{re-var} \times \text{re-index} \downarrow & & & \downarrow c' \\ V\&\Gamma & \xrightarrow{f\Theta} & \Theta \end{array}$$

To avoid any ambiguity that arises from this, we require that both ways around this square are equal. Using these and the other new commutativity squares, we can again define `ornForget` from an ornamental algebra analogous to the one for `U-ix`:

```

ornForget : {re-var : Cxf Δ Γ} {re-index : J → I} {D : DescI Me Γ I}
  → (OD : OrnDesc Me' Δ re-var J re-index D)
  → μ (toDesc OD) →3 λ d j → μ D (re-var d) (re-index j)
ornForget OD = fold (ornAlg OD)

```

The precise meaning of ornamental descriptions as descriptions is given by the conversion

```

toDesc : {re-var : Cxf Δ Γ} {re-index : J → I} {D : DescI Me Γ I}
  → OrnDesc Me' Δ re-var J re-index D → DescI Me' Δ J
toDesc [] = []
toDesc (C0 :: 0) = toCon C0 :: toDesc 0

toCon : {re-par : Cxf Δ Γ} {re-var : Vxf re-par W V}
  {re-index : J → I} {D : ConI Me Γ V I}
  → ConOrnDesc Me' re-var re-index D → ConI Me' Δ W J
toCon (1 j _ {me' = me})
  = 1 {me = me} j
toCon (p j h _ _ {me' = me} C0)
  = p {me = me} j h (toCon C0)
toCon {re-var = v} (σ S h _ _ {me' = me} C0)
  = σ (S ◦ var→par v) {me = me} h (toCon C0)
toCon {re-var = v} (δ R j t {me' = me} {iff' = iff} C0)
  = δ {me = me} {iff = iff} (j ◦ var→par v) (t ◦ var→par v) R (toCon C0)
toCon (Δσ S h _ _ {me' = me} C0)
  = σ S {me = me} h (toCon C0)
toCon (Δδ R t j {me' = me} {iff' = iff} C0)
  = δ {me = me} {iff = iff} t j R (toCon C0)
toCon (•δ fΛ m RR' _ _ {me' = me} {iff' = iff} C0)
  = δ {me = me} {iff = iff} fΛ m (toDesc RR') (toCon C0)

```

which makes use of the implicit metadata fields in the constructor ornaments to reconstruct the metadata on the target description.

With `OrnDesc` we can reproduce the examples of the ornamental descriptions for `U-ix`, and present some previously inexpressible types as ornamental descriptions. For this, we use the variants of some ornaments specialized to binding or ignoring fields, such as:

```
Oσ+ : (S : Γ & V ⊢ Type) {CD : ConI Me Γ V' I} {h : Vxf _ _ _}
  → {me : Me .σi S} {me' : Me' .σi (S ◦ var→par re-var)}
  → ConOrnDesc Me' (h ◦ Vxf→ re-var S) re-index CD
  → ConOrnDesc Me' re-var re-index (σ {Me} S {me = me'} h CD)
Oσ+ S {h = h} {me' = me'} CO
  = σ S id (h ◦ Vxf→ re-var S) (λ _ → refl) {me' = me'} CO
```

We can then give the familiar ornamental description from `List` to `Vec`:

```
VecOD : OrnDesc Plain (∅ ▷ λ _ → Type) id N ! ListD
VecOD = nilOD :: consOD :: []
where
nilOD = 1 (λ _ → zero) (λ _ → refl)
consOD = OΔσ+ (λ _ → N)
  ( Oσ- (λ ((_, A), _) → A)
    ( Op0 (λ (_, (_, n)) → n) (λ _ → refl)
      ( 1 (λ (_, (_, n)) → suc n) (λ _ → refl))))
```

Rather than defining `Random` on its own, we can use the new flexibility in `ρ` and describe random access lists as an ornament from binary numbers:

```
RandomOD : OrnDesc Plain (∅ ▷ λ _ → Type) ! τ id BinND
RandomOD = ZeroOD :: OneOD :: TwoOD :: []
where
ZeroOD = 1 _ (λ _ → refl)
OneOD = OΔσ- (λ ((_, A), _) → A)
  ( ρ (λ (_, A) → (_, Pair A)) - (λ _ → refl) (λ _ → refl)
    ( 1 _ (λ _ → refl)))
TwoOD = OΔσ- (λ ((_, A), _) → A)
  ( OΔσ- (λ ((_, A), _) → A)
    ( ρ (λ (_, A) → (_, Pair A)) - (λ _ → refl) (λ _ → refl)
      ( 1 _ (λ _ → refl))))
```

Likewise, we can give an ornament turning phalanges into digits

```
DigitOD : OrnDesc Plain (∅ ▷ λ _ → Type) ! τ id PhalanxND
DigitOD = OneOD :: TwoOD :: ThreeOD :: []
where
OneOD = OΔσ- (λ ((_, A), _) → A)
  ( 1 _ (λ _ → refl))
TwoOD = OΔσ- (λ ((_, A), _) → A)
  ( OΔσ- (λ ((_, A), _) → A)
    ( 1 _ (λ _ → refl)))
ThreeOD = OΔσ- (λ ((_, A), _) → A)
  ( OΔσ- (λ ((_, A), _) → A)
    ( OΔσ- (λ ((_, A), _) → A)
      ( 1 _ (λ _ → refl))))
```

and assemble these into finger trees with `δ`.

```

FingerOD : OrnDesc Plain (⊘ ▷ λ _ → Type) ! τ id CarpalND
FingerOD = EmptyOD :: SingleOD :: DeepOD :: []
where
  EmptyOD = 1_ (λ _ → refl)
  SingleOD = OΔσ- (λ ((_, A), _) → A)
    ( 1_ (λ _ → refl) )
  DeepOD = •δ (λ (p, _) → p) _ DigitOD (λ _ _ → refl) (λ _ _ → refl)
    ( ρ (λ (_, A) → (_, (A × A))) _ (λ _ → refl) (λ _ → refl)
    ( •δ (λ (p, _) → p) _ DigitOD (λ _ _ → refl) (λ _ _ → refl)
    ( 1_ (λ _ → refl) ) ) )

```

5 Generic Numerical Representations

The ornamental descriptions together with the descriptions and number systems from before complete the toolset we will use to construct numerical representations as ornaments.

In summary, using `DescI Number` to represent number systems, we paraphrase calculations like in Section 3.1 as ornaments, rather than direct definitions. In fact, we have already seen ornaments to numerical representations before, such as `ListOD` and `RandomOD`. Generalizing those ornaments, we construct numerical representations by means of an ornament-computing function, sending number systems to the ornamental descriptions that describe their numerical representations.

5.1 Unindexed Numerical Representations

In this section we demonstrate the generic computation of numerical representations. We proceed differently from the calculation of `Vec` from `N`. Indeed, we will give ornamental descriptions, rather than deriving direct definitions via step-by-step isomorphism reasoning. Nevertheless, the choices we make when inserting fields depending on the analysis of a number system follow the same strategy.

We will first present the unindexed numerical representations, explaining case-by-case which fields it adds and why. In the next section, we will demonstrate the indexed numerical representations as an ornament on top of the unindexed variant.

The unindexed representations are computed by `TreeOD` in the form of ornamental descriptions, sending a number system to the corresponding type of (nested) full trees over it. The ornament is computed by cases on the number system, and in each case the size of the numerical representation has to match up with the `value` of the number system.

Let us refer to the sole parameter of a numerical representation as `A`, and consider the case of a leaf of value `k`:

```

1-case : N → ConI Number ⊘ V τ
1-case k = 1 {me = k} _

```

In this case, the leaf contributes a constant k to the `value`, so a numerical representation should accordingly have k fields of A before this leaf, or equivalently a field containing k values of A . A recursive field of weight k

```


$\rho$ -case :  $\mathbb{N} \rightarrow \text{ConI Number } \emptyset \text{ V } \tau \rightarrow \text{ConI Number } \emptyset \text{ V } \tau$   

 $\rho$ -case  $k \text{ C} = \rho\emptyset \{me = k\} \_ \text{C}$


```

multiplies the value contributed by the recursive part by k . Hence, the numerical representation should have a recursive field, in such a way that a recursive value of size x actually represents $k * x$ values of A . On the other hand, an ordinary field S containing s , of which the value is computed as $f \ s$

```


$\sigma$ -case :  $(S : V \vdash \text{Type}) \rightarrow (\forall p \rightarrow S \rightarrow \mathbb{N})$   

 $\rightarrow \text{ConI Number } \emptyset \text{ V } \tau \rightarrow \text{ConI Number } \emptyset \text{ V } \tau$   

 $\sigma$ -case  $S \ f \ \text{C} = \sigma \_ S \{me = f\} \text{C}$


```

is simply represented in the numerical representation by adding a field with $f \ s$ values of A . Finally, a field containing another number system R with weight k

```


$\delta$ -case :  $\mathbb{N} \rightarrow \text{DescI Number } \emptyset \ \tau \rightarrow \text{ConI Number } \emptyset \text{ V } \tau \rightarrow \text{ConI Number } \emptyset \text{ V } \tau$   

 $\delta$ -case  $k \text{ R C} = \delta \{me = \text{refl}, \text{refl}, k\} \{\text{id-}\text{MetaF}\} \_ \_ \text{R C}$


```

directly contributes values of R multiplied by k . The outer numerical representation should then replace R with its numerical representation NR , which should, like the recursive field, let its values weigh k times their size.

To describe the numerical representation, we encode these fields of weight k with k -element vectors, and in the same way, the multiplication by k in the cases of ρ and δ is modelled by nesting over a k -element vector. Combining all these cases and translating them to the language of ornaments we define the unindexed numerical representation:

```

TreeOD : (D : DescI Number  $\emptyset \ \tau$ )  $\rightarrow$  OrnDesc Plain ( $\emptyset \triangleright \lambda \_ \rightarrow \text{Type}$ ) !  $\tau$  ! D
TreeOD D = Tree-desc D id-MetaF
module TreeOD where mutual
Tree-desc : (D : DescI Me  $\emptyset \ \tau$ )  $\rightarrow$  MetaF Me Number
 $\rightarrow$  OrnDesc Plain ( $\emptyset \triangleright \lambda \_ \rightarrow \text{Type}$ ) !  $\tau$  ! D

Tree-desc []  $\phi = []$ 
Tree-desc (C :: D)  $\phi = \text{Tree-con } C \ \phi :: \text{Tree-desc } D \ \phi$ 

Tree-con : {re-var : Vxf ! W V} (C : ConI Me  $\emptyset \text{ V } \tau$ )  $\rightarrow$  MetaF Me Number
 $\rightarrow$  ConOrnDesc { $\Delta = \emptyset \triangleright \lambda \_ \rightarrow \text{Type}$ } {W = W}
{J =  $\tau$ } Plain re-var ! C

Tree-con ( $\mathbb{1} \{me = k\} \ j$ )  $\phi$ 
= O $\Delta$  $\sigma$ - ( $\lambda \ ((\_ , A) , \_) \rightarrow \text{Vec } A \ (\phi . \mathbb{1} f \ k)$ )
(  $\mathbb{1} \_ (\lambda \_ \rightarrow \text{refl})$ )

Tree-con ( $\rho \{me = k\} \_ \_ \text{C}$ )  $\phi$ 
=  $\rho \ (\lambda \ (\_ , A) \rightarrow (\_ , \text{Vec } A \ (\phi . \rho f \ k))) \_ (\lambda \_ \rightarrow \text{refl}) (\lambda \_ \rightarrow \text{refl})$ 
( Tree-con C  $\phi$ )

Tree-con ( $\sigma \ S \{me = f\} \ h \ \text{C}$ )  $\phi$ 
= O $\sigma$ + S
( O $\Delta$  $\sigma$ - ( $\lambda \ ((\_ , A) , \_ , s) \rightarrow \text{Vec } A \ (\phi . \sigma f \_ f \_ s)$ )
```

```

( Tree-con C  $\phi$  )

Tree-con ( $\delta$  {me = me} {iff = iff} g j R C)  $\phi$ 
  with  $\phi$  . $\delta$ f _ _ me
... | refl , refl , k
  = • $\delta$  ( $\lambda$  { ((_, A) , _)  $\rightarrow$  (_, Vec A k) } ) !
    (Tree-desc R ( $\phi$  •MetaF iff))
    ( $\lambda$  _ _  $\rightarrow$  refl) ( $\lambda$  _ _  $\rightarrow$  refl)
  ( Tree-con C  $\phi$  )

```

In most cases, we straightforwardly use $O\Delta\sigma$ - to insert vectors of the correct size. However, in the case of ρ , we can trivially change the nesting function to take the parameter A and give Vec A k as a parameter to the recursive field instead. In the case of δ , we similarly place the parameters in a vector, but these are now directed to the recursively computed numerical representation of R. This case is also why we generalize the whole construction over $\phi : \text{MetaF Me Number}$, as R is allowed to have a Meta that is not Number , as long as it is convertible to Number . Consequently, everywhere we use the “weight” represented by k in the construction, we first apply ϕ to compute the actual weights and values from Me.

As an example, let us take a look at how TreeOD transforms CarpalND to its numerical representation, FingerOD . Applying TreeOD sends leaves with a value of k to Vec A k , so applying it to PhalanxND yields

```

DigitOD : OrnDesc Plain ( $\phi \triangleright \lambda \_ \rightarrow \text{Type}$ ) !  $\tau$  id PhalanxND
DigitOD = OneOD :: TwoOD :: ThreeOD :: []
  where
OneOD   =  $O\Delta\sigma$ - ( $\lambda$  ((_, A) , _)  $\rightarrow$  Vec A 1)
          ( 1 _ ( $\lambda$  _  $\rightarrow$  refl) )
TwoOD   =  $O\Delta\sigma$ - ( $\lambda$  ((_, A) , _)  $\rightarrow$  Vec A 2)
          ( 1 _ ( $\lambda$  _  $\rightarrow$  refl) )
ThreeOD =  $O\Delta\sigma$ - ( $\lambda$  ((_, A) , _)  $\rightarrow$  Vec A 3)
          ( 1 _ ( $\lambda$  _  $\rightarrow$  refl) )

```

which, after expanding vectors of k elements into k fields, is equivalent to the DigitOD from before. The same happens for the first two constructors of CarpalND , replacing them with an empty vector and a one-element vector respectively. The last constructor is more interesting:

```

FingerOD : OrnDesc Plain ( $\phi \triangleright \lambda \_ \rightarrow \text{Type}$ ) !  $\tau$  id CarpalND
FingerOD = EmptyOD :: SingleOD :: DeepOD :: []
  where
EmptyOD  =  $O\Delta\sigma$ - ( $\lambda$  ((_, A) , _)  $\rightarrow$  Vec A 0)
            ( 1 _ ( $\lambda$  _  $\rightarrow$  refl) )
SingleOD =  $O\Delta\sigma$ - ( $\lambda$  ((_, A) , _)  $\rightarrow$  Vec A 1)
            ( 1 _ ( $\lambda$  _  $\rightarrow$  refl) )
DeepOD   = • $\delta$  ( $\lambda$  ((_, p) , _)  $\rightarrow$  (_, Vec p 1)) !
            DigitOD ( $\lambda$  _ _  $\rightarrow$  refl) ( $\lambda$  _ _  $\rightarrow$  refl)
            ( p ( $\lambda$  ( _ , A)  $\rightarrow$  _ , Vec A 2) _ ( $\lambda$  _  $\rightarrow$  refl) ( $\lambda$  _  $\rightarrow$  refl) )
            ( • $\delta$  ( $\lambda$  ((_, p) , _)  $\rightarrow$  (_, Vec p 1)) !

```

```

DigitOD (λ _ _ → refl) (λ _ _ → refl)
( OΔσ- (λ ((_, A), _) → Vec A 0)
( 1_ (λ _ → refl)) )))

```

The `PhalanxND` in the last constructor gets replaced with `DigitOD` via `O•δ+`, and the recursive field gets replaced by a recursive field nesting over vectors of length. Again, this is equivalent to `FingerOD`, up to wrapping values in length one vectors and inserting empty vectors.

5.2 Indexed Numerical Representations

Like how `List` has an ornament `VecOD` to its `N`-indexed variant `Vec`, we can also construct an ornament, which we will call `TrieOD` `D`, from the numerical representation `TreeOD` `D` to its `D`-indexed variant:

```

TrieOD : (N : DescI Number 0 τ)
→ OrnDesc Plain (0 ▷ λ _ → Type)
id (μ N tt tt) ! (toDesc (TreeOD N))
TrieOD N = Trie-desc N N (λ _ _ → con) id-MetaF

```

To continue the analogy to `VecOD`, we can use that `TreeOD` already sorts out how the parameters should be nested and how many fields have to be added; as a consequence, this ornament only has to add fields reflecting the recursive indices, which are used to report indices corresponding to the number of values of `A` contained in the numerical representation.

We accomplish this by threading the partially applied constructor `n` of the number system `N` through the resulting description; by feeding it all the sizes of the fields added by `TreeOD`, we can use `n` to compute the total size of an ornamented constructor.

In addition to generalizing over `Me` to facilitate the `δ` case as we did for `TreeOD`, we now also generalize over the index type `N'`. When mapping over the lists of constructors (i.e., descriptions), the choice of constructor also selects the corresponding constructor of `N'`:

```

Trie-desc : ∀ {Me} (N' : DescI Me 0 τ) (D : DescI Me 0 τ)
(n : [ D ]D (μ N') →3 μ N') (φ : MetaF Me Number)
→ OrnDesc Plain (0 ▷ λ _ → Type)
id (μ N' tt tt) ! (toDesc (Tree-desc D φ) )
Trie-desc N' [] n φ = []
Trie-desc N' (C :: D) n φ = Trie-con N' C (λ p w x → n _ _ (inj1 x)) φ
:: Trie-desc N' D (λ p w x → n _ _ (inj2 x)) φ

```

We define `Trie-con` by induction on `C`, binding the sizes of the subtrees, to be fed as arguments to the selected constructor `n`. Since we are continuing where `Tree-con` left off, we can copy most fields:

```

Trie-con : ∀ {Me} (N' : DescI Me 0 τ) {re-var : Vxf id W V}
{re-var' : Vxf ! V U} (C : ConI Me 0 U τ)
(n : ∀ p w → [ C ]C (μ N') (tt, re-var' (re-var {p = p} w)) → μ N' tt tt)
(φ : MetaF Me Number)
→ ConOrnDesc {Δ = 0 ▷ λ _ → Type} {W = W}

```

```

      {j =  $\mu$  N' tt tt} Plain {re-par = id}
      re-var ! (toCon (Tree-con {re-var = re-var'} C  $\phi$ ))
Trie-con N' (1 {me = k} j) n  $\phi$ 
= 0 $\sigma$ - _
  ( 1 ( $\lambda$  { (p , w)  $\rightarrow$  n p w refl }) ( $\lambda$  _  $\rightarrow$  refl))

Trie-con N' ( $\rho$  {me = k} g j C) n  $\phi$ 
= 0 $\Delta$  $\sigma$ + ( $\lambda$  _  $\rightarrow$   $\mu$  N' tt tt)
  ( $\rho$  ( $\lambda$  { (- , A)  $\rightarrow$  _ }) ( $\lambda$  { (p , w , i)  $\rightarrow$  i })
    ( $\lambda$  _  $\rightarrow$  refl) ( $\lambda$  _  $\rightarrow$  refl)
  (Trie-con N' C ( $\lambda$  { p (w , i) x  $\rightarrow$  n p w (i , x) })  $\phi$ ))

Trie-con N' ( $\sigma$  S {me = f} h C) n  $\phi$ 
= 0 $\sigma$ + (S  $\circ$  var $\rightarrow$ par _)
  ( 0 $\sigma$ - _
    (Trie-con N' C ( $\lambda$  { p (w , s) x  $\rightarrow$  n p w (s , x) })  $\phi$ ))

Trie-con N' ( $\delta$  {me = me} {iff = iff} g j R C) n  $\phi$ 
with  $\phi$  . $\delta$ f _ _ me
... | refl , refl , k
= 0 $\Delta$  $\sigma$ + ( $\lambda$  _  $\rightarrow$   $\mu$  R tt tt)
  (  $\bullet$  $\delta$  ( $\lambda$  ((- , A) , -)  $\rightarrow$  (- , Vec A k)) ( $\lambda$  { (p , w , i)  $\rightarrow$  i })
    (Trie-desc R R ( $\lambda$  _ _  $\rightarrow$  con) ( $\phi$   $\circ$  MetaF iff))
    ( $\lambda$  _ _  $\rightarrow$  refl) ( $\lambda$  _ _  $\rightarrow$  refl)
  (Trie-con N' C ( $\lambda$  { p (w , i) x  $\rightarrow$  n p w (i , x) })  $\phi$ ))

```

We only have to add fields in the cases for ρ and δ , and in both they are promptly passed as expected indices to the next field using $\lambda \{ (p , w , i) \rightarrow i \}$; the only difference is that for δ , since `Trie-desc R` will be R -indexed, we add a field of R rather than N' . The values of all fields, including σ are passed to n ; since n starts as one constructor C of N' , when we arrive at `1`, the final argument of n can be filled with simply `refl` to determine the actual index.

Since the N' -index i bound in the ρ case forces the number of elements in the recursive field to i , the value in the σ case corresponds to the number of elements added after this field, and the R -index i bound in the δ case likewise forces the number of elements in the subdescription to be i , we know that when we arrive at `1`, the total number of elements is exactly given by n , and thus `Trie-con` is correct. In turn, we find that `Trie-desc` and `TrieOD` correctly construct indexed numerical representations.

6 Conclusion and Discussion

In conclusion, we formulated a universe encoding `DescI` with which we can describe number systems in `DescI Number`. We adapted the language of ornamental descriptions `OrnDesc` to `DescI`, such that the numerical representations of the number systems could be seen as ornaments on top of the number systems. These ornaments and descriptions let us implement the generic programs `TreeOD`

and `TrieOD` which, from a number system compute the associated (un)indexed numerical representation, of which we informally outlined proofs of correctness.

While it is possible to write down a direct proof of correctness for `TrieOD` by comparing it to `Lookup` via `value`, and from this extract a proof of correctness for `TreeOD`, one might expect there to be a more useful and less laborious angle of attack.

Namely, we expect that if we define `PathOD` as a generic ornament from a `DescI Number` to the corresponding finite type (such that `PathOD ND n` is equivalent to `Fin (value n)`), then we can show that `TrieOD ND n` has a `tabulate/lookup` pair for `PathOD ND n`, from which it follows that `TrieOD ND n A` is equivalent to `PathOD ND n → A`, and in consequence `TrieOD ND` corresponds to `Vec`.

Due to the `remember-forget` isomorphism [McB14], we have that `TreeOD ND` is equivalent to $\Sigma (\mu \text{ ND}) (\text{TrieOD ND})$, and in turn we also find that `TreeOD ND` is a normal functor (also referred to as Traversable). This yields traversability of `TreeOD ND`, with as corollaries `toList`²⁸ and properties such as that `toList` is a lifting of `value` (again in the sense of [DM14]).

However, it turns out that `PathOD` is not so easy to define, as we can see by the following.

6.1 Σ -descriptions are more natural for expressing finite types

Due to our representation of types as sums of products, representing the finite types of larger number systems quickly becomes much more complex. Consider the binary numbers from before:

```
data Leibniz : Type where
  0b      : Leibniz
  1b_ 2b_ : Leibniz → Leibniz
```

The finite type associated to `Leibniz` has more constructors than `Leibniz`:

```
data FinB : Leibniz → Type where
  0/1      : FinB (1b n)
  0/2 1/2 : FinB (2b n)

  0-1b_ 1-1b_ : FinB n → FinB (1b n)
  0-2b_ 1-2b_ : FinB n → FinB (2b n)
```

Crucially, the constructors of the finite type at (e.g.) index `1b n` do not share the same structure.

In general, given a description of a number system \mathbb{N} , the number of constructors of the finite type `FinN` of \mathbb{N} depends directly on the interpretation of \mathbb{N} , preventing the construction `FinN` by simple recursion on `DescI` (that is, without passing around lists of constructors instead). Furthermore, since our definition of ornaments insists ornaments preserve the number of constructors, there cannot be an ornament from an arbitrary number system to its finite type.

²⁸Note that the foldable structure we get from the generic `fold` is significantly harder to work with for this purpose.

The apparent asymmetry between number systems and finite types stems from the definition of σ in `DescI`. In `DescI` and similar sums-of-products universes [EC22; Sij16], the remainder of a constructor `C` after a σ `S` simply has its context extended by `S`. In contrast, a universe with Σ -descriptions [eff20; KG16; McB14] (in the terminology of [Sij16]) encodes a dependent field ($s : S$) by asking for a function `C` assigning values `s` to descriptions.

Compared to Σ -descriptions, a sums-of-products universe keeps out some more exotic descriptions which do not have an obvious associated Agda datatype²⁹.

However, this also prevents us from writing down the simpler form of finite types. If we instead started from Σ -descriptions, taking functions into `DescI` to encode dependent fields, we could compute a “type of paths” in a number system by adding and deleting the appropriate fields. Consider the universe:

```
data  $\Sigma$ -Desc (I : Type) : Type where
  1 : I  $\rightarrow$   $\Sigma$ -Desc I
   $\rho$  : I  $\rightarrow$   $\Sigma$ -Desc I  $\rightarrow$   $\Sigma$ -Desc I
   $\sigma$  : (S : Type)  $\rightarrow$  (S  $\rightarrow$   $\Sigma$ -Desc I)  $\rightarrow$   $\Sigma$ -Desc I
```

In this universe we can present the binary numbers as:

```
Leibniz $\Sigma$ D :  $\Sigma$ -Desc  $\tau$ 
Leibniz $\Sigma$ D =  $\sigma$  (Fin 3)  $\lambda$ 
  { zero  $\rightarrow$  1 _
  ; (suc zero)  $\rightarrow$   $\rho$  _ (1 _)
  ; (suc (suc zero))  $\rightarrow$   $\rho$  _ (1 _) }
```

The finite type for these numbers can be described by:

```
FinB $\Sigma$ D :  $\Sigma$ -Desc Leibniz
FinB $\Sigma$ D =  $\sigma$  (Fin 3)  $\lambda$ 
  { zero  $\rightarrow$   $\sigma$  (Fin 0)  $\lambda$  _  $\rightarrow$  1 0b
  ; (suc zero)  $\rightarrow$   $\sigma$  Leibniz  $\lambda$  n  $\rightarrow$   $\sigma$  (Fin 2)  $\lambda$ 
    { zero  $\rightarrow$   $\sigma$  (Fin 1)  $\lambda$  _  $\rightarrow$  1 (1b n)
    ; (suc zero)  $\rightarrow$   $\sigma$  (Fin 2)  $\lambda$  _  $\rightarrow$   $\rho$  n (1 (1b n)) }
  ; (suc (suc zero))  $\rightarrow$   $\sigma$  Leibniz  $\lambda$  n  $\rightarrow$   $\sigma$  (Fin 2)  $\lambda$ 
    { zero  $\rightarrow$   $\sigma$  (Fin 2)  $\lambda$  _  $\rightarrow$  1 (2b n)
    ; (suc zero)  $\rightarrow$   $\sigma$  (Fin 2)  $\lambda$  _  $\rightarrow$   $\rho$  n (1 (2b n)) } }
```

Since this description of `FinB` largely has the same structure as `Leibniz`, and as a consequence also the numerical representation associated to `Leibniz`, this would simplify proving that the indexed numerical representation is indeed equivalent to the representable representation (the maps out of `FinB`). In a framework of ornaments for Σ -descriptions [KG16; McB14], we can even describe the finite type as an ornament on the number system.

6.2 Branching numerical representations

A numerical representation constructed by `TrieOD` looks like a finger tree: the structure typically has a central chain, which rather than directly storing ele-

²⁹Consider the constructor $\sigma \mathbb{N} \lambda n \rightarrow \text{power } \rho \ n \ 1$ which takes a number `n` and asks for `n` recursive fields (where `power f n x` applies `f n` times to `x`). This description, resembling a rose tree, does not (trivially) lie in a sums-of-products universe.

ments directly in nodes, stores the elements in trees of which the depth increases with the level of the node.

In contrast, structures like Braun trees, as Hinze and Swierstra [HS22] compute from binary numbers, represent the weight of a node by branching themselves. Because this kind of branching is uniform, i.e., each branch looks the same, we can still give an equivalent construction. By combining `TreeOD` and `TrieOD`, and using `to` to apply `p` k -fold in the case of `p {if = k}`, rather than over k -element vectors, we can replicate the structure of a Braun tree from `BinND`. However, if we use the Σ -descriptions we discussed above, we can more elegantly present these structures by adding an internal branch over `Fin k`.

6.3 Indices do not depend on parameters

In `DescI`, we represent the indices of a description as a single constant type, as opposed to an extension of the parameter telescope [EC22]. This simplification keeps the treatment of ornaments and numerical representations more to the point, but rules out types like the identity type `≡`.

By replacing index computing functions $\Gamma \& V \vdash I$ with dependent functions

```
_&_<_>_ : (Γ : Tel τ) (V I : ExTel Γ) → Type
Γ & V ⊢ I = (pv : [ Γ & V ] tel) → [ I ] tel (fst pv)
```

we can allow indices to depend on parameters in our framework. As a consequence, we have to modify nested recursive fields to ask for the index type `[I] tel` precomposed with `g : Cxf Γ Γ`, and we have to replace the square like `i ◦ j' ~ i' ◦ over v` in the definition of ornaments with heterogeneous squares.

Other consequence of allowing indices to depend on parameters is that this allows *algebraic ornaments* [McB14] to be formulated in `OrnDesc` in their fully general form, and also lets us describe *singleton types*, which can, roughly speaking, be used to compute the additional information needed to invert `ornForget`.

6.4 No RoseTrees

In `DescI`, we encode nested types by allowing nesting over a function of parameters `Cxf Γ Γ`. This is less expressive than full nested types, which may also nest a recursive field under a strictly positive functor. For example, rose trees

```
data RoseTree (A : Type) : Type where
  rose : A → List (RoseTree A) → RoseTree A
```

cannot be directly expressed as a `DescI`³⁰.

If we were to describe full nested types, allowing applications of functors in the types of recursive arguments, we would have to convince Agda that these functors are indeed positive, possibly by using polarity annotations³¹. Alternatively, we could encode strictly positive functors in a separate universe, which only allows using parameters in strictly positive contexts [Sij16]. Finally, we

³⁰And, since `DescI` does not allow for higher-order inductive arguments like Escot and Cockx [EC22], we can also not give an essentially equivalent definition.

³¹See <https://github.com/agda/agda/pull/6385>.

could modify `DescI` in such a way that we can decide if a description uses a parameter strictly positively, for which we would modify ρ and σ , or add variants of ρ and σ restricted to strictly positive usage of parameters.

6.5 No levitation

Since our encoding does not support higher-order inductive arguments, let alone definitions by induction-recursion, there is no code for `DescI` in itself. Such self-describing universes have been described by Chapman et al. [Cha+10], and we expect that the additional features of `DescI`, i.e., parameters, nesting, and composition, would not obstruct a similar levitating variant of `DescI`. Due to the work of Dagand and McBride [DM14], ornaments might even be generalized to inductive-recursive descriptions.

If that is the case, then modifications of universes like `Meta` could be expressed internally. In particular, rather than defining `DescI` such that it can describe datatypes with the information of, e.g., number systems, `DescI` should be expressible as an ornamental description on `Desc`, in contrast to how `Desc` is an instance of `DescI` in our framework. This would allow treating information explicitly in `DescI`, and not at all in `Desc`.

Furthermore, constructions like `TrieOD`, which have the recursive structure of a `fold` over `DescI`, could be expressed by instantiating `fold` to `DescI`.

6.6 Metadata more tasteful externally than internally

On the other hand, while incorporating general metadata into `DescI` works out neatly in our case, and in general seems to work out if we think about one use-case at a time, it might not work so nicely in other situations. For example, if we are working with `Number`, but we are given a `DescI Plain` (i.e., `Desc`), then we would have to duplicate that description in `DescI Number` before we could use it. Even worse, if we want to give the constructors of a number system nice names using `Names`, we would have to rewrite our code and descriptions to use something like the product of `Number` and `Names`.

It might be more portable to take the same approach in handling metadata as True sums of products [VL14], where metadata is described externally to the universe and only combined again if needed by a generic function. From this point of view, a type of metadata can simply be a convenient function from `Desc` to `Type`, and if `Number` was presented in this way, then `TreeOD` would not have to ask for `DescI Number`, but rather for a `D` of `Desc` with `Number D`.

6.7 δ is conservative

We define our universe `DescI` with δ as a former of fields with known descriptions, and this makes it easier to write down `TreeOD`, even though δ is redundant. If more concise universes and ornaments are preferable, we can actually get all the features of δ and ornaments like $\bullet\delta$ by describing them using σ , annotations, and other ornaments.

Indeed, rather than using δ to add a field from a description R , we can simply use σ to add $S = \mu R$, and remember that S came from R in the information:

```
Delta : Meta
Delta . $\sigma$ i { $\Gamma = \Gamma$ } { $V = V$ } S
= Maybe (
   $\Sigma$ [  $\Delta \in \text{Tel } \tau$  ]  $\Sigma$ [  $J \in \text{Type}$  ]  $\Sigma$ [  $j \in \Gamma \ \& \ V \vdash J$  ]
   $\Sigma$ [  $g \in \Gamma \ \& \ V \vdash [\Delta] \text{tel } tt$  ]  $\Sigma$ [  $D \in \text{DescI } \text{Delta } \Delta J$  ]
  ( $\forall \text{pv} \rightarrow S \text{pv} \equiv \text{liftM2 } (\mu D) \ g \ j \text{pv}$ ))
```

We can then define δ as a pattern synonym matching on the `just` case, and σ matching on the `nothing` case.

Recall that, leaving out some details, the ornament $\bullet\delta$ lets us compose an ornament from D to D' with an ornament from R to R' , yielding an ornament from $\delta D R$ to $\delta D' R'$. This ornament can equivalently be modelled by first adding a new field $\mu R'$, and then deleting the original μR field. The ornament ∇ [Ko14] allows one to provide a default value for a field, deleting it from the description. Hence, we can model $\bullet\delta$ by binding a value r' of $\mu R'$ with $\text{O}\Delta\sigma+$ and deleting the field μR using a default value computed by `ornForget`.

This also partially explains why we did not refer to algebraic ornaments at all in our construction of `TrieOD`; We can see that `TrieOD` looks very similar to the algebraic ornament over `TreeOD`, which sends ornaments from D to E to an ornament to a D -indexed variant of E . However, the case of δ requires `TrieOD` to step in and re-index the subdescription. In contrast, the algebraic ornament would simply treat a δ like its equivalent σ , and even though this would produce a correct numerical representation, this amounts to presenting a `Vec` as a tuple of a length n , a `List` x s, and a proof that n is equal to `length xs`.

Thus, while it would be possible to present `TrieOD` as a kind of algebraic ornament, this would require redefining algebraic ornaments from algebras that are rather specific about how they treat a σ .

6.8 No sparse numerical representations

The encoding of number systems in a universe we explained in Section 3.2.1 corresponds to a generalization of dense number systems. Consequently, this excludes the skew binary numbers [Oka95] in their useful sparse representation.

Representations of sparse number systems can be regained by allowing addition *and* variable multiplication in a σ . In such a setup, skew binary numbers and other sparse representations could be described by adding their gaps as fields, and computing the appropriate multiplier from there. While not worked out in this thesis, this extension is compatible with the construction of numerical representations.

Another notable extension of `Number` is to let some recursive and composite fields be interpreted by multiplication, with which we could equip `U-fin` with its obvious interpretation into `N`. This can be compared to the last exponential law we did not use in Section 3.1, which is that $A^{BC} = (A^B)^C$. Furthermore, any indexed numerical representation acts as a representable functor F . If F and

G are numerical representations corresponding to number systems N and M , then the multiplication of N and M just corresponds to composition $F \circ G$.

References

- [AMM07] Thorsten Altenkirch, Conor McBride, and Peter Morris. “Generic Programming with Dependent Types”. In: Nov. 2007, pp. 209–257. ISBN: 978-3-540-76785-5. DOI: 10.1007/978-3-540-76786-2_4.
- [Bru91] N.G. de Bruijn. “Telescopic mappings in typed lambda calculus”. In: *Information and Computation* 91.2 (1991), pp. 189–204. ISSN: 0890-5401. DOI: [https://doi.org/10.1016/0890-5401\(91\)90066-B](https://doi.org/10.1016/0890-5401(91)90066-B). URL: <https://www.sciencedirect.com/science/article/pii/089054019190066B>.
- [Cha+10] James Chapman et al. “The Gentle Art of Levitation”. In: *Proceedings of the 15th ACM SIGPLAN International Conference on Functional Programming*. ICFP ’10. Baltimore, Maryland, USA: Association for Computing Machinery, 2010, pp. 3–14. ISBN: 9781605587943. DOI: 10.1145/1863543.1863547. URL: <https://doi.org/10.1145/1863543.1863547>.
- [Coc+22] Jesper Cockx et al. “Reasonable Agda is Correct Haskell: Writing Verified Haskell Using Agda2hs”. In: *Proceedings of the 15th ACM SIGPLAN International Haskell Symposium*. Haskell 2022. Ljubljana, Slovenia: Association for Computing Machinery, 2022, pp. 108–122. ISBN: 9781450394383. DOI: 10.1145/3546189.3549920. URL: <https://doi.org/10.1145/3546189.3549920>.
- [DM14] Pierre-Évariste Dagand and Conor McBride. “Transporting functions across ornaments”. In: *Journal of Functional Programming* 24.2-3 (Apr. 2014), pp. 316–383. DOI: 10.1017/s0956796814000069. URL: <https://doi.org/10.1017/s0956796814000069>.
- [DS06] Peter Dybjer and Anton Setzer. “Indexed induction–recursion”. In: *The Journal of Logic and Algebraic Programming* 66.1 (2006), pp. 1–49. ISSN: 1567-8326. DOI: <https://doi.org/10.1016/j.jlap.2005.07.001>. URL: <https://www.sciencedirect.com/science/article/pii/S1567832605000536>.
- [EC22] Lucas Escot and Jesper Cockx. “Practical Generic Programming over a Universe of Native Datatypes”. In: *Proc. ACM Program. Lang.* 6.ICFP (Aug. 2022). DOI: 10.1145/3547644. URL: <https://doi.org/10.1145/3547644>.
- [eff20] effectfully. *Generic*. 2020. URL: <https://github.com/effectfully/Generic>.
- [HP06] Ralf Hinze and Ross Paterson. “Finger trees: a simple general-purpose data structure”. In: *Journal of Functional Programming* 16.2 (2006), pp. 197–217. DOI: 10.1017/S0956796805005769.

- [HS22] Ralf Hinze and Wouter Swierstra. “Calculating Datastructures”. In: *Mathematics of Program Construction*. Ed. by Ekaterina Komen-dantskaya. Cham: Springer International Publishing, 2022, pp. 62–101. ISBN: 978-3-031-16912-0.
- [JG07] Patricia Johann and Neil Ghani. “Initial Algebra Semantics Is Enough!”. In: *Typed Lambda Calculi and Applications*. Ed. by Simona Ronchi Della Rocca. Berlin, Heidelberg: Springer Berlin Heidelberg, 2007, pp. 207–222. ISBN: 978-3-540-73228-0.
- [KG16] Hsiang-Shang Ko and Jeremy Gibbons. “Programming with orna-ments”. In: *Journal of Functional Programming* 27 (2016), e2. DOI: 10.1017/S0956796816000307.
- [Ko14] H Ko. “Analysis and synthesis of inductive families”. PhD thesis. Oxford University, UK, 2014.
- [Mag+10] José Pedro Magalhães et al. “A Generic Deriving Mechanism for Haskell”. In: *SIGPLAN Not.* 45.11 (Sept. 2010), pp. 37–48. ISSN: 0362-1340. DOI: 10.1145/2088456.1863529. URL: <https://doi.org/10.1145/2088456.1863529>.
- [Mar84] Per Martin-Löf. *Intuitionistic Type Theory*. Bibliopolis, 1984.
- [McB14] Conor McBride. “Ornamental Algebras, Algebraic Ornaments”. In: 2014.
- [Nor09] Ulf Norell. “Dependently Typed Programming in Agda”. In: *Ad-vanced Functional Programming: 6th International School, AFP 2008, Heijen, The Netherlands, May 2008, Revised Lectures*. Ed. by Pieter Koopman, Rinus Plasmeijer, and Doaitse Swierstra. Berlin, Heidelberg: Springer Berlin Heidelberg, 2009, pp. 230–266. ISBN: 978-3-642-04652-0. DOI: 10.1007/978-3-642-04652-0_5. URL: https://doi.org/10.1007/978-3-642-04652-0_5.
- [Oka95] Chris Okasaki. “Purely Functional Random-Access Lists”. In: *Pro-ceedings of the Seventh International Conference on Functional Pro-gramming Languages and Computer Architecture*. FPCA ’95. La Jolla, California, USA: Association for Computing Machinery, 1995, pp. 86–95. ISBN: 0897917197. DOI: 10.1145/224164.224187. URL: <https://doi.org/10.1145/224164.224187>.
- [Oka98] Chris Okasaki. *Purely Functional Data Structures*. USA: Cambridge University Press, 1998. ISBN: 0521631246.
- [Rey83] John C Reynolds. “Types, abstraction and parametric polymor-phism”. In: *Information Processing 83, Proceedings of the IFIP 9th World Computer Congres.* 1983, pp. 513–523.
- [Sij16] Yorick Sijsling. “Generic programming with ornaments and depen-dent types”. In: *Master’s thesis* (2016).
- [Tea23] Agda Development Team. *Agda*. 2023. URL: <https://agda.readthedocs.io/en/v2.6.3/>.

- [The23] The Agda Community. *Agda Standard Library*. Version 1.7.2. Feb. 2023. URL: <https://github.com/agda/agda-stdlib>.
- [VL14] Edsko de Vries and Andres Löh. “True Sums of Products”. In: *Proceedings of the 10th ACM SIGPLAN Workshop on Generic Programming*. WGP ’14. Gothenburg, Sweden: Association for Computing Machinery, 2014, pp. 83–94. ISBN: 9781450330428. DOI: 10.1145/2633628.2633634. URL: <https://doi.org/10.1145/2633628.2633634>.
- [Wad89] Philip Wadler. “Theorems for Free!” In: *Proceedings of the Fourth International Conference on Functional Programming Languages and Computer Architecture*. FPCA ’89. Imperial College, London, United Kingdom: Association for Computing Machinery, 1989, pp. 347–359. ISBN: 0897913280. DOI: 10.1145/99370.99404. URL: <https://doi.org/10.1145/99370.99404>.
- [WKS22] Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. Aug. 2022. URL: <https://plfa.inf.ed.ac.uk/22.08/>.

Appendices

A Folding

In Section 2.4.6 and Section 2.5 we used `fold` as a concept to explain a bit of generic programming. We give its definition here, but for `DescI` instead, since the fold of `U-ix` can be seen as a simplification of it.

`fold` : $\forall \{D : \text{DescI Me } \Gamma \text{ I}\} \{X\} \rightarrow \llbracket D \rrbracket D X \rightarrow_3 X \rightarrow \mu D \rightarrow_3 X$

As `fold f` is the algebra map `con` \Rightarrow `f`, the following commutes:

$$\begin{array}{ccc} F \mu F & \xrightarrow{F(\text{fold } f)} & F A \\ \text{con} \downarrow & & \downarrow f \\ \mu F & \xrightarrow{\text{fold } f} & A \end{array}$$

However, by defining `fold f (con x)` as `f (map (fold f) x)`, we prevent the termination checker from seeing that `fold` is only applied to terms strictly smaller than `x` (much like our fellow universe constructions find out somewhere along the line). To help out the termination checker, we inline `fold` into `map`, which gives us an equivalent definition:

```
mapDesc :  $\forall \{D' : \text{DescI Me } \Gamma \text{ I}\} (D : \text{DescI Me } \Gamma \text{ I}) \{X\}$ 
   $\rightarrow \forall p i \rightarrow \llbracket D' \rrbracket D X \rightarrow_3 X$ 
   $\rightarrow \llbracket D \rrbracket D (\mu D') p i \rightarrow \llbracket D \rrbracket D X p i$ 
```

```
mapCon :  $\forall \{D' : \text{DescI Me } \Gamma \text{ I}\} \{V\} (C : \text{ConI Me } \Gamma \text{ V I}) \{X\}$ 
   $\rightarrow \forall p i v \rightarrow \llbracket D' \rrbracket D X \rightarrow_3 X$ 
   $\rightarrow \llbracket C \rrbracket C (\mu D') (p, v) i \rightarrow \llbracket C \rrbracket C X (p, v) i$ 
```

```
fold f p i (con x) = f p i (mapDesc _ p i f x)
```

```
mapDesc (C :: D) p i f (inj1 x) = inj1 (mapCon C p i tt f x)
mapDesc (C :: D) p i f (inj2 y) = inj2 (mapDesc D p i f y)
```

```
mapCon (1 j)      p i v f x      = x
mapCon (p g j C)  p i v f (r, x) = fold f (g p) (j (p, v)) r
                                     , mapCon C p i v f x
mapCon (σ S w C)  p i v f (s, x) = s , mapCon C p i (w (v, s)) f x
mapCon (δ d j R C) p i v f (r, x) = r , mapCon C p i v f x
```

Here `mapDesc` (and `mapCon`) simply peel off and reassemble all non-recursive structure, applying `fold` to the recursive fields; `fold` is then defined in the usual way by applying its algebra `f` to itself mapped over `x`.

B Folding without Axiom K

The axiom of univalence (or cubical type theory) gives us another interesting context to study ornaments in. In the way we presented it, the theory of ornaments produces a lot of isomorphisms from relations between types, which

are not yet as powerful as they could be when comparing properties between related types. Univalence gives us the means to turn equivalences³² into equalities, allowing us to put an isomorphism between types to work by transporting properties over it.

Unfortunately, a direct port of ornaments into `--cubical` is quickly thwarted by the absence of Axiom K, as one would discover that the definitions of `mapDesc` and `mapCon` illegally pattern match on the types calculated by interpretations³³.

This can be remedied by presenting interpretations as datatypes³⁴. Effectively, we are applying the duality between type computing functions and indexed types. Since `Desc` and `Con` are unindexed types, they cannot accidentally carry equational content, and pattern matching on them does not generate transports in `[_]D` and `[_]C`. Hence, the definition of `fold` is (morally speaking) safe.

With that out of the way, we can define the interpretations as indexed types:

```
mutual
data μ (D : Desc Γ I) (p : [Γ]tel tt) : I → Type where
  con : ∀ {i} → IntpD (μ D) p i D → μ D p i

data IntpC (X : [Γ]tel tt → I → Type)
  (pv : [Γ & V]tel) (i : I)
  : Con Γ V I → Type where
  1-i : ∀ {i'}
    → i ≡ i' pv → IntpC X pv i (1 i')

  ρ-i : ∀ {g i' D}
    → X (g (pv .fst)) (i' pv) → IntpC X pv i D
    → IntpC X pv i (ρ g i' D)

  σ-i : ∀ {S D} {w : Vxf id (V ▷ S) W}
    → (s : S pv) → IntpC X (pv .fst , w (pv .snd , s)) i D
    → IntpC X pv i (σ S w D)

  δ-i : ∀ {d j D} {R : Desc Δ J}
    → (s : μ R (d pv) (j pv)) → IntpC X pv i D
    → IntpC X pv i (δ d j R D)

data IntpD (X : [Γ]tel tt → I → Type)
  (p : [Γ]tel tt) (i : I)
  : Desc Γ I → Type where
  ::-il : ∀ {C D} → IntpC X (p , tt) i C → IntpD X p i (C :: D)
```

³²Equivalences can be considered as a correction to isomorphisms for types which are not sets (in the sense of being discrete); since all types we describe here are sets, equivalences and isomorphisms coincide.

³³The Without K documentation explains why pattern matching on non-datatypes is not safe in general.

³⁴Albeit a bit dubiously; at the time of writing, this is also how you can circumvent a restriction on pattern matching emplaced by `--cubical-compatible`, see the relevant GitHub issue.


```

--ir : ∀ {C D} → IntpD X p          i D → IntpD X p i (C :: D)

[_]D : Desc Γ I → ([ Γ ]tel tt → I → Type) → [ Γ ]tel tt → I → Type
[_]D = λ D X p i → IntpD X p i D

[_]C : Con Γ V I → ([ Γ ]tel tt → I → Type) → [ Γ & V ]tel → I → Type
[_]C = λ C X pv i → IntpC X pv i C

```

Since the interpretations are datatypes now, we can pattern match on them to define `mapDesc` and `mapCon` in a way that is accepted:

```

mapDesc : ∀ {D' : Desc Γ I} (D : Desc Γ I) {X}
          → ∀ p i → [ D' ]D X →3 X → [ D ]D (μ D') p i → [ D ]D X p i

mapCon : ∀ {D' : Desc Γ I} {V} (C : Con Γ V I) {X}
          → ∀ p i v → [ D' ]D X →3 X → [ C ]C (μ D') (p , v) i → [ C ]C X (p , v) i

fold f p i (con x) = f p i (mapDesc _ p i f x)

mapDesc (C :: D) p i f (con il x) = con il (mapCon C p i tt f x)
mapDesc (C :: D) p i f (con ir y) = con ir (mapDesc D p i f y)

mapCon (1 j)      p i v f
  (1-i x) = 1-i x

mapCon (ρ g j C)  p i v f
  (ρ-i r x) = ρ-i (fold f (g p) (j (p , v)) r) (mapCon C p i v f x)

mapCon (σ S w C)  p i v f
  (σ-i s x) = σ-i s (mapCon C p i (w (v , s)) f x)

mapCon (δ d j R C) p i v f
  (δ-i r x) = δ-i r (mapCon C p i v f x)

```

C Nested types as uniformly recursive indexed types

Although `U-ix` has no direct support for expressing nested types, we can actually give equivalent encodings for some of them.

Indeed, indices are readily repurposed as parameters. If we apply this to, say, random-access lists, we can write:

```

RandomD-1 : U-ix ∅ Type
RandomD-1 = σ (λ _ → Type)
  ( 1 λ { ( _ , ( _ , A ) ) → A } )
  :: σ (λ _ → Type)
  ( σ (λ { ( _ , ( _ , A ) ) → A } )
    ( ρ (λ { ( _ , (( _ , A ) , _ ) ) → A × A } )
      ( 1 λ { ( _ , (( _ , A ) , _ ) ) → A } ) ) )
  :: σ (λ _ → Type)
  ( σ (λ { ( _ , ( _ , A ) ) → A × A } )

```

```

( ρ (λ { (− , ((− , A) , −)) → A × A } )
( 1 λ { (− , ((− , A) , −)) → A } )))
:: []

```

More interestingly, perhaps, the depth of a random-access list determines the types of its fields. Namely, at the highest level, `One` will ask for 1 element, one level down it asks for 2, and one more it asks for 4, and so on. Hence, in a way that vaguely resembles defunctionalization, we can define

```

power : ℕ → (A → A) → A → A
power zero f x = x
power (suc n) f x = f (power n f x)

```

```

data Pair (A : Type) : Type where
  pair : A → A → Pair A

```

and describe a field at depth n by `power n Pair A`. This can be applied to describe random-access lists which track their depth in their index instead:

```

RandomD-2 : U-ix (∅ ▷ const Type) ℕ
RandomD-2 = σ (λ _ → ℕ)
  ( 1 λ { (− , (− , n)) → n } )
  :: σ (λ _ → ℕ)
  ( σ (λ { ((− , A) , (− , n)) → power n Pair A } )
  ( ρ (λ { (− , ((− , n) , −)) → suc n } )
  ( 1 λ { (− , ((− , n) , −)) → n } ) ) )
  :: σ (λ _ → ℕ)
  ( σ (λ { ((− , A) , (− , n)) → power (suc n) Pair A } )
  ( ρ (λ { (− , ((− , n) , −)) → suc n } )
  ( 1 λ { (− , ((− , n) , −)) → n } ) ) )
  :: []

```

Since we cannot (yet) construct path types generically (Section 6.1), we cannot make this construction generic. If we did have such constructions, the argument for random-access lists generalizes to an operation that splits a nested datatype D into three parts:

1. a type of paths in D (not necessarily pointing to a field)
2. a lookup function that sends a path to the accumulated parameter transformation
3. the (uniform) datatype, indexed over the paths, using the lookup function to calculate the types of its fields.