

# Restoring (part of) the friendship between recursion schemes and without-K (provisional)

I'll have to grab a UU-template at some point

Samuel Klumpers  
6057314

March 14, 2023

## Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>How Cubical Agda helps our binary numbers (ready)</b>	<b>3</b>
2.1	Binary numbers . . . . .	4
2.2	Use as definition: functions from specifications . . . . .	5
2.3	Structure Identity Principle . . . . .	7
<b>3</b>	<b>Specifying types (ready)</b>	<b>8</b>
3.1	Numerical representations: from numbers to containers . . . . .	8
3.2	Numerical representations as ornaments . . . . .	10
3.3	Folding in . . . . .	12
<b>4</b>	<b>Reducing friction (work in progress)</b>	<b>13</b>
4.1	Well-founded monic algebras are initial . . . . .	13
4.1.1	Algebras of endofunctors . . . . .	14
4.1.2	Accessibility . . . . .	16
4.1.3	Proof sketch of 4.1 . . . . .	17
4.1.4	Example . . . . .	17
<b>5</b>	<b>Is equivalence too strong (finger trees)</b>	<b>17</b>
<b>6</b>	<b>Discussion and future work (aka the union of my to-do list and the actual future work section)</b>	<b>17</b>
<b>7</b>	<b>Temporary</b>	<b>18</b>
	This document is generated from a literate agda file!	

## Abstract

The preliminary goal of this thesis is to introduce, among others, the concepts of the structure identity principle, numerical representations, and ornamentations, which are then combined to simplify the presentation and verification of finger trees, as a demonstration of the generalizability and improved compactness and security of the resulting code.

## 1 Introduction

Most of the time when we are Agda-ing [Tea23] we are trying to un-Haskell ourselves, e.g., not take the head of an empty list. In this example, we can make `head` total by switching to length-indexed lists: vectors. We have now effectively doubled the size of our code base, since functions like `_++_` which we had for lists, will also have to be reimplemented for vectors.

To make things worse; often, after coping with the overloaded names resulting from Agda-ing by shoving them into a different namespace, we also find out that lists nor vectors are efficient containers to begin with. Maybe binary trees are better. We now need four times the number of definitions to keep everything working, and, if we start proving things, we will also have to prove everything fourfold. (Not to mention that reasoning about trees is probably going to be harder than reasoning about lists). This inefficiency has sparked (my) interest in ways to deal with the situation.

Following [DM14] and [KG16], we can describe the relation between list and vector using the mechanism of ornamentation. This leads them to define the concept of patches, which can aid us when defining `_++_` for the second time by forcing the new version to be coherent. In fact, the algebraic nature of ornaments can even get us the definition of the vector type for free, if we started by defining lists relative to natural numbers [McB14]. Such constructions rely heavily on descriptions of datastructures and often come with limitations in their expressiveness. These descriptions in turn impose additional ballast on the programmer, leading us to investigate reflection like in [EC22] as a means to bring datatypes and descriptions closer when possible.

From a different direction, [HS22] gives methods by which we can show two implementations of some structure to be equivalent. With this, we can simply transport all proofs about `_++_` we have for lists over to the implementation for trees, provided that we show them to be equivalent as appendable containers. This process can also be automated by some heavy generics, but instead, we resort to cubical; which hosts a range of research like [Ang+20] tailored to the problem describing equivalences of structures.

We can liken the situation to movement on a plane, where ornamentation moves us vertically by modifying constructors or indices, and structured equivalences move us horizontally to and from equivalent but more equivalent implementations. In this paper, we will investigate a variety of means of moving around structures and proofs, and ways to make this more efficient or less intrusive.

Currently, all sections mainly reintroduce or reformulate existing research, with some spots of new ideas and original examples here and there. In section 2, we will look at how proofs on unary naturals can be moved to binary naturals. Then in section 3 we recall how numeral systems in particular induce container types, which we attempt to reformulate in the language of ornaments in subsection 3.2.

## 2 How Cubical Agda helps our binary numbers (ready)

Let us quickly review the small set of features in Cubical Agda that we will be using extensively throughout this article.<sup>1</sup> We note that there are some downsides to cubical, such as that

`{-# OPTIONS --cubical #-}`

also implies the negation of axiom K, which in turn complicates both some termination checking and some universe levels. And, more obviously, we get saddled with the proof obligation that our types are sets should we use certain constructions.

Of course, this downside is more than offset by the benefits of changing our primitive notion of equality, which among other things, lets us access univalence, which drastically cut down the investment required to both show more complex structures to be equivalent (at least when compared to non-cubical). Here, equality arises not (directly) from the indexed inductive definition we are used to, but rather from the presence of the interval type `I`. This type represents a set of two points `i0` and `i1`, which are considered “identified” in the sense that they are connected by a path. To define a function out of this type, we also have to define the function on all the intermediate points, which is why we call such a function a “path”. Terms of other types are then considered identified when there is a path between them.

As an added benefit, this different perspective gives intuitive interpretations to some proofs of equality, like

`sym : x ≡ y → y ≡ x`

`sym p i = p (~ i)`

where `~_` is the interval reversal, swapping `i0` and `i1`, so that `sym` simply reverses the given path.

Furthermore, because we can now interpret paths in records and function differently, we get a host of “extensionality” for free. For example, a path in  $A \rightarrow B$  is indeed a function which takes each  $i$  in `I` to a function  $A \rightarrow B$ . Using this, function extensionality becomes tautological

`funExt : (∀ x → f x ≡ g x) → f ≡ g`

`funExt p i x = p x i`

Finally, the `Glue` type tells us that equivalent types fit together in a new type, in a way that guarantees univalence

Not sure if it would be helpful to have a more extensive introduction covering all features used.

To be precise, the `mapFold` in [KG16] gets painted red.

<sup>1</sup>[VMA19] gives a comprehensive introduction to cubical agda.

$ua : \forall \{A B : \text{Type } \ell\} \rightarrow A \simeq B \rightarrow A = B$

For our purposes, we can interpret univalence as “equivalent types are identified”, and, we can treat equivalences as the “HoTT-compatible” generalization of bijections. In particular, type isomorphisms like  $1 \rightarrow A \simeq A$  actually become paths  $1 \rightarrow A \equiv A$ , such that we can transport proofs along them. We will demonstrate this by a slightly more practical example.

## 2.1 Binary numbers

Let us motivate the cubical method by showing the equivalence of the “Peano” naturals and the “Leibniz” naturals. Recall that the Peano naturals are defined as

```
data N : Type where
  zero : N
  suc : N → N
```

This definition enjoys a simple induction principle and has many proofs of its properties in standard libraries. However, it is too slow to be of practical use: most arithmetic operations defined on `N` have time complexity in the order of the value of the result.

Of course, the alternative are the more performant binary numbers: the time complexities for binary numbers are usually logarithmic in the resultant values, but these are typically less well-covered in terms of proofs. This does not have to be a problem, because the `N` naturals and the binary numbers should be equivalent after all!

Let us make this formal. We define the Leibniz naturals as follows:

```
data Leibniz : Set where
  0b : Leibniz
  _1b : Leibniz → Leibniz
  _2b : Leibniz → Leibniz
```

Here, the `0b` constructor encodes 0, while the `_1b` and `_2b` constructors respectively add a 1 and a 2 bit, under the usual interpretation of binary numbers:

```
toN : Leibniz → N
toN 0b = 0
toN (n 1b) = 1 N.+ 2 N.· toN n
toN (n 2b) = 2 N.+ 2 N.· toN n
```

Let us construct the equivalence from `N` to `Leibniz`. First, we can also interpret a number in `N` as a binary number by repeating the successor operation on binary numbers:

```
bsuc : Leibniz → Leibniz
bsuc 0b = 0b 1b
bsuc (n 1b) = n 2b
bsuc (n 2b) = (bsuc n) 1b
```

```
fromN : N → Leibniz
fromN N.zero = 0b
```

is this too much code and too little explanation at once?

$$\text{fromN } (\mathbb{N}.\text{succ } n) = \text{bsuc } (\text{fromN } n)$$

To show that the operations are inverses, we observe that the interpretation respects successors

$$\begin{aligned} \text{toN-suc} &: \forall x \rightarrow \text{toN } (\text{bsuc } x) = \mathbb{N}.\text{succ } (\text{toN } x) \\ \text{toN-suc } 0b &= \text{refl} \\ \text{toN-suc } (x \ 1b) &= \text{refl} \\ \text{toN-suc } (x \ 2b) &= \text{cong} \\ &(\lambda k \rightarrow (1 \ \mathbb{N} + 2 \ \mathbb{N} \cdot k)) \\ &(\text{toN-suc } x) \cdot \text{cong } \mathbb{N}.\text{succ } (\text{NP} \cdot \text{succ } 2 \ (\text{toN } x)) \end{aligned}$$

and that the inverse respects even and odd numbers

$$\begin{aligned} \text{fromN-1+2} &: \forall x \rightarrow \text{fromN } (1 \ \mathbb{N} + 2 \ \mathbb{N} \cdot x) = (\text{fromN } x) \ 1b \\ \text{fromN-1+2} \ \mathbb{N}.\text{zero} &= \text{refl} \\ \text{fromN-1+2} \ (\mathbb{N}.\text{succ } x) &= \text{cong} \\ &(\text{bsuc} \circ \text{bsuc}) \\ &(\text{cong fromN } (\text{NP} \cdot \text{succ } x \ (x \ \mathbb{N} + \mathbb{N}.\text{zero})) \cdot \text{fromN-1+2} \ x) \end{aligned}$$

$$\begin{aligned} \text{fromN-2+2} &: \forall x \rightarrow \text{fromN } (2 \ \mathbb{N} + 2 \ \mathbb{N} \cdot x) = (\text{fromN } x) \ 2b \\ \text{fromN-2+2} \ x &= \text{cong bsuc } (\text{fromN-1+2} \ x) \end{aligned}$$

The wanted statement follows

$$\begin{aligned} \mathbb{N} \leftrightarrow \mathbb{L} &: \text{Iso } \mathbb{N} \ \text{Leibniz} \\ \mathbb{N} \leftrightarrow \mathbb{L} &= \text{iso fromN toN sec ret} \end{aligned}$$

where

$$\begin{aligned} \text{sec} &: \text{section fromN toN} \\ \text{sec } 0b &= \text{refl} \\ \text{sec } (n \ 1b) &= \text{fromN-1+2} \ (\text{toN } n) \cdot \text{cong } \_1b \ (\text{sec } n) \\ \text{sec } (n \ 2b) &= \text{fromN-2+2} \ (\text{toN } n) \cdot \text{cong } \_2b \ (\text{sec } n) \end{aligned}$$

$$\begin{aligned} \text{ret} &: \text{retract fromN toN} \\ \text{ret } \mathbb{N}.\text{zero} &= \text{refl} \\ \text{ret } (\mathbb{N}.\text{succ } n) &= \text{toN-suc } (\text{fromN } n) \cdot \text{cong } \mathbb{N}.\text{succ } (\text{ret } n) \end{aligned}$$

but since we now have a bijection, we also get an equivalence

$$\begin{aligned} \mathbb{N} \simeq \mathbb{L} &: \mathbb{N} \simeq \text{Leibniz} \\ \mathbb{N} \simeq \mathbb{L} &= \text{isoToEquiv } \mathbb{N} \leftrightarrow \mathbb{L} \end{aligned}$$

Finally, by univalence, we can identify  $\mathbb{N}$  and  $\text{Leibniz}$  naturals

$$\begin{aligned} \mathbb{N} = \mathbb{L} &: \mathbb{N} = \text{Leibniz} \\ \mathbb{N} = \mathbb{L} &= \text{ua } \mathbb{N} \simeq \mathbb{L} \end{aligned}$$

Using the path  $\mathbb{N} = \mathbb{L}$  we can already prove some otherwise difficult properties, e.g.,

$$\begin{aligned} \text{isSetL} &: \text{isSet Leibniz} \\ \text{isSetL} &= \text{subst isSet } \mathbb{N} = \mathbb{L} \ \mathbb{N}.\text{isSetN} \end{aligned}$$

Let us define an operation on  $\text{Leibniz}$  and demonstrate how we can also transport proofs about operations from  $\mathbb{N}$  to  $\text{Leibniz}$ .

## 2.2 Use as definition: functions from specifications

As an example, we will define addition of binary numbers. We could take

```

BinOp : Type → Type
BinOp A = A → A → A

```

```

_+_ : BinOp Leibniz
_+_ = subst BinOp N=L N._+_

```

But this would be rather inefficient, incurring an  $O(n + m)$  overhead when adding  $n + m$ , so we could better define addition directly. We would prefer to give a definition which makes use of the binary nature of `Leibniz`, while agreeing with the addition on `N`.

Such a definition can be derived from the specification “agrees with `_+_`”, so we implement the following syntax for giving definitions by equational reasoning, inspired by the “use-as-definition” notation from [HS22]:

```

Def : {X : Type a} → X → Type a
Def {X = X} x = Σ' X λ y → x ≡ y

```

```

defined-by : {X : Type a} {x : X} → Def x → X
defined-by = fst

```

```

by-definition : {X : Type a} {x : X} → (d : Def x) → x ≡ defined-by d
by-definition = snd

```

which infers the definition from the right endpoint of a path using an implicit pair type

```

record Σ' (A : Set a) (B : A → Set b) : Set (ℓ-max a b) where
  constructor _use-as-def
  field
    {fst} : A
    snd : B fst

open Σ'

infix 1 _use-as-def

```

*As of now, I am unsure if this reduces the effort of implementing a coherent function, or whether it is more typically possible to give a smarter or shorter proof by just giving a definition and proving an easier property of it<sup>2</sup>*

With this we can define addition on `Leibniz` and show it agrees with addition on `N` in one motion

```

{-# TERMINATING #-}
plus-def : ∀ x y → Def (fromN (toN x N.+ toN y))
plus-def 0b y = N↔L .rightInv y use-as-def
plus-def (x 1b) 0b =
  bsuc (fromN (toN x N.+ (toN x N.+ N.zero) N.+ N.zero))
  ≡< cong (bsuc ∘ fromN) (NP.+zero (2 N. toN x)) >
  bsuc (fromN (toN x N.+ (toN x N.+ N.zero)))

```

---

<sup>2</sup>I will put the alternative in the appendix for now

```

    =< fromN-1+2· (toN x) >
fromN (toN x) 1b
    =< cong _1b (N↔L .rightInv x) >
x 1b ■ use-as-def
plus-def (x 1b) (y 1b) =
  fromN ((1 N.+ 2 N.· toN x) N.+ (1 N.+ 2 N.· toN y))
    =< cong fromN (Eq.eqToPath (eq (toN x) (toN y))) >
  fromN (2 N.+ (2 N.· (toN x N.+ toN y)))
    =< fromN-2+2· (toN x N.+ toN y) >
  fromN (toN x N.+ toN y) 2b
    =< cong _2b (by-definition (plus-def x y)) >
  defined-by (plus-def x y) 2b ■ use-as-def
  where
    eq : ∀ x y
      → (1 N.+ 2 N.· x) N.+ (1 N.+ 2 N.· y) Eq.= 2 N.+ (2 N.· (x N.+ y))
    eq = NS.solve-∀
-- similar clauses omitted

plus : ∀ x y → Leibniz
plus x y = defined-by (plus-def x y)

plus-coherent : ∀ x y → fromN (x N.+ y) = plus (fromN x) (fromN y)
plus-coherent x y = cong fromN
  (cong₂ N._+ (sym (N↔L .leftInv x)) (sym (N↔L .leftInv _))) ·
  by-definition (plus-def (fromN x) (fromN y))

```

## 2.3 Structure Identity Principle

We see that as a consequence (modulo some [PathP](#) lemmas), we get a path from  $(\mathbb{N}, N.+)$  to  $(\text{Leibniz}, \text{plus})$ . More generally, we can view a type  $X$  combined with a function  $f : X \rightarrow X \rightarrow X$  as a kind of structure, which in fact coincides with a magma. We can see that paths between magmas correspond to paths between the underlying types  $X$  and paths over this between their operations  $f$ . This observation is further generalized by the Structure Identity Principle (SIP), formalized in [Ang+20]. Given a structure, which in our case is just a binary operation

```

MagmaStr : Type → Type
MagmaStr A = A → A → A

```

this principle produces an appropriate definition “structured equivalence”  $\iota$ . The  $\iota$  is such that if structures  $X, Y$  are  $\iota$ -equivalent, then they are identified. In this case, the  $\iota$  asks us to provide [plus-coherent](#), so we have just shown that the [plus](#) magma on [Leibniz](#)

```

MagmaL : Magma
fst MagmaL = Leibniz
snd MagmaL = plus

```

and the [\\_+\\_](#) magma on [N](#) and are identical

Replace  
with  
BinOp

```

MagmaN≅MagmaL : MagmaN ≡ MagmaL
MagmaN≅MagmaL = equivFun (MagmaΣPath _ _) proof
  where
    proof : MagmaN ≅[ MagmaEquivStr ] MagmaL
    fst proof = N≅L
    snd proof = plus-coherent

```

As a consequence, properties of `+_` directly yield corresponding properties of `plus`. For example,

```

plus-assoc : Associative == plus
plus-assoc = subst
  (λ A → Associative == (snd A))
MagmaN≅MagmaL
N-assoc

```

### 3 Specifying types (ready)

While the practical applications of the last example do not stretch very far<sup>3</sup>, the approach generalizes to the more relevant containers and their associated laws.

In the same vein as the last section, we could define a simple but inefficient array type, and a more efficient implementation using trees. Then we can show that these are equivalent, such that when the simple type satisfies a set of laws, trees will satisfy them as well. We could then start developing all sorts of complex implementations fine-tuned to each operation and figure out how these are equivalent to some simpler type, but let us first take a step back, and investigate how we can make this approach run smoothly in a simpler example.

Rather than inductively defining a container and then showing that it is represented by a lookup function, we can go the other way around and define a type by insisting that it is equivalent to such a function. This approach, in particular the case in which one calculates a container with the same shape as a numeral system, was dubbed numerical representations in [Oka98], and has some formalized examples in, e.g., [HS22] and [KG16]. Numerical representations form the starting point for defining more complex datastructures based on simpler ones, so let us demonstrate such a calculation.

#### 3.1 Numerical representations: from numbers to containers

We can compute the type of vectors starting from `N`.<sup>4</sup> For simplicity, we define them as a type computing function via the “use-as-definition” notation from before. We expect vectors to be represented by

<sup>3</sup>Considering that `N` is a candidate to be replaced by a more suitable unsigned integer type when compiling to Haskell anyway.

<sup>4</sup>This is adapted (and fairly abridged) from [HS22]



```

Lookup : Type → ℕ → Type
Lookup A n = Fin n → A

```

where we use the finite type `Fin` as an index into vector. Using this representation as a specification, we can compute both `Fin` and a type of vectors. The finite type can be computed from the evident definition

```

Fin-def : ∀ n → Def (Σ[ m ∈ ℕ ] m < n)
Fin-def zero = ⊥-strict (λ ()) use-as-def
Fin-def (suc n) =
  ua (<-split n) ·
  cong (T ⊔ _) (by-definition (Fin-def n)) use-as-def

```

```

Fin : ℕ → Type
Fin n = defined-by (Fin-def n)

```

using

```

<-split : ∀ n → (Σ[ m ∈ ℕ ] m < suc n) ≃ (T ⊔ (Σ[ m ∈ ℕ ] m < n))

```

Likewise, vectors can be computed by applying a sequence of type isomorphisms

```

Vec-def : ∀ A n → Def (Lookup A n)
Vec-def A zero = isContr→≡Unit isContr⊥→A use-as-def
Vec-def A (suc n) =
  ((T ⊔ Fin n) → A)
  ≡⟨ ua Π⊔≃ ⟩
  (T → A) × (Fin n → A)
  ≡⟨ cong₂ _×_
    (UnitToTypePath A)
    (by-definition (Vec-def A n)) ⟩
  A × (defined-by (Vec-def A n)) ■ use-as-def

```

```

Vec : ∀ A n → Type
Vec A n = defined-by (Vec-def A n)

```

*SIP doesn't mesh very well with indexed stuff, does HSIP help?*

Of course, a container would not be of much use without lookup functions, so we define an interface

```

record Array (V : Type → ℕ → Type) : Type₁ where
  field
    lookup : ∀ {A n} → V A n → Fin n → A
    tail : ∀ {A n} → V A (suc n) → V A n

```

which at the very least has to satisfy laws like

```

record ArrayLaws {C} (Arr : Array C) : Type₁ where
  field
    lookup•tail : ∀ {A n} (xs : C A (suc n)) (i : Fin n)
      → Arr .lookup (Arr .tail xs) i ≡ Arr .lookup xs (inr i)

```

We could directly show that `Vec` satisfies this, but now that we defined `Vec` from `Lookup` we might as well use this fact.

The implementation of arrays as functions is very straightforward

```

FunArray : Array Lookup
FunArray .lookup f = f
FunArray .tail f = f ◦ inr

```

and clearly satisfies our interface

```

FunLaw : ArrayLaws FunArray
FunLaw .lookup◦tail _ _ = refl

```

We can implement arrays based on `Vec` as well

```

VectorArray : Array Vec
VectorArray .lookup {n = n} = f n
  where
    f : ∀ {A} n → Vec A n → Fin n → A
    f (suc n) (x , xs) (inl _) = x
    f (suc n) (x , xs) (inr i) = f n xs i
VectorArray .tail (x , xs) = xs

```

and again, we can transport the proofs from `Lookup` to `Vec`.<sup>5</sup>

*As you can see, taking “use-as-definition” too literally prevents Agda from solving a lot of metavariables.*

*This computation can of course be generalized to any arity zeroless numeral system; unfortunately beyond this set of base types, this “straightforward” computation from numeral system to container loses its efficacy. In a sense, the  $n$ -ary natural numbers are exactly the base types for which the required steps are convenient type equivalences like  $(A + B) \rightarrow C = (A \rightarrow C) \times (B \rightarrow C)$ ?*

If one was determined to cobble together the path over path over path we need now.

### 3.2 Numerical representations as ornaments

We could perform the same computation for `Leibniz`, which would yield the type of binary trees, but we note that these computations proceed with roughly the same pattern: each constructor of the numeral system gets assigned a value, and is amended with a field holding a number of elements and subnodes using this value as a “weight”. But wait! Such modifications of constructors are already made formal by the concept of ornamentation!

Ornamentation, as exposed in [McB14] and [KG16], lets us formulate what it means for two types to have a “similar” recursive structure. This is achieved by interpreting (indexed inductive) datatypes from descriptions, between which an ornament is seen as a certificate of similarity, describing which fields or indices need to be introduced or dropped. Furthermore, a one-sided ornament: an ornamental description, lets us describe new datatypes by recording the modifications to an existing description.

This links back to the construction in the previous section, since `N` and `Vec` share the same recursive structure, so `Vec` can be formed by introducing indices

<sup>5</sup>Except in this oversimplified case the laws are trivial for `Vec` as well.

It seems like Agda forgets that we defined `Leibniz` if we move between different .tex files, so I’ll have to float all `AgdaTargets` to the top level at some point...

Again not sure how much space I

and adding a field holding an element at each node.<sup>6</sup>

However, instead deriving **List** from **N** generalizes to **Leibniz** with less notational overhead, so let's tackle that case first. For this, we have to give a description of **N** to work with

```
NatD : Desc T
NatD _ = σ Bool λ
  { false → y []
  ; true  → y [ tt ] }
```

Recall that  $\sigma$  adds a field, upon which the rest of the description may vary, and  $y$  lists the recursive fields and their indices (which can only be **tt**). We can now write down the ornament which adds fields to the **suc** constructor

```
NatD-ListO : Type → OrnDesc T ! NatD
NatD-ListO A (ok _) = σ Bool λ
  { false → y _
  ; true  → Δ A (λ _ → y (ok _, _)) }
```

Here, the  $\sigma$  and  $y$  are forced to match those of **NatD**, but the  $\Delta$  adds a new field. With the least fixpoint and description extraction from [KG16], this is sufficient to define **List**. Note that we cannot hope to give an unindexed ornament from **Leibniz**

```
LeibnizD : Desc T
LeibnizD _ = σ (Fin 3) λ
  { zero      → y []
  ; (suc zero) → y [ tt ]
  ; (suc (suc zero)) → y [ tt ] }
```

into trees, since trees have a very different recursive structure! Instead, we must keep track at what level we are in the tree so that we can ask for adequately many elements:

```
power : N → (A → A) → A → A
power N.zero f = λ x → x
power (N.suc n) f = f ∘ power n f
```

```
Two : Type → Type
Two X = X × X
```

```
LeibnizD-TreeO : Type → OrnDesc N ! LeibnizD
LeibnizD-TreeO A (ok n) = σ (Fin 3) λ
  { zero      → y _
  ; (suc zero) → Δ (power n Two A) λ _ → y (ok (suc n), _)
  ; (suc (suc zero)) → Δ (power (suc n) Two A) λ _ → y (ok (suc n), _) }
```

We use the **power** combinator to ensure that the digit at position  $n$ , which has weight  $2^n$  in the interpretation of a binary number, also holds its value times  $2^n$  elements. This makes sure that the number of elements in the tree shaped after a given binary number also is the value of that binary number.

This “folding in” technique using the indices to keep track of structure seems to apply more generally. Let us explore this a bit further, and return later to

Clearly this can use more explanation (the question is, how much?)

<sup>6</sup>These and similar examples are also documented in [KG16]

the generalization of the pattern from numeral systems to datastructures.

### 3.3 Folding in

Let us describe this procedure of folding a complex recursive structure into a simpler structure more generally. In particular, we will demonstrate that for linear datatypes, such as `N` and `Leibniz`, and for a given unindexed datatype, there is always an indexed datatype isomorphic to it at some index, and an ornament from the linear type to the indexed type.

Suppose we are given a description, the first thing we can do to simplify it is collect all fields in one place

```
RField : RDesc T → Type
RField (v is) = T
RField (σ S D) = Σ S λ s → RField (D s)
```

Next, we will certainly have to count the number of recursive occurrences we are tracking, so we define

```
-- note to self, I should probably make v _not_ overlap
-- so not everything links here
data Number : Type where
  1 : Number
  v : ∀ n → (Fin n → Number) → Number
```

where `1` records that we are at the top level, and `v` denotes that we are below a constructor with some number of recursive fields. This simplifies our task to implementing the types in

```
nested : Desc T → Desc Number
nested d n = σ (Fields (d tt) n) λ a → v [ subnodes a ]
```

such a way that we get an isomorphism

```
nested-eq : ∀ D → μ D tt ≈ μ (nested D) 1
```

Thus, `Fields` is forced to have a `leaf` constructor like

```
data Fields (d : RDesc T) : Number → Type where
  leaf : RField d → Fields d 1
  node : ∀ n {f : Fin n → Number}
    → ((i : Fin n) → Fields d (f i)) → Fields d (v n f)
```

if `nested` is to work at `1`. The `node` constructor makes sure that if we have collection of `Fields`, then we can gather them in a field at a higher level. We can then count the subnodes of a given `Fields` as

```
subnodes : ∀ {n} {d : RDesc T} → Fields d n → Number
subnodes (leaf x) = v (RSize _ x) λ _ → 1
subnodes (node n f) = v n (subnodes ∘ f)
```

where `RSize` counts the number of recursive fields of a particular branch

```
RSize : (d : RDesc T) → (a : RField d) → N
RSize (v is) a = length is
RSize (σ S D) a = RSize (D (fst a)) (snd a)
```

Note that `subnodes` effectively keeps the shape of the previous field, but unfolds the recursive fields at the bottom of the tree by one level.

*I then tried and realized how unpleasant even the functions from the original type to the nested type are to write.*

As a triviality, we get that any type, interpreted as a container, always decomposes as an ornament over a “numerical” base type. This links to the construction of binary heaps in [KG16], as in hindsight, starting from the usual binary heaps would yield binary numbers and their binary heap ornament (in a much less useful package).

Or at least, that was where I was trying to go with this, but I notice that this still is a bit further away.

## 4 Reducing friction (work in progress)

The setup some approaches in earlier sections require makes them tedious or impractical to apply. In this section we will look at some ways how part of this problem could be alleviated through generics, or by alternative descriptions of concepts like equivalences through the lens of initial algebras.

In later sections we will construct many more equivalences between more complicated types than before, so we will dive right into the latter. Reflecting upon section 2, we see that when one establishes an equivalence, most of the time is spent working out a series of tedious lemmas to show that the conversion functions are mutual inverses, which tend to be relatively easy to define. We take away two things from this; the first is that the conversion functions are perhaps too obvious, and the second is that we should really avoid talking about sections and retractions lest we incur tedium!<sup>7</sup> In fact, the machinery from [KG16] will come in handy to demonstrate to what extent our hand was actually forced in section 2.

We note that like how `N` is the fixpoint of `NatD`, one of the sides of the equivalence is almost always going to be a datatype, and hence the `μ` of a `Desc'`, so we will use this as the working assumption for the following section.

### 4.1 Well-founded monic algebras are initial

Unfortunately, the machinery from [KG16] relies on axiom K for a small but crucial part. To be precise, in a cubical setting, the type `μ` as given stops being initial for its base functor! In this section, we will be working with a simplified and repaired version. Namely, we simplify `Desc'` to

```
data Desc' : Set, where
  γ : (n : N) → Desc'
  σ : (S : Set) (D : S → Desc') → Desc'
```

To complete the definition of `μ`

```
data μ (D : Desc') : Set, where
  con : Base (μ D) D → μ D
```

we will need to implement `Base`. We remark that in [KG16], the recursion of `mapFold` is a structural descent in `[[ D' ]]` (`μ D`). Because `[[ ]]` is a function and

<sup>7</sup>The latter perhaps less so, because it is useful to show a map to be monic.

not a datatype, this descent becomes invalid, and `mapFold` fails the termination check. We resolve this by defining `Base` as a datatype

```
data Base (X : Set₁) : Desc' → Set₁ where
  in-γ : ∀ {n} → Vec X n → Base X (γ n)
  in-σ : ∀ {S D} → Σ[ s ∈ S ] (Base X (D s)) → Base X (σ S D)
```

such that this descent is allowed by the termination checker without axiom K.<sup>8</sup>

Recall that the `Base` functors of descriptions are special polynomial functors, and the fixpoint of a base functor is its initial algebra. The situation so far is summarized by the diagram

$$\begin{array}{ccc} & F\mu_F & \\ & \downarrow \text{con} & \\ X & \xleftarrow{e} \mu F & \end{array}$$

so, we are looking for sufficient conditions on  $X$  to get the equivalence  $e : X \cong \mu F$ . Note that when  $X \cong \mu F$ , then there necessarily is an initial algebra  $FX \rightarrow X$ . Conversely, if the algebra  $(X, f)$  is isomorphic to  $(\mu F, \text{con})$ , then  $X \cong \mu F$  would follow immediately, so it is equivalent to ask for the algebras to be isomorphic instead.

#### 4.1.1 Algebras of endofunctors

To characterize when such algebras are isomorphic, we reiterate some basic category theory, simultaneously rephrasing it in Agda terms.<sup>9</sup>

Let  $C$  be a category, and let  $a, b, c$  be objects of  $C$ , so that in particular we have identity arrows  $1_a : a \rightarrow a$  and for arrows  $g : b \rightarrow c, f : a \rightarrow b$  composite arrows  $gf : a \rightarrow c$  subject to associativity. In our case,  $C$  is the category of types, with ordinary functions as arrows.

Recall that an endofunctor, which is simply a functor  $F$  from  $C$  to itself, assigns objects to objects and sends arrows to arrows

```
F₀ : Type ℓ → Type ℓ
fmap : (A → B) → F₀ A → F₀ B
```

These assignments are subject to the identity and composition laws

```
f-id      : (x : F A)
            → mapF id x ≡ x

f-comp    : (g : B → C) (f : A → B) (x : F A)
            → mapF (g ∘ f) x ≡ mapF g (mapF f x)
```

An  $F$ -algebra is just a pair of an object  $a$  and an arrow  $Fa \rightarrow a$

```
record Algebra (F : Type ℓ → Type ℓ) : Type (ℓ-suc ℓ) where
  field
```

<sup>8</sup>This has, again by the absence of axiom K, the consequence of pushing the universe levels up by one. However, this is not too troublesome, as equivalences can go between two levels, and indeed types are equivalent to their lifts.

<sup>9</sup>We are not reusing a pre-existing category theory library for the simple reasons that it is not that much work to write out the machinery explicitly, and that such libraries tend to phrase initial objects in the correct way, which is too restrictive for us.

Maybe  
category  
theory ref-  
erence

**Carrier** : **Type**  $\ell$   
**forget** :  $F$  **Carrier**  $\rightarrow$  **Carrier**

Algebras themselves again form a category  $C^F$ . The arrows of  $C^F$  are the arrows  $f$  of  $C$  such that the following square commutes

$$\begin{array}{ccc} Fa & \xrightarrow{Ff} & Fb \\ U_a \downarrow & & \downarrow U_b \\ a & \xrightarrow{f} & b \end{array}$$

So we define

**Alg** $\rightarrow$ **Sqr**  $F$   $A$   $B$   $f = f \circ A$  .**forget**  $\equiv B$  .**forget**  $\circ F$  .**fmap**  $f$

and

**record** **Alg** $\rightarrow$  (**RawF** : **RawFunctor**  $\ell$ )  
 (**AlgA** **AlgB** : **Algebra** (**RawF** .**F**<sub>0</sub>)) : **Type**  $\ell$  **where**  
**constructor** **alg** $\rightarrow$

**field**

**mor** : **AlgA** .**Carrier**  $\rightarrow$  **AlgB** .**Carrier**  
**coh** :  $\parallel$  **Alg** $\rightarrow$ **Sqr** **RawF** **AlgA** **AlgB** **mor**  $\parallel$  ,

Note that we take the propositional truncation of the square, such that algebra maps with the same underlying morphism become propositionally equal

**Alg** $\rightarrow$ **Path** : {**F** : **RawFunctor**  $\ell$ } {**A B** : **Algebra** (**F** .**F**<sub>0</sub>)}  
 $\rightarrow (g$   $f$  : **Alg** $\rightarrow$   $F$   $A$   $B$ )  $\rightarrow g$  .**mor**  $\equiv f$  .**mor**  $\rightarrow g \equiv f$

The identity and composition in  $C^F$  arise directly from those of the underlying arrows in  $C$ .

Recall that an object  $\emptyset$  is initial when for each other object  $a$ , there is a unique arrow  $! : \emptyset \rightarrow a$ . By reversing the proofs of initiality of  $\mu$  and the main result of this section, we obtain a slight variation upon the usual definition. Namely, unicity is often expressed as contractability of a type

**isContr**  $A = \Sigma [x \in A] (\forall y \rightarrow x \equiv y)$

Instead, we again use a truncation

**weakContr**  $A = \Sigma [x \in A] (\forall y \rightarrow \parallel x \equiv y \parallel_1)$

but note that this also, crucially, slightly stronger than connectedness. We I think

define initiality for arbitrary relations

**record** **Initial** (**C** : **Type**  $\ell$ ) (**R** :  $C \rightarrow C \rightarrow$  **Type**  $\ell'$ )  
 (**Z** :  $C$ ) : **Type** ( $\ell$ -**max** ( $\ell$ -**suc**  $\ell$ )  $\ell'$ ) **where**

**field**

**initiality** :  $\forall X \rightarrow$  **weakContr** (**R** **Z**  $X$ )

such that it closely resembles the definition of least element. Then,  $A$  is an initial algebra when

**InitAlg** **RawF**  $A =$  **Initial** (**Algebra** (**RawF** .**F**<sub>0</sub>)) (**Alg** $\rightarrow$  **RawF**)  $A$

By basic category theory (using the usual definition of initial objects), two initial objects  $a$  and  $b$  are always isomorphic; namely, initiality guarantees that there are arrows  $f : a \rightarrow b$  and  $g : b \rightarrow a$ , which by initiality must compose to the identities again.

Similarly, we get that

$$\begin{aligned} \text{InitAlg} \simeq & : (F : \text{Functor } \ell) (A B : \text{Algebra } (F \text{ .RawF } F_0)) \\ & \rightarrow \text{InitAlg } (F \text{ .RawF}) A \rightarrow \text{InitAlg } (F \text{ .RawF}) B \\ & \rightarrow A \text{ .Carrier} \simeq B \text{ .Carrier} \end{aligned}$$

However, we only have the equalities from the isomorphism inside a propositional truncation. But fortunately, being an equivalence is a property, so we can eliminate from the truncations to get the wanted result.

Note that even though we warned ourselves, we are still talking about sections and retractions to establish that  $f$  is an equivalence! However, this result also makes sure we will not have to speak of them again.<sup>10</sup>

#### 4.1.2 Accessibility

In terms of “when is  $X \cong \mu D$ ”, this reduces the problem to showing that  $X$  is an initial algebra for the base functor of  $D$ ;  $\mu D$  is initial by its fold, and by induction on  $\mu D$  using the squares of algebra maps.

**Remark 4.1.** We need (in general) not hope  $\mu D$  is a strict initial object in the category of algebras. For a strict initial object, having a map  $a \rightarrow \emptyset$  implies  $a \cong \emptyset$ . This is not the case here. In general strict initial objects satisfy  $a \times \emptyset \cong \emptyset$ . Consider the algebras  $(\mathbb{N}, (0, \text{suc}))$  and  $(2^{\mathbb{N}}, f)$ , for which clearly not  $2^{\mathbb{N}} \times \mathbb{N} \cong \mathbb{N}$ . (In particular,  $F$  being a left adjoint is sufficient to make  $C^F$  have strict initial objects, but then the carrier of the initial algebra is simply  $\perp$ .)

I think.

Looking back at section 2, we see that **Leibniz** is an initial  $F : X \mapsto 1 + X$  algebra because for any other algebra, the image of **0b** is fixed, and by **bsuc** all other values are determined by chasing around the square. Thus, we are looking for a similar structure on  $f : FX \rightarrow X$  that supports recursion.

Clearly we will need something stronger than  $FX \cong X$ , as in general a functor can have many fixpoints. For this, we define what it means for an element  $x$  to be accessible by  $f$ . This definition uses a mutually recursive datatype as follows: We state that an element  $x$  of  $X$  is accessible when there is an accessible  $y$  in its fiber over  $f$

```
data Acc D f x where
  acc : (y : fiber f x) → Acc' D f D (fst y) → Acc D f x
```

Accessibility of an element  $x$  of **Base A E** is defined by cases on  $E$ ; if  $E$  is **y n** and  $x$  is a **Vec A n**, then  $x$  is accessible if all its elements are; if  $x$  is **σ S E'**, then  $x$  is accessible if **snd x** is

```
data Acc' D f where
  acc-y : All (Acc D f) x → Acc' D f (y n) (in-y x)
  acc-σ : Acc' D f (E s) x → Acc' D f (σ S E) (in-σ (s , x))
```

Consequently,  $X$  is well-founded for an algebra when all its elements are accessible

$$\text{Wf } D f = \forall x \rightarrow \text{Acc } D f x$$

We can see well-foundedness as an upper bound on the size of  $X$ , if it were larger than  $\mu D$ , some of its elements would inevitably get out of reach of an

<sup>10</sup>In this section...



algebra. *Now* having  $FX \cong X$  also gives us a lower bound, but remark that having a well-founded injection  $f : FX \rightarrow X$  is already sufficient, as accessibility makes this an epi and an iso. Thus, we claim

**Claim 4.1.** If there is a mono  $f : FX \rightarrow X$  and  $X$  is well-founded for  $f$ , then  $X$  is an initial  $F$ -algebra.

#### 4.1.3 Proof sketch of 4.1

Let us be on our way. Suppose  $X$  is well-founded for the mono  $f : FX \rightarrow X$ . To show that  $(X, f)$  is initial, let us take another algebra  $(Y, g)$ , and show that there is a unique arrow  $(X, f) \rightarrow (Y, g)$ .

By **Acc**-recursion and because all  $x$  are accessible, we can define a plain map into  $Y$

$$\begin{aligned} \text{Wf-rec} : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X.\text{forget}) \\ &\rightarrow (\dot{F} D A \rightarrow A) \rightarrow X.\text{Carrier} \rightarrow A \end{aligned}$$

This construction is an instance of the broad concept of “well-founded recursion”, so we let ourselves be inspired by these methods. In particular, we prove an irrelevance lemma

$$\text{Wf-rec-irrelevant} : \forall x' y' x a b \rightarrow \text{rec } x' x a \equiv \text{rec } y' x b$$

which implies the unfolding lemma

$$\begin{aligned} \text{unfold-Wf-rec} : \forall x' &\rightarrow \text{rec } (cx \ x') (cx \ x') (\text{wf } (cx \ x')) \\ &\equiv f (\text{Base-map } (\lambda y \rightarrow \text{rec } y y (\text{wf } y)) \ x') \end{aligned}$$

The unfolding lemma ensures that the map we defined by **Wf-rec** is a map of algebras.

The proof that this map is unique proceeds analogously to that in the proof that  $\mu D$  is initial, but now by **Acc**-recursion

$$\begin{aligned} \text{Wf+inj} \rightarrow \text{Init} : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X.\text{forget}) \\ &\rightarrow \text{injective } (X.\text{forget}) \rightarrow \text{InitAlg } (\text{Raw } \dot{F} D) X \end{aligned}$$

Thus, we conclude that  $X$  is initial.

The main result is then a corollary of initiality of  $X$  and the isomorphism of initial objects

$$\begin{aligned} \text{Wf+inj} \equiv \mu : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X.\text{forget}) \\ &\rightarrow \text{injective } (X.\text{forget}) \rightarrow X.\text{Carrier} \equiv \mu D \end{aligned}$$

#### 4.1.4 Example

Let us redo the proof in section 2, now using this result.

## 5 Is equivalence too strong (finger trees)

## 6 Discussion and future work (aka the union of my to-do list and the actual future work section)

I think, but I'm not sure how much surjective=epi, injective=mono and “(Type a) is balanced” I can safely sweep under the carpet.  
stdlib

## 7 Temporary

### Todo list

Not sure if it would be helpful to have a more extensive introduction covering all features used. . . . .	3
To be precise, the mapFold in [KG16] gets painted red. . . . .	3
is this too much code and too little explanation at once? . . . . .	4
Replace with BinOp . . . . .	7
If one was determined to cobble together the path over path over path we need now. . . . .	10
It seems like Agda forgets that we defined Leibniz if we move between different .tex files, so I'll have to float all AgdaTargets to the top level at some point... . . . . .	10
Again not sure how much space I should use to reiterate Desc, Orn, and OrnDesc. . . . .	10
Clearly this can use more explanation (the question is, how much?) . . . .	11
Or at least, that was where I was trying to go with this, but I notice that this still is a bit further away. . . . .	13
Maybe category theory reference . . . . .	14
I think . . . . .	15
I think. . . . .	16
I think, but I'm not sure how much surjective=epi, injective=mono and "(Type a) is balanced" I can safely sweep under the carpet. . . . .	17
stdlib . . . . .	17

### References

- [Ang+20] Carlo Angiuli et al. *Internalizing Representation Independence with Univalence*. 2020. DOI: 10.48550/ARXIV.2009.05547. URL: <https://arxiv.org/abs/2009.05547>.
- [DM14] PIERRE-ÉVARISTE DAGAND and CONOR McBRIDE. “Transporting functions across ornaments”. In: *Journal of Functional Programming* 24.2-3 (Apr. 2014), pp. 316–383. DOI: 10.1017/s0956796814000069. URL: <https://doi.org/10.1017%2Fs0956796814000069>.
- [EC22] Lucas Escot and Jesper Cockx. “Practical Generic Programming over a Universe of Native Datatypes”. In: *Proc. ACM Program. Lang.* 6.ICFP (Aug. 2022). DOI: 10.1145/3547644. URL: <https://doi.org/10.1145/3547644>.
- [HS22] Ralf Hinze and Wouter Swierstra. “Calculating Datastructures”. In: *Mathematics of Program Construction*. Ed. by Ekaterina Komenantskaya. Cham: Springer International Publishing, 2022, pp. 62–101. ISBN: 978-3-031-16912-0.

- [KG16] HSIANG-SHANG KO and JEREMY GIBBONS. “Programming with ornaments”. In: *Journal of Functional Programming* 27 (2016), e2. DOI: 10.1017/S0956796816000307.
- [McB14] Conor McBride. “Ornamental Algebras, Algebraic Ornaments”. In: 2014.
- [Oka98] Chris Okasaki. *Purely Functional Data Structures*. USA: Cambridge University Press, 1998. ISBN: 0521631246.
- [Tea23] Agda Development Team. *Agda*. 2023. URL: <https://agda.readthedocs.io/en/v2.6.3/>.
- [VMA19] Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. “Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types”. In: *Proc. ACM Program. Lang.* 3.ICFP (July 2019). DOI: 10.1145/3341691. URL: <https://doi.org/10.1145/3341691>.