

Generic Proofs and Constructions in Agda

Samuel Klumpers
6057314

April 11, 2023

Contents

I	Proposal	3
1	Outline	3
2	Program Vivisection	3
3	This is going to be (re)moved: “Introduction”	4
4	Research Question and Contributions	5
5	Related work	5
6	Planning	6
II	Preliminary work	6
7	Proof Transport via the Structure Identity Principle	6
7.1	Unary numbers are binary numbers	7
7.2	Functions from specifications	9
7.3	The Structure Identity Principle	10
8	Types from Specifications: Ornamentation and Calculation	11
8.1	From numbers to containers	11
8.2	Numerical representations as ornaments	13
8.3	Heterogeneization	15
9	More equivalences for less effort	16
9.1	Well-founded monic algebras are initial	16
9.1.1	Datatypes as initial algebras	17
9.1.2	Accessibility	19
9.1.3	Proof sketch of Claim 9.1	20

9.1.4	Example	20
-------	-------------------	----

Part I

Proposal

1 Outline

In this document I propose a master thesis project, which will investigate and attempt to counter the obstacles one can encounter when replacing one data-structure with a more complicated one, while keeping the end-result verifiable.

Section 2 explains why you might want to do this, and describes some challenges that can arise from this, along with known ways to tackle them.

Section 4 poses some remaining and new questions regarding this topic, and proposes methods and ideas to tackle and answer the open challenges and questions.

Section 5 lists literature and frameworks relevant to this and similar problems.

Section 6 proposes a planning for the described project.

part II summarizes and contains the preliminary work done for this project.

2 Program Vivisection

“Spend some more time on the context”

Agda [Tea23] is a functional programming language and a proof assistant, taking inspiration from languages like Haskell and other proof assistants like Coq. We can write programs like we would in Haskell, and then express and prove their properties all inside Agda. This allows us to demonstrate the correctness of programs by formal proof rather than by testing; of course, producing an often tedious proof typically demands more effort than covering the relevant code with testcases.

In this thesis, we will explore some methods of proving properties of our programs, focussing on the problems or inconveniences that may arise, and how to deal with them. Let us sketch some problems and their remedies to get an idea of what awaits us, before we dive into the nitty-gritty details.

First, merely adapting a program to Agda may already require changes to the datatypes used in it; for example, if a program manipulating a `List` uses the unsafe `head` function, then one is forced to replace the `List` by a datatype that ensures non-emptiness, such as a `NonEmpty` list or a length-aware vector `Vec`. On the other hand, there might be sections of a program where the concrete length is not relevant for correctness and only gets in the way. As a result, one might find themselves duplicating common functions like concatenation `_++_` to only alter their signatures.

Clearly we would not be writing this if there was no way out, which there is! Often, the “new” datatype (`Vec`) is simply a variation on the old datatype (`List`) making small adjustments to the existing constructors; in this case, we decorate the `nil` and `cons` constructors with natural numbers representing the

length. This kind of modification of types falls in the framework of ornamentation as described by Ko and Gibbons [KG16]; if two types are reified to their *descriptions*, then *ornaments* express whether the types are “similar” by acting as a recipe to produce one type from the other. By restricting the operations to the copying of corresponding parts, and the introduction of fields or dropping of indices, the existence of such an ornament ensures that the types have the same recursive structure.

Write here about: “Something about patches.”

Write here about: “For each invariant a new datatype? Still ornaments”

Now that we know we can decently satisfyingly organize similar datatypes, it is time to look at dissimilar datatypes. It is certainly not foolish to prototype a program using simpler types or implementations, and only replace these with more performant alternatives in critical places; knowing that this is eventually going to happen, one might as well prepare for it. While this may quickly turn into a refactoring nightmare in the general case, we can hope for a more satisfying transition if we restrict our attention to a more narrow scope. As an example, we might start programming something with a `List`, but replace this with a `Tree` if we notice that the program spends most of its time in `lookup` operations. We are of course reimplementing the operations on `Tree` utilizing their binary nature to gain a speedup, but it also seems as though we are about to double the number of necessary proofs; however, we have two ways to avoid this problem.

We will look at the more specific solution first. This solution is guided by the realization that `List` and `Tree`, like most other containers, still have something in common even if their recursive structure is very different. That is, both resemble a number system, and, Okasaki [Oka98] notes that this resemblance to number systems is “surprisingly common”. In the case of lists and Braun trees¹, one can present both by deriving them from unary and binary numbers respectively, as is made formal by Hinze and Swierstra [HS22]. One can then apply this *numerical representation* to simplify or make trivial the proofs of the properties we hesitated to duplicate before.

Write here about: “If we instead hide our datatypes behind interfaces, we can use proof transport as an alternative.”

Write here about: “Something about fingertrees, leading into the research question and proposed work”

3 This is going to be (re)moved: “Introduction”

Something similar happens when replacing an implementation with a more efficient one. For example, when implementing binary trees as a more efficient alternative to lists, the proofs of the same properties will differ between list and tree, and tend to be more difficult for the latter. Switching between implementations of an interface not only duplicates code, but also (and sometimes more than) doubles the effort required to verify both.

¹Braun trees are a kind of binary tree, of which its shape is determined by its size.

concrete example?

Other work like [HS22]

don't use \cite as noun

simplifies the proofs relating to certain containers directly, formally executing the way of thought of numerical representations as noted in [Oka98].

When two types are isomorphic and equivalent under an interface, proofs of properties of these implementations should be interconvertible. By using structured equivalences and univalence, [Ang+20] characterizes equivalences under interfaces.

In Section 7, we will follow [Ang+20], and look at how proofs on unary naturals can be transported to the binary naturals. Then in Section 8 we recall how numeral systems in particular induce container types in [HS22], which we attempt to reformulate in the language of ornaments in Subsection 8.2, using the framework of [KG16]. In Section 9 we investigate how we can make the earlier methods more easily accessible to the user, and, ourselves, when we give a description of finger trees in ??.

Ok, but make the research question more concrete

4 Research Question and Contributions

The research question of this project will be: *can we describe finger trees [HP06] in the frameworks of numerical representations and ornamentation [KG16], simplifying the verification of their properties as flexible two-sided arrays?* We expect to encounter a variety of obstacles along the way, and so the project will focus on finding existing or developing new theory to deal with these. For example, an obstacle can be that the corresponding number system has many representations for the same number, which we expect to describe using quotients [VMA19] and reason about using representation independence [Ang+20]. If this is accomplished or deemed infeasible at an early stage, we can generalize the results we have to other related problems; for example, we may view the problem of generating arbitrary values for testing as an instance of an enumeration problem, but then through the lens of ornaments.

5 Related work

Write here about: “Organize this into a human text”

IRI	Equivalences from relations, transport over structures
CalcdData	Prototyping datatypes with properties deriving from the underlying numbers, enjoy the
ProgOrn	Using simple types as stepping stones to more impressive types, generalization of num
PurelyFunct	Not so much work in the same direction, but definitely a source of inspiration, and a g
HetBin	A predecessor/inspiration for heterogeneization
GLookup	Perhaps an inspiration to do more with the numerical representations/ornaments expe
OrnTrans	How many functions do we keep when we turn our naturals into lists or vectors?
HSIP	Can we push SIP to eat indexed structures? (I still have to attempt to digest this)
PractGen	Source of inspiration for some generic constructions (and perhaps a list of things we d
TranspGalois	It claims to achieve what univalence does, without the Cubical/HoTT overhead? (Hav
CoveringBases	Inspiration/application for enumeration
GuideToRecSch	Like mapFold, but more probably
InitFinal?	The least fixpoint of $X \mapsto \top + A \times X$ is a list, the largest is a stream. What happens
AmortCoind	Could we derive the amortized bounds on (our or Ralf's) fingertrees using this? (Not s

6 Planning

Write here about: “This will be done”

Part II

Preliminary work

Write here about: “This has been done”

7 Proof Transport via the Structure Identity Principle

Let us quickly review some features of Cubical Agda [VMA19] that we will use in this section.

In Cubical Agda, the primitive notion of equality arises not (directly) from the indexed inductive definition we are used to, but rather from the presence of the interval type `I`. This type represents a set of two points `i0` and `i1`, which are considered “identified” in the sense that they are connected by a path. To define a function out of this type, we also have to define the function on all the intermediate points, which is why we call such a function a “path”. Terms of other types are then considered identified when there is a path between them.

While the benefits are overwhelming for us

Which?

, this is not completely without downsides, such as that the negation of axiom K complicates both some termination checking and some universe levels.²

²In particular, this prompts rather far-reaching (but not fundamental) changes to the code

Furthermore, if we use certain homotopical constructions, and we wish to eliminate from our types as if they were sets, then we will also have to prove that they actually are sets.

On the positive side, this different perspective gives intuitive interpretations to some proofs of equality, like

```
sym : x = y → y = x
sym p i = p (~ i)
```

where `~_` is the interval reversal, swapping `i0` and `i1`, so that `sym` simply reverses the given path.

Furthermore, because we can now interpret paths in record and function types in a new way, we get a host of “extensionality” for free. For example, a path in $A \rightarrow B$ is indeed a function which takes each i in I to a function $A \rightarrow B$. Using this, function extensionality becomes tautological

```
funExt : (∀ x → f x = g x) → f = g
funExt p i x = p x i
```

Finally, much of our work will rest on equivalences, as the “HoTT-compatible” generalization of bijections. This is because in Cubical Agda, we have the univalence theorem

```
ua : ∀ {A B : Type ℓ} → A ≃ B → A = B
```

stating that “equivalent types are identified”, such that type isomorphisms like $1 \rightarrow A \simeq A$ actually become paths $1 \rightarrow A \equiv A$, making it so that we can transport proofs along them. We will demonstrate this by a slightly more practical example in the next section.

7.1 Unary numbers are binary numbers

Let us demonstrate an application of univalence by exploiting the equivalence of the “Peano” naturals and the “Leibniz” naturals. Recall that the Peano naturals are defined as

```
data N : Type where
  zero : N
  suc : N → N
```

This definition enjoys a simple induction principle and is well-covered in most libraries. However, the definition is also impractically slow, since most arithmetic operations defined on `N` have time complexity in the order of the value of the result.

As an alternative we can use the more performant binary numbers, for which for example addition has logarithmic time complexity. It would seem like these are less comprehensively proven about in typical libraries, but fortunately, this does not have to be a problem: the `N` naturals and the binary numbers should be equivalent after all!

Let us make this formal. We define the Leibniz naturals as follows:

```
data Leibniz : Set where
  0b : Leibniz
```

of previous work, such as to the machinery of ornaments [KG16] in Section 9.

```

    _1b : Leibniz → Leibniz
    _2b : Leibniz → Leibniz

```

Here, the `0b` constructor encodes 0, while the `_1b` and `_2b` constructors respectively add a 1 and a 2 bit, under the usual interpretation of binary numbers:

```

toN : Leibniz → ℕ
toN 0b = 0
toN (n 1b) = 1 + 2 * toN n
toN (n 2b) = 2 + 2 * toN n
[[ ]] = toN

```

This defines one direction of the equivalence from `ℕ` to `Leibniz`, for the other direction, we can interpret a number in `ℕ` as a binary number by repeating the successor operation on binary numbers:

```

bsuc : Leibniz → Leibniz
bsuc 0b = 0b 1b
bsuc (n 1b) = n 2b
bsuc (n 2b) = (bsuc n) 1b

```

```

fromN : ℕ → Leibniz
fromN 0 = 0b
fromN (suc n) = bsuc (fromN n)

```

To show that `toN` is an isomorphism, we have to show that it is the inverse of `fromN`. By induction on `Leibniz` and basic arithmetic on `ℕ` we see that

```

toN-suc : ∀ x → [[ bsuc x ]] = suc [[ x ]]

```

so `toN` respects successors. Similarly, by induction on `ℕ` we get

```

fromN-1+2· : ∀ x → fromN (1 + double x) = (fromN x) 1b

```

and

```

fromN-2+2· : ∀ x → fromN (2 + double x) = (fromN x) 2b

```

so that `fromN` respects even and odd numbers. We can then prove that applying `toN` and `fromN` after each other is the identity by repeating these lemmas

```

N↔L : Iso ℕ Leibniz
N↔L = iso fromN toN sec ret
where
  sec : section fromN toN
  ret : retract fromN toN

```

This isomorphism can be promoted to an equivalence

```

N≅L : ℕ ≅ Leibniz
N≅L = isoToEquiv N↔L

```

which, finally, lets us identify `ℕ` and `Leibniz` by univalence

```

N=L : ℕ = Leibniz
N=L = ua N≅L

```

The path `N=L` then allows us to transport properties from `ℕ` directly to `Leibniz`; as an example, we have not yet shown that `Leibniz` is discrete, i.e., has decidable equality. Using substitution, we can quickly derive this³

³Of course, this gives a rather inefficient equality test, but for the homotopical consequences


```

discreteL : Discrete Leibniz
discreteL = subst Discrete N=L discreteN

```

This can be generalized even further to transport proofs about operations from `N` to `Leibniz`.

7.2 Functions from specifications

As an example, we will define addition of binary numbers. We could transport `_+_` as a binary operation

```

BinOp : Type → Type
BinOp A = A → A → A

```

from `N` to `Leibniz` to get

```

_+'_ : BinOp Leibniz
_+'_ = subst BinOp N=L N._+_

```

Unfortunately this is rather inefficient, incurring an $O(n + m)$ overhead when adding n and m . It is more efficient to define addition on `Leibniz` directly, making use of the binary nature of `Leibniz`, while agreeing with the addition on `N`. Such a definition can be derived from the specification “agrees with `_+_`”, so we implement a syntax for giving definitions by equational reasoning, inspired by the “use-as-definition” notation used by Hinze and Swierstra [HS22]: Using an implicit pair type

```

record Σ' (A : Set a) (B : A → Set b) : Set (ℓ-max a b) where
  constructor _use-as-def
  field
    {fst} : A
    snd : B fst

```

we define

```

Def : {X : Type a} → X → Type a
Def {X = X} x = Σ' X λ y → x ≡ y

defined-by : {X : Type a} {x : X} → Def x → X
by-definition : {X : Type a} {x : X} → (d : Def x) → x ≡ defined-by d

```

which extracts a definition as the right endpoint of a given path.

With this we can define addition on `Leibniz` and show it agrees with addition on `N` in one motion

```

plus-def : ∀ x y → Def (fromN ([ x ] + [ y ]))
plus-def 0b y =
  fromN [ y ]
  =< N↔L .rightInv y >
  y ■ use-as-def
plus-def (x 1b) (y 1b) =
  fromN ((1 + double [ x ]) + (1 + double [ y ]))
  =< solved >
  fromN (2 + (double ([ x ] + [ y ])))
  =< fromN-2+2· ([ x ] + [ y ]) >

```

this is not a problem.

```

    fromN ([ x ] + [ y ]) 2b
  =< cong _2b (by-definition (plus-def x y)) >
    defined-by (plus-def x y) 2b ■ use-as-def
-- ...

```

Now we can easily extract the definition of `plus` and its correctness with respect to `_+_`

```

plus : ∀ x y → Leibniz
plus x y = defined-by (plus-def x y)

plus-coherent : ∀ x y → fromN (x + y) = plus (fromN x) (fromN y)
plus-coherent x y = cong fromN
  (cong₂ _+_ (sym (N↔L .leftInv x)) (sym (N↔L .leftInv _))) ·
  by-definition (plus-def (fromN x) (fromN y))

```

We remark that `Def` is close in concept to refinement types⁴, but extracts the value from the proof, rather than requiring it before.⁵

7.3 The Structure Identity Principle

We point out that `N` with `N.+` and `Leibniz` with `plus` form magmas. That is, inhabitants of

magmaa

Using the well-known fact that a path in a Σ type is just a Σ of paths, we get a path from $(\mathbf{N}, N.+)$ to $(\text{Leibniz}, \text{plus})$. This observation is further generalized by the Structure Identity Principle (SIP) as a form of representation independence [Ang+20]. Given a structure, which in our case is just a binary operation

```

MagmaStr : Type → Type
MagmaStr = BinOp

```

this principle produces an appropriate definition “structured equivalence” ι . The ι is such that if structures X, Y are ι -equivalent, then they are identified. In the case of `MagmaStr`, the ι asks us to provide something with the same type as `plus-coherent`, so we have just shown that the `plus` magma on `Leibniz`

```

MagmaL : Magma
fst MagmaL = Leibniz
snd MagmaL = plus

```

and the `_+_` magma on `N` and are identical

```

MagmaN≅MagmaL : MagmaN = MagmaL
MagmaN≅MagmaL = equivFun (MagmaΣPath _ _) proof
  where
    proof : MagmaN ≅[ MagmaEquivStr ] MagmaL
    fst proof = N≅L
    snd proof = plus-coherent

```

As a consequence, properties of `_+_` directly yield corresponding properties of `plus`. For example,

⁴À la Data.Refinement.

⁵Unfortunately, normalizing an application of a `defined-by` function also causes a lot of unnecessary wrapping and unwrapping, so `Def` is mostly only useful for presentation.

```

plus-assoc : Associative == plus
plus-assoc = subst
  (λ A → Associative == (snd A))
MagmaN≅MagmaL
N-assoc

```

Express what this accomplishes, and why this is impressive compared to without univalence

8 Types from Specifications: Ornamentation and Calculation

Suppose that we started writing and verifying some code using a vector-based implementation of the two-sided flexible array interface, but later decide to reimplement more efficiently using trees. It would be a shame to lay aside our vector lemmas, and rebuild the correctness proofs for trees from scratch. So, we take a step back and note that both vectors and trees can be represented by their `lookup` function. In fact, we can ask for more, and rather than defining an array-like type and then showing that it is represented by a lookup function, we can go the other way around and define types by insisting that they are equivalent to such a function. This approach, in particular the case in which one calculates a container with the same shape as a numeral system, was dubbed numerical representations by Okasaki [Oka98], and has some formalized examples in by Hinze and Swierstra [HS22] and Ko and Gibbons [KG16]. Numerical representations are our starting point for defining more complex datastructures based on simpler ones, so let us demonstrate such a calculation.

8.1 From numbers to containers

We can compute the type of vectors starting from `N`.

Is there a simple twist or other interesting example that we can run through instead, or would anything else be too abrupt without starting from this simple case?

⁶ For simplicity, we define them as a type computing function via the “use-as-definition” notation from before. We expect vectors to be represented by

```

Lookup : Type → N → Type
Lookup A n = Fin n → A

```

where we use the finite type `Fin` as an index into vector. Using this representation as a specification, we can compute both `Fin` and a type of vectors. The finite type can be computed from the evident definition

⁶This is adapted (and fairly abridged) from Calculating Datastructures [HS22]

```

Fin-def :  $\forall n \rightarrow \text{Def } (\Sigma[ m \in \mathbb{N} ] m < n)$ 
Fin-def zero =
  ( $\Sigma[ m \in \mathbb{N} ] m < 0$ )
  =<  $\perp$ -strict ( $\lambda ()$ ) >
   $\perp$  ■ use-as-def
Fin-def (suc n) =
  ( $\Sigma[ m \in \mathbb{N} ] m < \text{suc } n$ )
  =< ua (<-split n) >
   $\top \wr (\Sigma[ m \in \mathbb{N} ] m < n)$ 
  =< cong ( $\top \wr$ ) (by-definition (Fin-def n)) >
   $\top \wr$  defined-by (Fin-def n) ■ use-as-def

```

```

Fin :  $\mathbb{N} \rightarrow \text{Type}$ 
Fin n = defined-by (Fin-def n)

```

using

```

<-split :  $\forall n \rightarrow (\Sigma[ m \in \mathbb{N} ] m < \text{suc } n) \simeq (\top \wr (\Sigma[ m \in \mathbb{N} ] m < n))$ 

```

Likewise, vectors can be computed by applying a sequence of type isomorphisms

```

Vec-def :  $\forall A n \rightarrow \text{Def } (\text{Lookup } A n)$ 
Vec-def A zero =
  ( $\perp \rightarrow A$ )
  =< isContr  $\rightarrow$   $\equiv$  Unit isContr  $\perp \rightarrow A$  >
   $\top$  ■ use-as-def
Vec-def A (suc n) =
  (( $\top \wr$  Fin n)  $\rightarrow A$ )
  =< ua  $\Pi \wr$  >
  ( $\top \rightarrow A$ )  $\times$  (Fin n  $\rightarrow A$ )
  =< cong2  $\_ \times \_$ 
    (UnitToTypePath A)
    (by-definition (Vec-def A n)) >
  A  $\times$  (defined-by (Vec-def A n)) ■ use-as-def

```

```

Vec :  $\forall A n \rightarrow \text{Type}$ 
Vec A n = defined-by (Vec-def A n)

```

SIP doesn't mesh very well with indexed stuff, does HSIP help?

We can implement the following interface using `Vec`

```

record Array (V : Type  $\rightarrow \mathbb{N} \rightarrow \text{Type}$ ) : Type1 where
  field

```

```

  lookup :  $\forall \{A n\} \rightarrow V A n \rightarrow \text{Fin } n \rightarrow A$ 
  tail :  $\forall \{A n\} \rightarrow V A (\text{suc } n) \rightarrow V A n$ 

```

and show that this satisfies some usual laws like

```

record ArrayLaws {C} (Arr : Array C) : Type1 where
  field

```

```

  lookup•tail :  $\forall \{A n\} (xs : C A (\text{suc } n)) (i : \text{Fin } n)$ 
     $\rightarrow \text{Arr}.\text{lookup } (\text{Arr}.\text{tail } xs) i \equiv \text{Arr}.\text{lookup } xs (\text{inr } i)$ 

```

However, now that we defined `Vec` from `Lookup` we might as well use that.

The implementation of arrays as functions is very straightforward

```
FunArray : Array Lookup
FunArray .lookup f = f
FunArray .tail f = f ∘ inr
```

and clearly satisfies our interface

```
FunLaw : ArrayLaws FunArray
FunLaw .lookup∘tail _ _ = refl
```

We can implement arrays based on `Vec` as well⁷

```
VectorArray : Array Vec
VectorArray .lookup {n = n} = f n
  where
    f : ∀ {A} n → Vec A n → Fin n → A
    f (suc n) (x , xs) (inl _) = x
    f (suc n) (x , xs) (inr i) = f n xs i
VectorArray .tail (x , xs) = xs
```

Now, rather than rederiving the laws for vectors, the equality allows us to transport them from `Lookup` to `Vec`.⁸

As you can see, taking “use-as-definition” too literally prevents Agda from solving a lot of metavariables.

This computation can of course be generalized to any arity zeroless numeral system; unfortunately beyond this set of base types, this “straightforward” computation from numeral system to container loses its efficacy. In a sense, the n -ary natural numbers are exactly the base types for which the required steps are convenient type equivalences like $(A + B) \rightarrow C = (A \rightarrow C) \times (B \rightarrow C)$?

8.2 Numerical representations as ornaments

Reflecting on this derivation for `N`, we could perform the same computation for `Leibniz` to get Braun trees. However, we note that these computations proceed with roughly the same pattern: each constructor of the numeral system gets assigned a value, and is amended with a field holding a number of elements and subnodes using this value as a “weight”. This kind of “modifying constructors” is formalized by ornamentation [KG16], which lets us formulate what it means for two types to have a “similar” recursive structure. This is achieved by interpreting (indexed inductive) datatypes from descriptions, between which an ornament is seen as a certificate of similarity, describing which fields or indices need to be introduced or dropped to go from one description to the other. Furthermore, one-sided ornaments (ornamental descriptions) lets us describe new

⁷Note that, like any other type computing representation, we pay the price by not being able to pattern match directly on our type.

⁸Except that due to the simplicity of this case, the laws are trivial for `Vec` as well.

datatypes by recording the modifications to an existing description.

Put some minimal definitions here.

This links back to the construction in the previous section, since **N** and **Vec** share the same recursive structure, so **Vec** can be formed by introducing indices and adding a field holding an element at each node.

Instead, deriving **List** from **N** generalizes to **Leibniz** with less notational overhead, so we tackle that case first. We use the following description of **N**

```
NatD : Desc  $\top$   $\ell$ -zero
NatD _ =  $\sigma$  Bool  $\lambda$ 
  { false  $\rightarrow$   $\mathbf{v}$  []
  ; true  $\rightarrow$   $\mathbf{v}$  [ tt ] }
```

Here, σ adds a field to the description, upon which the rest of the description can vary, and \mathbf{v} lists the recursive fields and their indices (which can only be **tt**).

We can now write down the ornament which adds fields to the **suc** constructor

```
NatD-ListO : Type  $\rightarrow$  OrnDesc  $\top$  ! NatD
NatD-ListO A (ok _) =  $\sigma$  Bool  $\lambda$ 
  { false  $\rightarrow$   $\mathbf{v}$  _
  ; true  $\rightarrow$   $\Delta$  A ( $\lambda$  _  $\rightarrow$   $\mathbf{v}$  (ok _, _)) }
```

Here, the σ and \mathbf{v} are forced to match those of **NatD**, but the Δ adds a new field. With the least fixpoint and description extraction, this is sufficient to define **List**. Note that we cannot hope to give an unindexed ornament from **Leibniz**

```
LeibnizD : Desc  $\top$   $\ell$ -zero
LeibnizD _ =  $\sigma$  (Fin 3)  $\lambda$ 
  { zero  $\rightarrow$   $\mathbf{v}$  []
  ; (suc zero)  $\rightarrow$   $\mathbf{v}$  [ tt ]
  ; (suc (suc zero))  $\rightarrow$   $\mathbf{v}$  [ tt ] }
```

into trees, since trees have a very different recursive structure! Thus, we must keep track at what level we are in the tree so that we can ask for adequately many elements:

```
power : N  $\rightarrow$  (A  $\rightarrow$  A)  $\rightarrow$  A  $\rightarrow$  A
power N.zero f =  $\lambda$  x  $\rightarrow$  x
power (N.suc n) f = f  $\circ$  power n f
```

```
Two : Type  $\rightarrow$  Type
Two X = X  $\times$  X
```

```
LeibnizD-TreeO : Type  $\rightarrow$  OrnDesc N ! LeibnizD
LeibnizD-TreeO A (ok n) =  $\sigma$  (Fin 3)  $\lambda$ 
  { zero  $\rightarrow$   $\mathbf{v}$  _
  ; (suc zero)  $\rightarrow$   $\Delta$  (power n Two A)  $\lambda$  _  $\rightarrow$   $\mathbf{v}$  (ok (suc n), _)
  ; (suc (suc zero))  $\rightarrow$   $\Delta$  (power (suc n) Two A)  $\lambda$  _  $\rightarrow$   $\mathbf{v}$  (ok (suc n), _) }
```

We use the **power** combinator to ensure that the digit at position n , which has weight 2^n in the interpretation of a binary number, also holds its value times 2^n elements. This makes sure that the number of elements in the tree shaped after a given binary number also is the value of that binary number.

8.3 Heterogeneization

The situation in which one wants to collect a variety of types is not uncommon, and is typically handled by tuples. However, if e.g., you are making a game in Haskell, you might feel the need to maintain a list of “Drawables”, which may be of different types. Such a list would have to be a kind of “heterogeneous list”. In Haskell, this can be resolved by using an existentially quantified list, which, informally speaking, can contain any type implementing some constraint, but can only be inspected as if it contains the intersection of all types implementing this constraint.

This of course ports fairly directly to Agda, but is cumbersome when we just want to make a pile of different types, and becomes impractical if we want to be able to inspect the elements. The alternative is to split our heterogeneous list into two parts; one tracking the types, and one tracking the values. In practice, this means that we implement a heterogeneous list as a list of values indexed over a list of types. This approach and mainly its specialization to binary trees is investigated by Swierstra [SWI20].

We will demonstrate that we can express this “lift a type over itself” operation as an ornament. For this, we make a small adjustment to `RDesc` to track a type parameter separately from the fields. Using this we define an ornament-computing function, which given a description computes an ornamental description on top of it:

Oops

```

HetO' : (D : RDesc ⊤ ℓ-zero) (E : RDesc ⊤ ℓ-zero) (x : F (λ _ → D) (μ (λ _ → E) Type) Type tt) → ROrnDes
HetO' (γ is) E x = γ (map-γ is x)
  where
    map-γ : (is : List ⊤) → P is (μ (λ _ → E) Type) → P is (Inv !)
    map-γ [] _ = _
    map-γ (c :: is) (x , xs) = ok x , map-γ is xs
HetO' (σ S D) E (s , x) = ∇ s (HetO' (D s) E x)
HetO' (ṗ D) E (A , x) = Δ[ _ ∈ A ] ṗ (HetO' D E x)

HetO : (D : RDesc ⊤ ℓ-zero) → OrnDesc (μ (λ _ → D) Type tt) ! λ _ → D
HetO D (ok (con x)) = HetO' D D x

```

This ornament relates the original unindexed type to a type indexed over it; we see that this ornament largely keeps all fields and structure identical, only performing the necessary bookkeeping in the index, and adding extra fields before parameters.

As an example, we adapt the list description

```

ListD : Desc ⊤ ℓ-zero
ListD _ = σ Bool λ
  { false → γ []
  ; true  → ṗ (γ [ tt ]) }

List' : Type ℓ → Type ℓ
List' A = μ ListD A tt

```

which is easily heterogeneized to an `HList`. In fact, `HetO` seems to act functorially; if we lift `Maybe` like

```

MaybeD : Desc T ℓ-zero
MaybeD _ = σ Bool (λ
  { false → γ []
  ; true → γ̇ (γ []) })

Maybe : Type ℓ → Type ℓ
Maybe A = μ MaybeD A tt

HMaybeD = L HetO (MaybeD tt) J
HMaybe = μ HMaybeD T

then we can lift functions like head as
head : List' A → Maybe A
head (con (false , _)) = con (false , _)
head (con (true , a , _)) = con (true , a , _)

hhead : (As : List' Type) → HList As → HMaybe (head As)
hhead (con (false , _)) (con _) = con _
hhead (con (true , A , _)) (con (a , _)) = con (a , _ , _)

```

9 More equivalences for less effort

The setup some approaches in earlier sections require makes them tedious or impractical to apply. In this section we will look at some ways how part of this problem could be alleviated through generics, or by alternative descriptions of concepts like equivalences through the lens of initial algebras.

In later sections we will construct many more equivalences between more complicated types than before, so we will dive right into the latter. Reflecting upon Section 7, we see that when one establishes an equivalence, most of the time is spent working out a series of tedious lemmas to show that the conversion functions are mutual inverses, which tend to be relatively easy to define. We take away two things from this; the first is that the conversion functions are perhaps too obvious, and the second is that we should really avoid talking about sections and retractions lest we incur tedium!⁹ We will reuse the machinery of Ko and Gibbons [KG16] to illustrate how the definitions in Section 7 were actually forced for a large part.

First, we remark that μ is internalization of the representation of simple¹⁰ datatypes as W-types. Thus, we will assume that one of the sides of the equivalence is always represented as an initial algebra of a polynomial functor, and hence the μ of a `Desc'`.

9.1 Well-founded monic algebras are initial

Unfortunately, the machinery developed by Ko and Gibbons [KG16] relies on axiom K for a small but crucial part. To be precise, in a cubical setting, the

⁹The latter perhaps less so, because it is useful to show a map to be monic.

¹⁰Of course, indexed datatypes are indexed W-types, mutually recursive datatypes are represented yet differently...

type μ as given stops being initial for its base functor! In this section, we will be working with a simplified and repaired version. Namely, we simplify `Desc'` to

```
data Desc' : Set1 where
  γ : (n : ℕ) → Desc'
  σ : (S : Set) (D : S → Desc') → Desc'
```

To complete the definition of μ

```
data μ (D : Desc') : Set1 where
  con : Base (μ D) D → μ D
```

we will need to implement `Base`. We remark that in the original setup, the recursion of `mapFold` is a structural descent in $\llbracket D' \rrbracket (\mu D)$. Because $\llbracket _ \rrbracket$ is a type computing function and not a datatype, this descent becomes invalid¹¹, and `mapFold` fails the termination check. We resolve this by defining `Base` as a datatype

```
data Base (X : Set1) : Desc' → Set1 where
  in-γ : ∀ {n} → Vec X n → Base X (γ n)
  in-σ : ∀ {S D} → Σ[ s ∈ S ] (Base X (D s)) → Base X (σ S D)
```

such that this descent is allowed by the termination checker without axiom K.¹²

Recall that the `Base` functors of descriptions are special polynomial functors, and the fixpoint of a base functor is its initial algebra. We are looking for sufficient conditions on X to get the equivalence $e : X \cong \mu F$. Note that when $X \cong \mu F$, then there necessarily is an initial algebra $FX \rightarrow X$. Conversely, if the algebra (X, f) is isomorphic to $(\mu F, \text{con})$, then $X \cong \mu F$ would follow immediately, so it is equivalent to ask for the algebras to be isomorphic instead.

9.1.1 Datatypes as initial algebras

To characterize when such algebras are isomorphic, we reiterate some basic category theory, simultaneously rephrasing it in Agda terms.¹³

Let C be a category, and let a, b, c be objects of C , so that in particular we have identity arrows $1_a : a \rightarrow a$ and for arrows $g : b \rightarrow c, f : a \rightarrow b$ composite arrows $gf : a \rightarrow c$ subject to associativity. In our case, C is the category of types, with ordinary functions as arrows.

Recall that an endofunctor, which is simply a functor F from C to itself, assigns objects to objects and sends arrows to arrows

```
F0 : Type ℓ → Type ℓ
fmap : (A → B) → F0 A → F0 B
```

These assignments are subject to the identity and composition laws

```
f-id      : (x : F A)
            → mapF id x ≡ x
```

¹¹Refer to the without K page.

¹²This has, again by the absence of axiom K, the consequence of pushing the universe levels up by one. However, this is not too troublesome, as equivalences can go between two levels, and indeed types are equivalent to their lifts.

¹³We are not reusing a pre-existing category theory library for the simple reasons that it is not that much work to write out the machinery explicitly, and that such libraries tend to phrase initial objects in the correct way, which is too restrictive for us.

f-comp : $(g : B \rightarrow C) (f : A \rightarrow B) (x : F A)$
 $\rightarrow \text{mapF } (g \circ f) x = \text{mapF } g (\text{mapF } f x)$

An F -algebra is just a pair of an object a and an arrow $Fa \rightarrow a$

record Algebra ($F : \text{Type } \ell \rightarrow \text{Type } \ell$) : $\text{Type } (\ell\text{-suc } \ell)$ **where**
field

Carrier : $\text{Type } \ell$
forget : $F \text{ Carrier} \rightarrow \text{Carrier}$

Algebras themselves again form a category C^F . The arrows of C^F are the arrows f of C such that the following square commutes

$$\begin{array}{ccc} Fa & \xrightarrow{Ff} & Fb \\ U_a \downarrow & & \downarrow U_b \\ a & \xrightarrow{f} & b \end{array}$$

So we define

Alg→Sqr $F A B f = f \circ A.\text{forget} = B.\text{forget} \circ F.\text{fmap } f$

and

record Alg→ ($\text{RawF} : \text{RawFunctor } \ell$)
 $(\text{AlgA } \text{AlgB} : \text{Algebra } (\text{RawF } .F_0)) : \text{Type } \ell$ **where**
constructor alg→

field
mor : $\text{AlgA}.\text{Carrier} \rightarrow \text{AlgB}.\text{Carrier}$
coh : $\parallel \text{Alg→Sqr } \text{RawF } \text{AlgA } \text{AlgB } \text{mor} \parallel_1$

Note that we take the propositional truncation of the square, such that algebra maps with the same underlying morphism become propositionally equal

Alg→Path : $\{F : \text{RawFunctor } \ell\} \{A B : \text{Algebra } (F.F_0)\}$
 $\rightarrow (g f : \text{Alg→} F A B) \rightarrow g.\text{mor} = f.\text{mor} \rightarrow g = f$

The identity and composition in C^F arise directly from those of the underlying arrows in C .

Recall that an object \emptyset is initial when for each other object a , there is a unique arrow $! : \emptyset \rightarrow a$. By reversing the proofs of initiality of μ and the main result of this section, we obtain a slight variation upon the usual definition. Namely, unicity is often expressed as contractability of a type

isContr $A = \Sigma [x \in A] (\forall y \rightarrow x = y)$

Instead, we again use a truncation

weakContr $A = \Sigma [x \in A] (\forall y \rightarrow \parallel x = y \parallel_1)$

but note that this also, crucially, slightly stronger than connectedness. We define initiality for arbitrary relations

record Initial ($C : \text{Type } \ell$) ($R : C \rightarrow C \rightarrow \text{Type } \ell'$)
 $(Z : C) : \text{Type } (\ell\text{-max } (\ell\text{-suc } \ell) \ell')$ **where**
field

initiality : $\forall X \rightarrow \text{weakContr } (R Z X)$

such that it closely resembles the definition of least element. Then, A is an initial algebra when

$$\text{InitAlg } \text{RawF } A = \text{Initial } (\text{Algebra } (\text{RawF } .F_0)) (\text{Alg} \rightarrow \text{RawF}) A$$

By basic category theory (using the usual definition of initial objects), two initial objects a and b are always isomorphic; namely, initiality guarantees that there are arrows $f : a \rightarrow b$ and $g : b \rightarrow a$, which by initiality must compose to the identities again.

Similarly, we get that

$$\begin{aligned} \text{InitAlg-}\simeq : (F : \text{Functor } \ell) (A B : \text{Algebra } (F .\text{RawF } .F_0)) \\ \rightarrow \text{InitAlg } (F .\text{RawF}) A \rightarrow \text{InitAlg } (F .\text{RawF}) B \\ \rightarrow A .\text{Carrier} \simeq B .\text{Carrier} \end{aligned}$$

However, we only have the equalities from the isomorphism inside a propositional truncation. But fortunately, being an equivalence is a property, so we can eliminate from the truncations to get the wanted result.

9.1.2 Accessibility

As a consequence, we get that X is isomorphic to μD when X is an initial algebra for the base functor of D ; μD is initial by its fold, and by induction on μD using the squares of algebra maps.

Remark 9.1. We need (in general) not hope μD is a strict initial object in the category of algebras. For a strict initial object, having a map $a \rightarrow \emptyset$ implies $a \cong \emptyset$. This is not the case here: strict initial objects satisfy $a \times \emptyset \cong \emptyset$, but for the $X \mapsto 1 + X$ -algebras \mathbb{N} and $2^{\mathbb{N}}$ clearly $2^{\mathbb{N}} \times \mathbb{N} \cong \mathbb{N}$ does not hold. On the other hand, the “obvious” sufficient condition to let C^F have strict initial objects is that F is a left adjoint, but then the carrier of the initial algebra is simply \perp .

Looking back at Section 7, we see that **Leibniz** is an initial $F : X \mapsto 1 + X$ algebra because for any other algebra, the image of **0b** is fixed, and by **bsuc** all other values are determined by chasing around the square. Thus, we are looking for a similar structure on $f : FX \rightarrow X$ that supports recursion.

Clearly we will need something stronger than $FX \cong X$, as in general a functor can have many fixpoints. For this, we define what it means for an element x to be accessible by f . This definition uses a mutually recursive datatype as follows: We state that an element x of X is accessible when there is an accessible y in its fiber over f

$$\begin{aligned} \text{data Acc } D f x \text{ where} \\ \text{acc} : (y : \text{fiber } f x) \rightarrow \text{Acc}' D f D (\text{fst } y) \rightarrow \text{Acc } D f x \end{aligned}$$

Accessibility of an element x of **Base A E** is defined by cases on E ; if E is **y n** and x is a **Vec A n**, then x is accessible if all its elements are; if x is **σ S E'**, then x is accessible if **snd x** is

$$\begin{aligned} \text{data Acc}' D f \text{ where} \\ \text{acc-y} : \text{All } (\text{Acc } D f) x \rightarrow \text{Acc}' D f (\text{y } n) (\text{in-y } x) \\ \text{acc-}\sigma : \text{Acc}' D f (E s) x \rightarrow \text{Acc}' D f (\sigma S E) (\text{in-}\sigma (s, x)) \end{aligned}$$

Consequently, X is well-founded for an algebra when all its elements are accessible

$$\text{Wf } D f = \forall x \rightarrow \text{Acc } D f x$$

We can see well-foundedness as an upper bound on the size of X , if it were larger than μD , some of its elements would inevitably get out of reach of an algebra. *Now* having $FX \cong X$ also gives us a lower bound, but remark that having a well-founded injection $f : FX \rightarrow X$ is already sufficient, as accessibility gives a section of f , making it an iso. This leads us to claim

Claim 9.1. If there is a mono $f : FX \rightarrow X$ and X is well-founded for f , then X is an initial F -algebra.

9.1.3 Proof sketch of Claim 9.1

Let us be on our way. Suppose X is well-founded for the mono $f : FX \rightarrow X$. To show that (X, f) is initial, let us take another algebra (Y, g) , and show that there is a unique arrow $(X, f) \rightarrow (Y, g)$.

This section is about as digestable as a brick.

By **Acc**-recursion and because all x are accessible, we can define a plain map into Y

$$\begin{aligned} \text{Wf-rec} : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X \text{.forget}) \\ &\rightarrow (\dot{F} D A \rightarrow A) \rightarrow X \text{.Carrier} \rightarrow A \end{aligned}$$

This construction is an instance of the concept of “well-founded recursion”¹⁴, so we let ourselves be inspired by these methods. In particular, we prove an irrelevance lemma

$$\text{Wf-rec-irrelevant} : \forall x' y' x a b \rightarrow \text{rec } x' x a = \text{rec } y' x b$$

which implies the unfolding lemma

$$\begin{aligned} \text{unfold-Wf-rec} : \forall x' &\rightarrow \text{rec } (cx \ x') (cx \ x') (\text{wf } (cx \ x')) \\ &= f (\text{Base-map } (\lambda y \rightarrow \text{rec } y y (\text{wf } y)) x') \end{aligned}$$

The unfolding lemma ensures that the map we defined by **Wf-rec** is a map of algebras. The proof that this map is unique proceeds analogously to that in the proof that μD is initial, but here we instead use **Acc**-recursion

$$\begin{aligned} \text{Wf+inj} \rightarrow \text{Init} : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X \text{.forget}) \\ &\rightarrow \text{injective } (X \text{.forget}) \rightarrow \text{InitAlg } (\text{Raw } \dot{F} D) X \end{aligned}$$

Thus, we conclude that X is initial. The main result is then a corollary of initiality of X and the isomorphism of initial objects

$$\begin{aligned} \text{Wf+inj} = \mu : (D : \text{Desc}') (X : \text{Algebra } (\dot{F} D)) &\rightarrow \text{Wf } D (X \text{.forget}) \\ &\rightarrow \text{injective } (X \text{.forget}) \rightarrow X \text{.Carrier} = \mu D \end{aligned}$$

9.1.4 Example

Let us redo the proof in Section 7, now using this result. Recall the description of naturals **NatD**. To show that **Leibniz** is isomorphic to **Nat**, we will need a **NatD**-algebra and a proof of its well-foundedness. We define the algebra

$$\begin{aligned} \text{bsuc}' : \text{Base Leibniz}_1 \text{ NatD} &\rightarrow \text{Leibniz}_1 \\ \text{bsuc}' \text{ zero} &= 0b_1 \\ \text{bsuc}' (\text{succ } n) &= \text{bsuc}_1 n \end{aligned}$$

¹⁴This is formalized in the standard-library with many other examples.

```

L-Alg : Algebra (F NatD)
L-Alg .Carrier = Leibniz1
L-Alg .forget = bsuc'

```

For well-foundedness, we use something similar to view-patterns (the main difference being that we look through the entire structure, instead of a single layer)

```

data Peano-View : Leibniz1 → Type, where
  as-zero : Peano-View 0b1
  as-suc : (n : Leibniz1) (v : Peano-View n) → Peano-View (bsuc1 n)

view-1b : ∀ {n} → Peano-View n → Peano-View (n 1b1)
view-2b : ∀ {n} → Peano-View n → Peano-View (n 2b1)
view : (n : Leibniz1) → Peano-View n

```

where the mutually recursive proof of `view` is “almost trivial”. Well-foundedness follows fairly immediately

```

view→Acc : ∀ {n} → Peano-View n → Acc NatD bsuc' n
Wf-bsuc : Wf NatD bsuc'
Wf-bsuc n = view→Acc (view n)

```

Injectivity of `bsuc1` happens to be harder to prove from retractions than directly, so we prove it directly, from which the wanted statement follows

```

L≅μN : Leibniz1 ≅ μ NatD
L≅μN = Wf+inj≅μ NatD L-Alg Wf-bsuc λ x y p → inj-bsuc x y p

```

Note that in this case it took us more code to prove the same statement! However, we stress that the code that we did write became more forced, and might be more amenable to automation.

References

- [Ang+20] Carlo Angiuli et al. *Internalizing Representation Independence with Univalence*. 2020. DOI: 10.48550/ARXIV.2009.05547. URL: <https://arxiv.org/abs/2009.05547>.
- [HP06] RALF HINZE and ROSS PATERSON. “Finger trees: a simple general-purpose data structure”. In: *Journal of Functional Programming* 16.2 (2006), pp. 197–217. DOI: 10.1017/S0956796805005769.
- [HS22] Ralf Hinze and Wouter Swierstra. “Calculating Datastructures”. In: *Mathematics of Program Construction*. Ed. by Ekaterina Komen-dantskaya. Cham: Springer International Publishing, 2022, pp. 62–101. ISBN: 978-3-031-16912-0.
- [KG16] HSIANG-SHANG KO and JEREMY GIBBONS. “Programming with ornaments”. In: *Journal of Functional Programming* 27 (2016), e2. DOI: 10.1017/S0956796816000307.
- [Oka98] Chris Okasaki. *Purely Functional Data Structures*. USA: Cambridge University Press, 1998. ISBN: 0521631246.

- [SWI20] WOUTER SWIERSTRA. “Heterogeneous binary random-access lists”. In: *Journal of Functional Programming* 30 (2020), e10. DOI: 10.1017/S0956796820000064.
- [Tea23] Agda Development Team. *Agda*. 2023. URL: <https://agda.readthedocs.io/en/v2.6.3/>.
- [VMA19] Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. “Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types”. In: *Proc. ACM Program. Lang.* 3.ICFP (July 2019). DOI: 10.1145/3341691. URL: <https://doi.org/10.1145/3341691>.