Restoring (part of) the friendship between recursion schemes and without-K (provisional) $_{\mbox{\sc l'll have to grab a UU-template at some point}$

$\begin{array}{c} \text{Samuel Klumpers} \\ 6057314 \end{array}$

March 14, 2023

Contents

1	Introduction	2
2	How Cubical Agda helps our binary numbers (ready) 2.1 Binary numbers	3 4 5 7
3	Specifying types (ready) 3.1 Numerical representations: from numbers to containers 3.2 Numerical representations as ornaments 3.3 Folding in	8 8 10 12
4	Reducing friction (work in progress) 4.1 Well-founded monic algebras are initial	13 13 14 16 17 17
5	Is equivalence too strong (finger trees)	17
6	Discussion and future work (aka the union of my to-do list and the actual future work section)	17
7	Temporary This document is generated from a literate agda file!	18

Abstract

The preliminary goal of this thesis is to introduce, among others, the concepts of the structure identity principle, numerical representations, and ornamentations, which are then combined to simplify the presentation and verification of finger trees, as a demonstration of the generalizability and improved compactness and security of the resulting code.

1 Introduction

Most of the time when we are Agda-ing [Tea23] we are trying to un-Haskell ourselves, e.g., not take the head of an empty list. In this example, we can make head total by switching to length-indexed lists: vectors. We have now effectively doubled the size of our code base, since functions like _++_ which we had for lists, will also have to be reimplemented for vectors.

To make things worse; often, after coping with the overloaded names resulting from Agda-ing by shoving them into a different namespace, we also find out that lists nor vectors are efficient containers to begin with. Maybe binary trees are better. We now need four times the number of definitions to keep everything working, and, if we start proving things, we will also have to prove everything fourfold. (Not to mention that reasoning about trees is probably going to be harder than reasoning about lists). This inefficiency has sparked (my) interest in ways to deal with the situation.

Following [DM14] and [KG16], we can describe the relation between list and vector using the mechanism of ornamentation. This leads them to define the concept of patches, which can aid us when defining _++_ for the second time by forcing the new version to be coherent. In fact, the algebraic nature of ornaments can even get us the definition of the vector type for free, if we started by defining lists relative to natural numbers [McB14]. Such constructions rely heavily on descriptions of datastructures and often come with limitations in their expressiveness. These descriptions in turn impose additional ballast on the programmer, leading us to investigate reflection like in [EC22] as a means to bring datatypes and descriptions closer when possible.

From a different direction, [HS22] gives methods by which we can show two implementations of some structure to be equivalent. With this, we can simply transport all proofs about _++_ we have for lists over to the implementation for trees, provided that we show them to be equivalent as appendable containers. This process can also be automated by some heavy generics, but instead, we resort to cubical; which hosts a range of research like [Ang+20] tailored to the problem describing equivalences of structures.

We can liken the situation to movement on a plane, where ornamentation moves us vertically by modifying constructors or indices, and structured equivalences move us horizontally to and from equivalent but more equivalent implementations. In this paper, we will investigate a variety of means of moving around structures and proofs, and ways to make this more efficient or less intrusive.

Currently, all sections mainly reintroduce or reformulate existing research, with some spots of new ideas and original examples here and there. In section 2, we will look at how proofs on unary naturals can be moved to binary naturals. Then in section 3 we recall how numeral systems in particular induce container types, which we attempt to reformulate in the language of ornaments in subsection 3.2.

2 How Cubical Agda helps our binary numbers (ready)

Let us quickly review the small set of features in Cubical Agda that we will be using extensively throughout this article.¹ We note that there are some downsides to cubical, such as that

```
{-# OPTIONS --cubical #-}
```

also implies the negation of axiom K, which in turn complicates both some termination checking and some universe levels. And, more obviously, we get saddled with the proof obligation that our types are sets should we use certain constructions.

Of course, this downside is more than offset by the benefits of changing our primitive notion of equality, which among other things, lets us access univalence, which drastically cut down the investment required to both show more complex structures to be equivalent (at least when compared to non-cubical). Here, equality arises not (directly) from the indexed inductive definition we are used to, but rather from the presence of the interval type I. This type represents a set of two points i0 and i1, which are considered "identified" in the sense that they are connected by a path. To define a function out of this type, we also have to define the function on all the intermediate points, which is why we call such a function a "path". Terms of other types are then considered identified when there is a path between them.

As an added benefit, this different perspective gives intuitive interpretations to some proofs of equality, like

```
sym : x = y \rightarrow y = x
sym p i = p (\sim i)
```

where \sim is the interval reversal, swapping i0 and i1, so that sym simply reverses the given path.

Furthermore, because we can now interpret paths in records and function differently, we get a host of "extensionality" for free. For example, a path in $A \to B$ is indeed a function which takes each i in I to a function $A \to B$. Using this, function extensionality becomes tautological

```
funExt : (\forall x \rightarrow f x = g x) \rightarrow f = g
funExt p i x = p x i
```

Finally, the Glue type tells us that equivalent types fit together in a new type, in a way that guarantees univalence

Not sure if it would be helpful to have a more extensive introduction covering all features used.

To be precise, the mapFold in [KG16] gets painted red.

¹[VMA19] gives a comprehensive introduction to cubical agda.

```
ua : \forall \{A \ B : \mathsf{Type} \ \ell\} \longrightarrow A \cong B \longrightarrow A \equiv B
```

For our purposes, we can interpret univalence as "equivalent types are identified", and, we can treat equivalences as the "HoTT-compatible" generalization of bijections. In particular, type isomorphisms like $1 \to A \simeq A$ actually become paths $1 \to A \equiv A$, such that we can transport proofs along them. We will demonstrate this by a slightly more practical example.

2.1 Binary numbers

Let us motivate the cubical method by showing the equivalence of the "Peano" naturals and the "Leibniz" naturals. Recall that the Peano naturals are defined as

```
data N : Type where
zero : N
suc : N → N
```

This definition enjoys a simple induction principle and has many proofs of its properties in standard libraries. However, it is too slow to be of practical use: most arithmetic operations defined on \mathbb{N} have time complexity in the order of the value of the result.

Of course, the alternative are the more performant binary numbers: the time complexities for binary numbers are usually logarithmic in the resultant values, but these are typically less well-covered in terms of proofs. This does not have to be a problem, because the $\mathbb N$ naturals and the binary numbers should be equivalent after all!

Let us make this formal. We define the Leibniz naturals as follows:

```
data Leibniz: Set where

0b: Leibniz

_1b: Leibniz → Leibniz

2b: Leibniz → Leibniz
```

Here, the 0b constructor encodes 0, while the _1b and _2b constructors respectively add a 1 and a 2 bit, under the usual interpretation of binary numbers:

```
is this too
much code
and too
little ex-
planation
at once?
```

```
to \mathbb{N}: Leibniz \rightarrow \mathbb{N}

to \mathbb{N} 0b = 0

to \mathbb{N} (n 1b) = 1 N.+ 2 N.· to \mathbb{N} n

to \mathbb{N} (n 2b) = 2 N.+ 2 N.· to \mathbb{N} n
```

Let us construct the equivalence from \mathbb{N} to Leibniz. First, we can also interpret a number in \mathbb{N} as a binary number by repeating the successor operation on binary numbers:

```
bsuc : Leibniz \rightarrow Leibniz
bsuc 0b = 0b \ 1b
bsuc (n \ 1b) = n \ 2b
bsuc (n \ 2b) = (bsuc \ n) \ 1b
from \mathbb{N} : \mathbb{N} \rightarrow \text{Leibniz}
from \mathbb{N} : \mathbb{N} , \text{zero} = 0b
```

```
from \mathbb{N} (\mathbb{N}.suc n) = bsuc (from \mathbb{N} n)
To show that the operations are inverses, we observe that the interpretation
respects successors
        toN-suc : \forall x \rightarrow toN (bsuc x) \equiv N.suc (toN x)
        toN-suc 0b = refl
        toN-suc (x 1b) = refl
        toN-suc (x 2b) = cong
           (\lambda k \rightarrow (1 \text{ N.+ 2 N.} \cdot k))
           (toN-suc\ x) \cdot cong\ N.suc\ (NP.-suc\ 2\ (toN\ x))
and that the inverse respects even and odd numbers
        from \mathbb{N}-1+2·: \forall x \rightarrow \text{from } \mathbb{N} \text{ (1 N.+ 2 N.· } x) = (\text{from } \mathbb{N} \text{ x) 1b}
        from \mathbb{N}-1+2 \cdot \mathbb{N}.zero = refl
        from \mathbb{N}-1+2· (\mathbb{N}.suc x) = cong
           (bsuc • bsuc)
           (cong from \mathbb{N} (NP.+-suc x (x N.+ \mathbb{N}.zero)) · from \mathbb{N}-1+2· x)
        from \mathbb{N}-2+2·: \forall x \rightarrow \text{from } \mathbb{N} (2 \text{ N.+ } 2 \text{ N.· } x) = (\text{from } \mathbb{N} x) 2b
        from \mathbb{N}-2+2· x = \text{cong bsuc (from } \mathbb{N}-1+2· x)
The wanted statement follows
        N↔L: Iso N Leibniz
        \mathbb{N} \leftrightarrow L = iso \text{ from} \mathbb{N} \text{ to} \mathbb{N} \text{ sec ret}
           where
           sec: section from N to N
           sec 0b = refl
           sec (n 1b) = from \mathbb{N} - 1 + 2 \cdot (to \mathbb{N} n) \cdot cong \_1b (sec n)
           sec (n \ 2b) = from \mathbb{N} - 2 + 2 \cdot (to \mathbb{N} \ n) \cdot cong \ 2b (sec \ n)
           ret: retract from N to N
           ret N.zero = refl
           ret (\mathbb{N}.suc n) = to\mathbb{N}-suc (from\mathbb{N} n) · cong \mathbb{N}.suc (ret n)
but since we now have a bijection, we also get an equivalence
        \mathbb{N} \simeq \mathbb{L} : \mathbb{N} \simeq \text{Leibniz}
        \mathbb{N} \simeq L = isoToEquiv \mathbb{N} \leftrightarrow L
Finally, by univalence, we can identify N and Leibniz naturals
        \mathbb{N} = \mathbb{L} : \mathbb{N} = \text{Leibniz}
        N=L = ua N \simeq L
     Using the path N=L we can already prove some otherwise difficult properties,
e.g.,
        isSetL: isSet Leibniz
        isSetL = subst isSet N=L N.isSetN
Let us define an operation on Leibniz and demonstrate how we can also transport
```

2.2 Use as definition: functions from specifications

proofs about operations from N to Leibniz.

As an example, we will define addition of binary numbers. We could take

```
BinOp : Type \rightarrow Type
BinOp A = A \rightarrow A \rightarrow A
_+'_ : BinOp Leibniz
_+'_ = subst BinOp \mathbb{N}=L N._+_
```

But this would be rather inefficient, incurring an O(n+m) overhead when adding n+m, so we could better define addition directly. We would prefer to give a definition which makes use of the binary nature of Leibniz, while agreeing with the addition on \mathbb{N} .

Such a definition can be derived from the specification "agrees with _+_", so we implement the following syntax for giving definitions by equational reasoning, inspired by the "use-as-definition" notation from [HS22]:

```
Def : \{X: \mathsf{Type}\ a\} \to X \to \mathsf{Type}\ a

Def \{X = X\}\ x = \Sigma'\ X\ \lambda\ y \to x \equiv y

defined-by : \{X: \mathsf{Type}\ a\}\ \{x: X\} \to \mathsf{Def}\ x \to X

defined-by = fst

by-definition : \{X: \mathsf{Type}\ a\}\ \{x: X\} \to (d: \mathsf{Def}\ x) \to x \equiv \mathsf{defined-by}\ d

by-definition = snd
```

which infers the definition from the right endpoint of a path using an implicit pair type

```
record \Sigma' (A: Set a) (B: A \to Set b): Set (\ell-max a b) where constructor _use-as-def field {fst}: A snd: B fst open \Sigma' infix 1_use-as-def
```

As of now, I am unsure if this reduces the effort of implementing a coherent function, or whether it is more typically possible to give a smarter or shorter proof by just giving a definition and proving an easier property of it²

With this we can define addition on Leibniz and show it agrees with addition on $\mathbb N$ in one motion

```
{-# TERMINATING #-}
plus-def: \forall x \ y \rightarrow \text{Def (from} \mathbb{N} \text{ (to} \mathbb{N} \ x \ \text{N.+ to} \mathbb{N} \ y))}
plus-def 0b y = \mathbb{N} \leftrightarrow \mathbb{L} \text{ .rightlnv } y \text{ use-as-def}
plus-def (x \text{ 1b}) 0b =
bsuc (from \mathbb{N} \text{ (to} \mathbb{N} \ x \ \mathbb{N.+ \mathbb{N.zero}) \mathbb{N.+ \mathbb{N.zero})} \mathbb{N} \text{.rightlnv } y \text{ use-as-def}

| \text{suc (from} \mathbb{N} \text{ (to} \mathbb{N} \ x \ \mathbb{N.+ \mathbb{N.zero}) \mathbb{N.+ \mathbb{N.zero})}) \mathbb{N} \mathbb{N} \text{ bsuc (from} \mathbb{N} \text{ (to} \mathbb{N} \ x \ \mathbb{N.+ \mathbb{N.zero}))}
```

 $^{^{2}\}mathrm{I}$ will put the alternative in the appendix for now

```
\equiv \langle \text{ from} \mathbb{N} - 1 + 2 \cdot (\text{to} \mathbb{N} \ x) \rangle
    from \mathbb{N} (to \mathbb{N} x) 1b
        \equiv \langle \text{ cong } \_1b \ (\mathbb{N} \leftrightarrow \mathbb{L} \text{ .rightInv } x) \rangle
    x 1b ■ use-as-def
plus-def (x \ 1b) (y \ 1b) =
    from \mathbb{N} ((1 N.+ 2 N.+ to \mathbb{N} x) N.+ (1 N.+ 2 N.+ to \mathbb{N} y))
        \equiv \langle \text{ cong from} \mathbb{N} \text{ (Eq.eqToPath (eq (to} \mathbb{N} x) \text{ (to} \mathbb{N} y))) \rangle
    from \mathbb{N} (2 N.+ (2 N.+ (to \mathbb{N} x N.+ to \mathbb{N} y)))
        \equiv \langle \text{ from} \mathbb{N} - 2 + 2 \cdot (\text{to} \mathbb{N} \times \mathbb{N} + \text{to} \mathbb{N} \times \mathbb{N}) \rangle
    from \mathbb{N} (to \mathbb{N} x \mathbb{N}.+ to \mathbb{N} y) 2b
        \equiv \langle \text{ cong } \_2b \text{ (by-definition (plus-def } x \text{ } y)) \rangle
    defined-by (plus-def x y) 2b use-as-def
        where
        eq: \forall x \nu
              \rightarrow (1 N.+ 2 N.· x) N.+ (1 N.+ 2 N.· y) Eq.≡ 2 N.+ (2 N.· (x N.+ y))
        eq = NS.solve-∀
-- similar clauses omitted
plus : \forall x y \rightarrow Leibniz
plus x y = \text{defined-by (plus-def } x y)
plus-coherent : \forall x \ y \rightarrow \text{from} \mathbb{N} \ (x \ \mathbb{N} + y) = \text{plus} \ (\text{from} \mathbb{N} \ x) \ (\text{from} \mathbb{N} \ y)
plus-coherent x y = \text{cong from} \mathbb{N}
    (cong_2 \ N._+ \ (sym \ (\mathbb{N} \leftrightarrow L \ .leftInv \ x)) \ (sym \ (\mathbb{N} \leftrightarrow L \ .leftInv \ \_))) \cdot
        by-definition (plus-def (from \mathbb{N} x) (from \mathbb{N} y))
```

2.3 Structure Identity Principle

We see that as a consequence (modulo some PathP lemmas), we get a path from $(\mathbb{N}, \mathbb{N}.+)$ to (Leibniz, plus). More generally, we can view a type X combined with a function $f: X \to X \to X$ as a kind of structure, which in fact coincides with a magma. We can see that paths between magmas correspond to paths between the underlying types X and paths over this between their operations f. This observation is further generalized by the Structure Identity Principle (SIP), formalized in [Ang+20]. Given a structure, which in our case is just a binary operation

```
MagmaStr : Type \rightarrow Type
MagmaStr A = A \rightarrow A \rightarrow A
```

this principle produces an appropriate definition "structured equivalence" ι . The ι is such that if structures X,Y are ι -equivalent, then they are identified. In this case, the ι asks us to provide plus-coherent, so we have just shown that the plus magma on Leibniz

```
MagmaL : Magma
fst MagmaL = Leibniz
snd MagmaL = plus
and the _+_ magma on N and are identical
```

Replace with BinOp

```
MagmaN≃MagmaL : MagmaN = MagmaL

MagmaN≃MagmaL = equivFun (MagmaΣPath _ _) proof

where

proof : MagmaN ≃[ MagmaEquivStr ] MagmaL

fst proof = N≃L

snd proof = plus-coherent

As a consequence, properties of _+_ directly yield corresponding properties of plus. For example,

plus-assoc : Associative _=_ plus

plus-assoc = subst

(λ A → Associative _=_ (snd A))

MagmaN≃MagmaL

N-assoc
```

3 Specifying types (ready)

While the practical applications of the last example do not stretch very far³, the approach generalizes to the more relevant containers and their associated laws.

In the same vein as the last section, we could define a simple but inefficient array type, and a more efficient implementation using trees. Then we can show that these are equivalent, such that when the simple type satisfies a set of laws, trees will satisfy them as well. We could then start developing all sorts of complex implementations fine-tuned to each operation and figure out how these are equivalent to some simpler type, but let us first take a step back, and investigate how we can make this approach run smoothly in a simpler example.

Rather than inductively defining a container and then showing that it is represented by a lookup function, we can go the other way around and define a type by insisting that it is equivalent to such a function. This approach, in particular the case in which one calculates a container with the same shape as a numeral system, was dubbed numerical representations in [Oka98], and has some formalized examples in, e.g., [HS22] and [KG16]. Numerical representations form the starting point for defining more complex datastructures based on simpler ones, so let us demonstrate such a calculation.

3.1 Numerical representations: from numbers to containers

We can compute the type of vectors starting from \mathbb{N} .⁴ For simplicity, we define them as a type computing function via the "use-as-definition" notation from before. We expect vectors to be represented by

 $^{^3}$ Considering that $\mathbb N$ is a candidate to be replaced by a more suitable unsigned integer type when compiling to Haskell anyway.

⁴This is adapted (and fairly abridged) from [HS22]

```
Lookup : Type \rightarrow \mathbb{N} \rightarrow Type
Lookup A n = Fin n \rightarrow A
```

where we use the finite type Fin as an index into vector. Using this representation as a specification, we can compute both Fin and a type of vectors. The finite type can be computed from the evident definition

```
Fin-def : \forall n \rightarrow \text{Def} (\Sigma [m \in \mathbb{N}] m < n)
        Fin-def zero = \perp-strict (\lambda ()) use-as-def
        Fin-def (suc n) =
           ua (<-split n) ·
           cong (\top \ \ \ \ \ \ \ ) (by-definition (Fin-def n)) use-as-def
        Fin : \mathbb{N} \to \mathsf{Type}
        Fin n = \text{defined-by (Fin-def } n)
using
        <-split : \forall n → (Σ[ m ∈ ℕ ] m < suc n) ≃ (⊤ ⊎ (Σ[ m ∈ ℕ ] m < n))
Likewise, vectors can be computed by applying a sequence of type isomorphisms
        Vec-def : \forall A \ n \rightarrow Def (Lookup A \ n)
        Vec-def A zero = isContr \rightarrow = Unit isContr \perp \rightarrow A use-as-def
        Vec-def A (suc n) =
           ((\top \ \forall \ \mathsf{Fin} \ n) \to A)
             ≡⟨ua Π⊎≃⟩
           (\top \to A) \times (\mathsf{Fin} \ n \to A)
             ≡⟨ cong<sub>2</sub> _×_
                 (UnitToTypePath A)
                 (by-definition (Vec-def A n)) \rangle
           A \times (\text{defined-by (Vec-def } A \ n))  use-as-def
        Vec : \forall A \ n \rightarrow \mathsf{Type}
        Vec A n = defined-by (Vec-def A n)
```

SIP doesn't mesh very well with indexed stuff, does HSIP help?

Of course, a container would not be of much use without lookup functions, so we define an interface

```
record Array (V: \mathsf{Type} \to \mathbb{N} \to \mathsf{Type}): \mathsf{Type_1} where field \mathsf{lookup}: \forall \{A \ n\} \to V \ A \ n \to \mathsf{Fin} \ n \to A \mathsf{tail}: \forall \{A \ n\} \to V \ A \ (\mathsf{suc} \ n) \to V \ A \ n which at the very least has to satisfy laws like \mathsf{record} ArrayLaws \{C\} (Arr: \mathsf{Array} \ C): \mathsf{Type_1} where field \mathsf{lookup} \cdot \mathsf{tail}: \forall \{A \ n\} \ (xs: C \ A \ (\mathsf{suc} \ n)) \ (i: \mathsf{Fin} \ n) \to \mathsf{Arr}. \mathsf{lookup} \ (\mathsf{Arr} \cdot \mathsf{tail} \ xs) \ i \equiv \mathsf{Arr}. \mathsf{lookup} \ xs \ (\mathsf{inr} \ i)
```

We could directly show that Vec satisfies this, but now that we defined Vec from Lookup we might as well use this fact.

The implementation of arrays as functions is very straightforward

```
FunArray : Array Lookup
FunArray .lookup f = f
FunArray .tail f = f \circ inr
and clearly satisfies our interface
FunLaw : ArrayLaws FunArray
FunLaw .lookup \circ tail \_ = refl

We can implement arrays based on Vec as well
VectorArray : Array Vec
VectorArray .lookup \{n = n\} = f n
where
f : \forall \{A\} \ n \rightarrow \text{Vec } A \ n \rightarrow \text{Fin } n \rightarrow A
f (\text{suc } n) \ (x \ , xs) \ (\text{inf } i) = x
f (\text{suc } n) \ (x \ , xs) \ (\text{inf } i) = f \ n \ xs \ i
VectorArray .tail (x \ , xs) = xs
and again, we can transport the proofs from Lookup to Vec. ^5
```

As you can see, taking "use-as-definition" too literally prevents Agda from solving a lot of metavariables.

This computation can of course be generalized to any arity zeroless numeral system; unfortunately beyond this set of base types, this "straightforward" computation from numeral system to container loses its efficacy. In a sense, the n-ary natural numbers are exactly the base types for which the required steps are convenient type equivalences like $(A+B) \to C = (A \to C) \times (B \to C)$?

If one was determined to cobble together the path over path we need now.

3.2 Numerical representations as ornaments

We could perform the same computation for Leibniz, which would yield the type of binary trees, but we note that these computations proceed with roughly the same pattern: each constructor of the numeral system gets assigned a value, and is amended with a field holding a number of elements and subnodes using this value as a "weight". But wait! Such modifications of constructors are already made formal by the concept of ornamentation!

Ornamentation, as exposed in [McB14] and [KG16], lets us formulate what it means for two types to have a "similar" recursive structure. This is achieved by interpreting (indexed inductive) datatypes from descriptions, between which an ornament is seen as a certificate of similarity, describing which fields or indices need to be introduced or dropped. Furthermore, a one-sided ornament: an ornamental description, lets us describe new datatypes by recording the modifications to an existing description.

This links back to the construction in the previous section, since \mathbb{N} and Vec share the same recursive structure, so Vec can be formed by introducing indices

It seems like Agda forgets that we defined Leibniz if we move between different .tex files. so I'll have to float all AgdaTargets to the top level at some point...

Again not sure how much space I

⁵Except in this oversimplified case the laws are trivial for Vec as well.

and adding a field holding an element at each node.⁶

However, instead deriving List from \mathbb{N} generalizes to Leibniz with less notational overhead, so lets tackle that case first. For this, we have to give a description of \mathbb{N} to work with

```
NatD : Desc \top
NatD _ = \sigma Bool \lambda
{ false \rightarrow y []
; true \rightarrow y [ tt ] }
```

Recall that σ adds a field, upon which the rest of the description may vary, and ν lists the recursive fields and their indices (which can only be tt). We can now write down the ornament which adds fields to the suc constructor

```
NatD-ListO : Type \rightarrow OrnDesc \top ! NatD NatD-ListO A (ok _) = \sigma Bool \lambda { false \rightarrow \gamma _ ; true \rightarrow \Delta A (\lambda _ \rightarrow \gamma (ok _ , _)) }
```

Here, the σ and γ are forced to match those of NatD, but the Δ adds a new field. With the least fixpoint and description extraction from [KG16], this is sufficient to define List. Note that we cannot hope to give an unindexed ornament from Leibniz

```
LeibnizD : Desc \top

LeibnizD \_ = \sigma (Fin 3) \lambda

{ zero \rightarrow v []

; (suc zero) \rightarrow v [ tt ]

; (suc (suc zero)) \rightarrow v [ tt ] }
```

into trees, since trees have a very different recursive structure! Instead, we must keep track at what level we are in the tree so that we can ask for adequately many elements:

```
power : \mathbb{N} \to (A \to A) \to A \to A
power \mathbb{N}.\mathsf{zero}\ f = \lambda\ x \to x
power (\mathbb{N}.\mathsf{suc}\ n)\ f = f \circ \mathsf{power}\ n\ f

Two : Type \to Type
Two X = X \times X

LeibnizD-TreeO : Type \to OrnDesc \mathbb{N} ! LeibnizD

LeibnizD-TreeO A\ (\mathsf{ok}\ n) = \sigma\ (\mathsf{Fin}\ 3)\ \lambda
\{\mathsf{zero}\ \to \mathsf{v}\ \_\ ;\ (\mathsf{suc}\ \mathsf{zero}) \to \Delta\ (\mathsf{power}\ n\ \mathsf{Two}\ A)\ \lambda\ \_\to \mathsf{v}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ ,\ \_)
\{\mathsf{suc}\ (\mathsf{suc}\ \mathsf{zero})) \to \Delta\ (\mathsf{power}\ (\mathsf{suc}\ n)\ \mathsf{Two}\ A)\ \lambda\ \_\to \mathsf{v}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ ,\ \_)\ \}
```

We use the power combinator to ensure that the digit at position n, which has weight 2^n in the interpretation of a binary number, also holds its value times 2^n elements. This makes sure that the number of elements in the tree shaped after a given binary number also is the value of that binary number.

This "folding in" technique using the indices to keep track of structure seems to apply more generally. Let us explore this a bit further, and return later to

Clearly this can use more explanation (the question is, how much?)

 $^{^6\}mathrm{These}$ and similar examples are also documented in [KG16]

the generalization of the pattern from numeral systems to datastructures.

3.3 Folding in

Let us describe this procedure of folding a complex recursive structure into a simpler structure more generally. In particular, we will demonstrate that for linear datatypes, such as N and Leibniz, and for a given unindexed datatype, there is always an indexed datatype isomorphic to it at some index, and an ornament from the linear type to the indexed type.

Suppose we are given a description, the first thing we can do to simplify it is collect all fields in one place

```
RField : RDesc \top \to \mathsf{Type}
RField (\forall is) = \top
RField (\sigma S D) = \Sigma S \lambda s \to \mathsf{RField}(D s)
```

Next, we will certainly have to count the number of recursive occurrences we are tracking, so we define

```
-- note to self, I should probably make Y _not_ overlap
-- so not everything links here
data Number : Type where
1 : Number
v : ∀ n → (Fin n → Number) → Number
```

where $\mathbb{1}$ records that we are at the top level, and y denotes that we are below a constructor with some number of recursive fields. This simplifies our task to implementing the types in

```
nested: Desc \top \rightarrow Desc Number nested d n = \sigma (Fields (d tt) n) \lambda a \rightarrow \gamma [ subnodes a ] such a way that we get an isomorphism nested-eq: \forall D \rightarrow \mu D tt \cong \mu (nested D) \mathbb{1} Thus, Fields is forced to have a leaf constructor like data Fields (d: \text{RDesc } \top): \text{Number} \rightarrow \text{Type where} leaf: RField d \rightarrow \text{Fields } d \mathbb{1} node: \forall n \{f: \text{Fin } n \rightarrow \text{Number}\} \rightarrow ((i: \text{Fin } n) \rightarrow \text{Fields } d (f i)) \rightarrow \text{Fields } d (f i)
```

if nested is to work at 1. The node constructor makes sure that if we have collection of Fields, then we can gather them in a field at a higher level. We can then count the subnodes of a given Fields as

```
subnodes : \forall {n} {d : RDesc \top} \rightarrow Fields d n \rightarrow Number subnodes (leaf x) = y (RSize \_x) \lambda \_\rightarrow 1 subnodes (node n f) = y n (subnodes \circ f) where RSize counts the number of recursive fields of a particular branch RSize : (d : RDesc \top) \rightarrow (a : RField d) \rightarrow N RSize (y is) a = length is RSize (y is) a = RSize (y (fst y) (snd y)
```

Note that subnodes effectively keeps the shape of the previous field, but unfolds the recursive fields at the bottom of the tree by one level.

I then tried and realized how unpleasant even the functions from the original type to the nested type are to write.

As a trivialty, we get that any type, interpreted as a container, always decomposes as an ornament over a "numerical" base type. This links to the construction of binary heaps in [KG16], as in hindsight, starting from the usual binary heaps would yield binary numbers and their binary heap ornament (in a much less useful package).

4 Reducing friction (work in progress)

The setup some approaches in earlier sections require makes them tedious or impractical to apply. In this section we will look at some ways how part of this problem could be alleviated through generics, or by alternative descriptions of concepts like equivalences through the lens of initial algebras.

In later sections we will construct many more equivalences between more complicated types than before, so we will dive right into the latter. Reflecting upon section 2, we see that when one establishes an equivalence, most of the time is spent working out a series of tedious lemmas to show that the conversion functions are mutual inverses, which tend to be relatively easy to define. We take away two things from this; the first is that the conversion functions are perhaps too obvious, and the second is that we should really avoid talking about sections and retractions lest we incur tedium! In fact, the machinery from [KG16] will come in handy to demonstrate to what extent our hand was actually forced in section 2.

We note that like how $\mathbb N$ is the fixpoint of NatD, one of the sides of the equivalence is almost always going to be a datatype, and hence the μ of a Desc', so we will use this as the working assumption for the following section.

4.1 Well-founded monic algebras are initial

Unfortunately, the machinery from [KG16] relies on axiom K for a small but crucial part. To be precise, in a cubical setting, the type μ as given stops being initial for its base functor! In this section, we will be working with a simplified and repaired version. Namely, we simplify Desc' to

```
data \mathsf{Desc'}: \mathsf{Set}_1 \mathsf{where}
\mathsf{v}: (n: \mathbb{N}) \to \mathsf{Desc'}
\sigma: (S: \mathsf{Set}) (D: S \to \mathsf{Desc'}) \to \mathsf{Desc'}
To complete the definition of \mathsf{\mu}
data \mathsf{\mu} (D: \mathsf{Desc'}): \mathsf{Set}_1 \mathsf{where}
\mathsf{con}: \mathsf{Base} (\mathsf{\mu} D) D \to \mathsf{\mu} D
```

we will need to implement Base. We remark that in [KG16], the recursion of mapFold is a structural descent in [D'] (μ D). Because [_] is a function and

Or at least, that was where I was trying to go with this, but I notice that this still is a bit further away.

 $^{^{7}}$ The latter perhaps less so, because it is useful to show a map to be monic.

not a data type, this descent becomes invalid, and <code>mapFold</code> fails the termination check. We resolve this by defining <code>Base</code> as a data type

```
data Base (X : \mathsf{Set}_1) : \mathsf{Desc}' \to \mathsf{Set}_1 where in-\mathsf{v} : \forall \{n\} \to \mathsf{Vec} \ X \ n \to \mathsf{Base} \ X \ (\mathsf{v} \ n) in-\sigma : \forall \{S \ D\} \to \Sigma [\ s \in S \ ] \ (\mathsf{Base} \ X \ (D \ s)) \to \mathsf{Base} \ X \ (\sigma \ S \ D)
```

such that this descent is allowed by the termination checker without axiom K.⁸ Recall that the Base functors of descriptions are special polynomial functors, and the fixpoint of a base functor is its initial algebra. The situation so far is summarized by the diagram

$$F\mu_F$$

$$\downarrow^{\text{con}}$$

$$X \longleftrightarrow --- \to \mu F$$

so, we are looking for sufficient conditions on X to get the equivalence $e: X \cong \mu F$. Note that when $X \cong \mu F$, then there necessarily is an initial algebra $FX \to X$. Conversely, if the algebra (X,f) is isomorphic to $(\mu F, \operatorname{con})$, then $X \cong \mu F$ would follow immediately, so it is equivalent to ask for the algebras to be isomorphic instead.

4.1.1 Algebras of endofunctors

To characterize when such algebras are isomorphic, we reiterate some basic category theory, simultaneously rephrasing it in Agda terms.⁹

Let C be a category, and let a,b,c be objects of C, so that in particular we have identity arrows $1_a:a\to a$ and for arrows $g:b\to c,f:a\to b$ composite arrows $gf:a\to c$ subject to associativity. In our case, C is the category of types, with ordinary functions as arrows.

Maybe category theory reference

Recall that an endofunctor, which is simply a functor F from C to itself, assigns objects to objects and sends arrows to arrows

```
F_0: \mathsf{Type}\ \ell \to \mathsf{Type}\ \ell

\mathsf{fmap}: (A \to B) \to F_0\ A \to F_0\ B
```

These assignments are subject to the identity and composition laws

f-id :
$$(x: FA)$$
 $\rightarrow \text{mapF} \text{ id } x = x$

f-comp : $(g: B \rightarrow C)$ $(f: A \rightarrow B)$ $(x: FA)$
 $\rightarrow \text{mapF} (g \circ f)$ $x = \text{mapF}$ g $(\text{mapF} f x)$

An F -algebra is just a pair of an object a and an arrow $Fa \rightarrow a$

record Algebra $(F: \text{Type } \ell \rightarrow \text{Type } \ell): \text{Type } (\ell\text{-suc } \ell)$ where field

⁸This has, again by the absence of axiom K, the consequence of pushing the universe levels up by one. However, this is not too troublesome, as equivalences can go between two levels, and indeed types are equivalent to their lifts.

⁹We are not reusing a pre-existing category theory library for the simple reasons that it is not that much work to write out the machinery explicitly, and that such libraries tend to phrase initial objects in the correct way, which is too restrictive for us.

```
Carrier : Type \ell forget : F Carrier \rightarrow Carrier
```

Algebras themselves again form a category C^F . The arrows of C^F are the arrows f of C such that the following square commutes

$$\begin{array}{ccc} Fa & \xrightarrow{Ff} & Fb \\ U_a \downarrow & & \downarrow U_b \\ a & \xrightarrow{f} & b \end{array}$$

So we define

```
\begin{split} & \text{Alg} \rightarrow \text{-Sqr} \ F \ A \ B \ f = f \circ A \ . \text{forget} \equiv B \ . \text{forget} \circ F \ . \text{fmap} \ f \\ & \text{and} \\ & \text{record} \ \text{Alg} \rightarrow (RawF : \text{RawFunctor} \ \ell) \\ & \quad (AlgA \ AlgB : \text{Algebra} \ (RawF \ . \text{F}_0)) : \ \text{Type} \ \ell \ \text{where} \\ & \text{constructor} \ \text{alg} \rightarrow \\ & \text{field} \\ & \text{mor} : AlgA \ . \text{Carrier} \rightarrow AlgB \ . \text{Carrier} \\ & \text{coh} : \ \| \ \text{Alg} \rightarrow \text{-Sqr} \ RawF \ AlgA \ AlgB \ \text{mor} \ \|_1 \end{split}
```

Note that we take the propositional truncation of the square, such that algebra maps with the same underlying morphism become propositionally equal

```
\begin{split} \mathsf{Alg} & \to \mathsf{-Path} : \{F : \mathsf{RawFunctor} \ \ell\} \ \{A \ B : \mathsf{Algebra} \ (F \ .\mathsf{F_0})\} \\ & \to (g \ f : \mathsf{Alg} \to F \ A \ B) \to g \ .\mathsf{mor} \ \equiv f \ .\mathsf{mor} \to g \ \equiv f \end{split}
```

The identity and composition in C^F arise directly from those of the underlying arrows in C.

Recall that an object \emptyset is initial when for each other object a, there is an unique arrow $!: \emptyset \to a$. By reversing the proofs of initiality of μ and the main result of this section, we obtain a slight variation upon the usual definition. Namely, unicity is often expressed as contractability of a type

```
isContr A = \Sigma[x \in A] (\forall y \rightarrow x \equiv y)
```

Instead, we again use a truncation

```
weakContr A = \Sigma[x \in A] (\forall y \rightarrow ||x = y||_1)
```

but note that this also, crucially, slightly stronger than connectedness. We think define initiality for arbitrary relations

```
record Initial (C: \mathsf{Type}\ \ell)\ (R: C \to C \to \mathsf{Type}\ \ell')
(Z: C): \mathsf{Type}\ (\ell\text{-max}\ (\ell\text{-suc}\ \ell)\ \ell') where field initiality: \forall\ X \to \mathsf{weakContr}\ (R\ Z\ X)
```

such that it closely resembles the definition of least element. Then, A is an initial algebra when

```
InitAlg RawF A = Initial (Algebra (RawF .F<sub>0</sub>)) (Alg <math>\rightarrow RawF) A
```

By basic category theory (using the usual definition of initial objects), two initial objects a and b are always isomorphic; namely, initiality guarantees that there are arrows $f:a\to b$ and $g:b\to a$, which by initiality must compose to the identities again.

Similarly, we get that

```
InitAlg-\simeq: (F: Functor \ell) (A B: Algebra (F.RawF .F<sub>0</sub>))

\rightarrow InitAlg (F.RawF) A \rightarrow InitAlg (F.RawF) B

\rightarrow A.Carrier \simeq B.Carrier
```

However, we only have the equalities from the isomorphism inside a propositional truncation. But fortunately, being an equivalence is a property, so we can eliminate from the truncations to get the wanted result.

Note that even though we warned ourselves, we are still talking about sections and retractions to establish that f is an equivalence! However, this result also makes sure we will not have to speak of them again.¹⁰

4.1.2 Accessibility

In terms of "when is $X \cong \mu D$ ", this reduces the problem to showing that X is an initial algebra for the base functor of D; μD is initial by its fold, and by induction on μD using the squares of algebra maps.

Remark 4.1. We need (in general) not hope μD is a strict initial object in the category of algebras. For a strict initial object, having a map $a \to \emptyset$ implies $a \cong \emptyset$. This is not the case here. In general strict initial objects satisfy $a \times \emptyset \cong \emptyset$. Consider the algebras $(\mathbb{N}, (0, \mathrm{suc}))$ and $(2^{\mathbb{N}}, f)$, for which clearly not $2^{\mathbb{N}} \times \mathbb{N} \cong \mathbb{N}$. (In particular, F being a left adjoint is sufficient to make C^F have strict initial objects, but then the carrier of the initial algebra is simply \bot .)

I think

Looking back at section 2, we see that Leibniz is an initial $F: X \mapsto 1 + X$ algebra because for any other algebra, the image of 0b is fixed, and by bsuc all other values are determined by chasing around the square. Thus, we are looking for a similar structure on $f: FX \to X$ that supports recursion.

Clearly we will need something stronger than $FX \cong X$, as in general a functor can have many fixpoints. For this, we define what it means for an element x to be accessible by f. This definition uses a mutually recursive datatype as follows: We state that an element x of X is accessible when there is an accessible y in its fiber over f

```
data Acc D f x where
acc: (y : fiber f x) \rightarrow Acc' D f D (fst y) \rightarrow Acc D f x
```

Accessibility of an element x of Base A E is defined by cases on E; if E is y n and x is a Vec A n, then x is accessible if all its elements are; if x is σ S E', then x is accessible if $\operatorname{snd} x$ is

```
data Acc' D f where

acc-y: All (Acc D f) x \to Acc' D f (y n) (in-y x)

acc-\sigma: Acc' D f (E s) x \to Acc' D f (\sigma S E) (in-\sigma (s, x))
```

Consequently, X is well-founded for an algebra when all its elements are accessible

```
Wf D f = \forall x \rightarrow Acc D f x
```

We can see well-foundedness as an upper bound on the size of X, if it were larger than μD , some of its elements would inevitably get out of reach of an

¹⁰In this section...

algebra. Now having $FX \cong X$ also gives us a lower bound, but remark that having a well-founded injection $f: FX \to X$ is already sufficient, as accessibility makes this an epi and an iso. Thus, we claim

Claim 4.1. If there is a mono $f: FX \to X$ and X is well-founded for f, then X is an initial F-algebra.

4.1.3 Proof sketch of 4.1

Let us be on our way. Suppose X is well-founded for the mono $f: FX \to X$. To show that (X, f) is initial, let us take another algebra (Y, g), and show that there is a unique arrow $(X, f) \to (Y, g)$.

By Acc-recursion and because all x are accessible, we can define a plain map into Y

```
Wf-rec : (D : \mathsf{Desc'}) \ (X : \mathsf{Algebra} \ (\dot{\mathsf{F}} \ D)) \to \mathsf{Wf} \ D \ (X . \mathsf{forget})
 \to (\dot{\mathsf{F}} \ D \ A \to A) \to X . \mathsf{Carrier} \to A
```

This construction is an instance of the broad concept of "well-founded recursion", so we let ourselves be inspired by these methods. In particular, we prove an irrelevance lemma

```
unfold-Wf-rec : \forall x' \rightarrow \text{rec } (\text{cx } x') (\text{cx } x') (wf (\text{cx } x'))
= f (\text{Base-map } (\lambda y \rightarrow \text{rec } y y (wf y)) x')
```

The unfolding lemma ensures that the map we defined by Wf-rec is a map of algebras.

The proof that this map is unique proceeds analogously to that in the proof that μD is initial, but now by Acc-recursion

Thus, we conclude that X is initial.

The main result is then a corollary of initiality of X and the isomorphism of initial objects

```
Wf+inj=\mu: (D: Desc') (X: Algebra (\dot{\mathsf{F}} D)) \rightarrow Wf D (X .forget) \rightarrow injective (X .forget) \rightarrow X .Carrier = \mu D
```

4.1.4 Example

Let us redo the proof in section 2, now using this result.

- 5 Is equivalence too strong (finger trees)
- 6 Discussion and future work (aka the union of my to-do list and the actual future work section)

I think. but I'm not sure how much surjective=epi, injective=mono and "(Type a) is balanced" I can safely sweep under the carpet.

stdlib

7 Temporary

Todo list

	f it would be helpful to have a more extensive introduction	3	
covering all features used			
is this too much code and too little explanation at once?			
Replace with BinOp			
-	determined to cobble together the path over path we	7	
	now	10	
	ike Agda forgets that we defined Leibniz if we move between		
differ	ent .tex files, so I'll have to float all AgdaTargets to the top		
level	at some point	10	
Again not	sure how much space I should use to reiterate Desc, Orn, and		
$Orn \Gamma$		10	
	is can use more explanation (the question is, how much?)	11	
	, that was where I was trying to go with this, but I notice that	4.0	
	till is a bit further away	13	
Maybe category theory reference			
I think			
	at I'm not sure how much surjective=epi, injective=mono and	16	
	be a) is balanced I can safely sweep under the carpet	17	
, , , ,	is balanced. I can salely sweep under the carpet	17	
stano		11	
Refere	naos		
rterere	nces		
[Ang+20]	Carlo Angiuli et al. Internalizing Representation Independence	with	
. 0 1	Univalence. 2020. DOI: 10.48550/ARXIV.2009.05547. URL: htt		
	//arxiv.org/abs/2009.05547.	-	
[DM14]	PIERRE-ÉVARISTE DAGAND and CONOR McBRIDE. "Tr	ans-	
	porting functions across ornaments". In: Journal of Functional		
	gramming 24.2-3 (Apr. 2014), pp. 316–383. DOI: 10.1017/s09567		
	URL: https://doi.org/10.1017%2Fs0956796814000069.		
[EC22]	Lucas Escot and Jesper Cockx. "Practical Generic Programm	ning	
	over a Universe of Native Datatypes". In: <i>Proc. ACM Program. L</i>		
	6.ICFP (Aug. 2022). DOI: 10.1145/3547644. URL: https://d		
	org/10.1145/3547644.		
[HS22]	Ralf Hinze and Wouter Swierstra. "Calculating Datastructures"	In:	
[11044]	Mathematics of Program Construction. Ed. by Ekaterina Kon		
	dantskaya. Cham: Springer International Publishing, 2022, pp.		
	101. ISBN: 978-3-031-16912-0.	02	
	101. 1021 010 0 001 10012 0.		

- [KG16] HSIANG-SHANG KO and JEREMY GIBBONS. "Programming with ornaments". In: *Journal of Functional Programming* 27 (2016), e2. DOI: 10.1017/S0956796816000307.
- $[{\rm McB14}]$ Conor McBride. "Ornamental Algebras, Algebraic Ornaments". In: 2014.
- [Oka98] Chris Okasaki. Purely Functional Data Structures. USA: Cambridge University Press, 1998. ISBN: 0521631246.
- [Tea23] Agda Development Team. Agda. 2023. URL: https://agda.readthedocs.io/en/v2.6.3/.
- [VMA19] Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. "Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types". In: *Proc. ACM Program. Lang.* 3.ICFP (July 2019). DOI: 10.1145/3341691. URL: https://doi.org/10.1145/3341691.